CALCULS de MECANIQUE CELESTE avec MAPLE

appliqués aux mouvements des planètes et des satellites

Luc Duriez

Laboratoire d'Astronomie de l'Université de Lille 1 et IMCCE de l'Observatoire de Paris

Luc.Duriez@univ-lille1.fr

Dernière révision le avril 2007

Ce travail a pour but de donner des procédures et des instructions MAPLE permettant d'effectuer des développements analytiques sur un certain nombre de quantités relatives au mouvements des planètes et des satellites : Développements du mouvement képlérien, développements de fonctions perturbatrices de planètes et de satellites et des équations de Lagrange permettant de calculer analytiquement ces perturbations. C'est aussi une illustration du Cours de Mécanique Céleste enseigné dans la License de Mathématique à l'Université de Lille1. Les notions utilisées dans ce document d'applications pourront être davantage développées dans ce cours. Quand ce sera le cas, un renvoi vers la section correspondante de ce cours sera indiqué comme ici par une balise placée dans la marge (renvoi exécutable aussi par un "clic" sur celle-ci): Une application pratique des techniques exposées dans ce document montre comment construire les perturbations des éléments d'orbite des planètes du système solaire à l'ordre 1 des masses. Un renvoi éventuel vers cette application sera indiqué par cette autre balise: C Pla

Le présent document est une copie des résultats obtenus en exécutant le programme Maple "DevMC_Duriez.mws", adapté toutefois pour que ces résultats soient affichables à l'écran de manière raisonnable; on a ainsi en général tronqué les développements analytiques aux termes principaux (de plus bas degré par exemple s'il s'agit de développements en séries entières). C'est pourquoi apparaissent de nombreuses indications sur la façon de calculer, bien que ce fichier PDF ne soit pas exécutable par Maple.

Si vous disposez du logiciel Maple version 5 ou ultérieure, vous pourrez exécuter complètement le programme joint: "DevMC_Duriez.mws" et le modifier éventuellement pour faire vos propres expériences de calcul; alors, avant de démarrer tout calcul, il convient d'exécuter les quatre instructions suivantes, en indiquant notamment le *chemin* du répertoire où l'on a placé les fichiers de procédures (*.pr) (dont par exemple le fichier "Serutil.pr" qui contient des sous-programmes utiles à la manipulation de séries, voir l'Annexe); c'est aussi à cet endroit que seront sauvegardées les séries lors d'instructions *save* . Anx

```
> restart:
> chemin:='Maple/';
```

$$chemin := Maple/$$

```
> read cat(chemin,'Serutil.pr'):
> readlib(poisson):readlib(mtaylor):
```

1. Introduction

Le but de ce fichier MAPLE est de montrer comment on peut utiliser ce logiciel de calcul formel pour développer des expressions mathématiques rencontrées en Mécanique Céleste lorsqu'on veut représenter le mouvement des planètes et de leurs satellites.

A la base de ces représentations, il y a le **mouvement elliptique képlérien** d'un corps orbitant autour d'un autre, qu'on décrit ici par les **éléments d'orbite classiques** : a demi-grand axe, e excentricité, i inclinaison du plan de l'orbite sur un plan fixe de référence, Ω longitude du nœud sur ce plan, ω argument du péricentre et v anomalie vraie. (**R**emarque importante:dans le cours l'anomalie vraie est notée w). $C_{12.1}$

Ce mouvement périodique s'effectue avec la vitesse angulaire moyenne n reliée à a par la 3ième loi de Kepler $n^2\,a^3 = \mu$ où μ est la constante d'attraction. On introduit alors l'anomalie moyenne M telle que $\frac{dM}{dt} = n$, et les longitudes vraie $\lambda = \Omega + \omega + v$ et moyenne $L = \Omega + \omega + M$, ainsi que la longitude du péricentre $\varpi = \Omega + \omega$. On a aussi l'expression de la distance r au foyer en fonction de l'anomalie excentrique E: $r = a(1 - e\cos E)$ et l'équation de Kepler $M = E - e\sin E$. $C_{(3.25)}$

On ne veut pas ici reconstruire toute la formulation de ce mouvement depuis le début, mais partir d'un formulaire de base qui peut se résumer en fait aux deux expressions précédentes de r et de M. C'est l'objet de la section 2, où diverses quantités utiles pour représenter les mouvements d'un corps sont exprimées en fonctions des éléments de son orbite, et notamment les **coefficients de Hansen**. C13.6

Les développements du mouvement képlérien ainsi construits, on les utilise dans les section 3 et 4 pour exprimer la fonction perturbatrice d'une planète perturbée par une autre, ou celle d'un satellite perturbé par la non-sphéricité de sa planète. On expose dans chaque cas plusieurs méthodes, mais l'essentiel est de voir comment on peut utiliser MAPLE pour déterminer directement certaines parties de ces fonctions perturbatrices, sans avoir à en calculer l'ensemble : C'est le **calcul direct d'une inégalité donnée**, cette notion d'inégalité étant simplement celle du regroupement dans un développement de Fourier, des termes de même argument, par exemple ceux de la "grande inégalité": $(2\,L_{Jupiter} - 5\,L_{Saturne})$. On peut même faire le développement formel non plus d'une inégalité isolée, C mais de toute une classe d'inégalités, le seul choix étant leur caractéristique (par exemple on exprime explicitement l'ensemble des inégalités de la forme $(p\,L_{Jupiter} - (p+3)\,L_{Saturne})$ où p reste indéterminé).

Enfin, la section 5 permet d'utiliser ces fonctions perturbatrices pour exprimer les équations du mouvement perturbé des planètes et des satellites. On se contente ici des équations de Lagrange, et on donne des indications pour passer à une formulation hamiltonienne.

Un autre fichier (PlaMC_Duriez.mws) complète le présent document et indiqué par cette balise: Il explicite le calcul des perturbations séculaires des planètes et montre comment obtenir des perturbations périodiques choisies (par exemple pour leur importance ou parce qu'elles peuvent concerner une résonance). Ce fichier est simplement construit à partir d'une sélection des instructions MAPLE prises dans le présent document, sélection nécessaire et suffisante pour réaliser le calcul des équations de Lagrange dans un problème de type planétaire. On pourra sans doute s'en inspirer pour traiter de manière analogue des problèmes de satellites. Pla

2. Développements du mouvement képlérien de planètes ou de satellites

2.1. Introduction

On considère le mouvement képlérien elliptique de planètes autour du Soleil ou de satellites autour d'une planète. Ce mouvement est au départ représenté pour chaque corps par des éléments osculateurs classiques ($a, e, i, L, \varpi, \Omega$). On se propose d'exprimer diverses quantités intervenant dans le calcul de position des planètes ou des satellites, en fonction de ces éléments ou en fonction d'autres variables plus appropriées qui leur sont liées. On utilise essentiellement les propriétés exposées dans les références [1] et [2] . C(3.46)

2.2. Développements en excentricité du mouvement képlérien d'une planète

On développe en séries de Fourier les quantités $\frac{a}{r}$, $\frac{r}{a}$ et l'équation du centre $(v - M)$ en fonction de l'excentricité e et de l'anomalie moyenne M au voisinage de $e = 0$. On transforme ces développements trigonométriques en séries entières en adoptant les variables complexes non singulières: C13.8

$$X = e\ \exp\left(I\,M\right) \qquad \text{et son conjugué} \qquad Xb = e\ \exp\left(-I\,M\right)$$

On en déduit le calcul des **coefficients de Hansen**, $X_{k,\,n,\,m}(e)$, qui sont les coefficients du développement en série de Fourier de la quantité : C13.6

$$(\frac{r}{a})^n\ \exp\left(I\,m\,(v - M)\right) = \sum_k X_{k,\,n,\,m}(e)\ \exp\left(I\,(k - m)\,M\right)$$

en fonction de $(e,\ M)$ ou de $(X,\ Xb)$.

Avec $M = L - \varpi$, différence des longitudes moyenne et du péricentre, on introduit aussi la variable $x = e\ \exp(I\varpi)$ et son conjugué noté xb , puisqu'on a: C13.9

$$X = e\ \exp(I(L - \varpi)) \qquad \text{soit encore :} \qquad X = xb\ \exp(I\,L) \qquad \text{et} \qquad Xb = x\ \exp(-I\,L)$$

2.2.1. Initialisations

On commence par définir le degré maximun souhaité pour les développements en excentricité. On choisit ici degmax=20 pour montrer qu'on peut atteindre facilement des degrés élevés. Cependant, au delà du degré 12, le calcul 'direct' des coefficients de Hansen peut nécessiter trop de temps de calcul et de mémoire; on verra comment il convient alors d'organiser les calculs autrement. Pour l'affichage des résultats, on limite généralement le degré à une valeur plus faible (la procédure PRINTRONC affiche une série tronquée au degré maxi souhaité mais en montrant aussi le terme de degré le plus élevé)

Anx2.1

```
> degmax:=20; degmaxprint:=5;
```

$$degmax := 20$$

$$degmaxprint := 5$$

2.2.2. Calcul de a/r et de l'équation du centre v-M

On construit d'abord le développement en série de Fourier de (a/r), sachant que le mouvement képlérien s'exprime en fonction de l'anomalie excentrique E par les relations :

$$M = E - e\,\sin(E) \qquad \text{et} \qquad r = a(1-\cos E) = a\,\frac{dM}{dE} \qquad \text{d'où} \qquad \frac{a}{r} = \frac{dM}{dE}$$

et :

$$\frac{a}{r} = 1 + \frac{2}{\pi}\sum_{n=1}^{\infty}\int_0^{\pi}\cos(n\,(E - e\sin(E)))\,dE\,\cos(n\,M) \qquad \text{soit:} \qquad \frac{a}{r} = 1 + \sum_{n=1}^{\infty} 2\,J_n(n\,e)\cos(n\,M)$$

où interviennent les fonctions de Bessel de première espèce $J_n(x)$. C13.2

```
> 'a/r':=mtaylor(1+2*sum(BesselJ(n,n*e)*cos(n*M),n=1..degmax),[e],degma
> x+1):  PRINTRONC('a/r',degmaxprint,e);
```

$$\Big(-\frac{152587890625}{125536739328}\cos(10\,M) - \frac{1594323}{7175168000}\cos(6\,M) - \frac{1}{7242504192000}\cos(2\,M) - \frac{1628413597910449}{21776781312000}\cos(14\,M)$$
$$+ \frac{16}{147349125}\cos(4\,M) + \frac{67108864}{1915538625}\cos(8\,M) + \frac{17592186044416}{97692469875}\cos(16\,M)$$
$$- \frac{1853020188851841}{9270317056000}\cos(18\,M) + \frac{1220703125000}{14849255421}\cos(20\,M) + \frac{12754584}{875875}\cos(12\,M)\Big)e^{20},\ \ + \ldots + ,$$
$$\Big(\frac{1}{192}\cos(M) + \frac{625}{384}\cos(5\,M) - \frac{81}{128}\cos(3\,M)\Big)\,e^5 + \Big(\frac{4}{3}\cos(4\,M) - \frac{1}{3}\cos(2\,M)\Big)\,e^4 + \Big(-\frac{1}{8}\cos(M) + \frac{9}{8}\cos(3\,M)\Big)\,e^3$$
$$+ \cos(2\,M)\,e^2 + \cos(M)\,e + 1$$

L'équation du centre est obtenue en intégrant de l'équation $\frac{dv}{dM} = (\frac{a}{r})^2 \sqrt{1-e^2}$ issue de l'intégrale des aires : $r^2 \, (\frac{dv}{dt}) = C = n\, a^2 \sqrt{1-e^2}$

```
> 'v-M':=int(poisson('a/r'^2*sqrt(1-e*e),[e],degmax+1),M)- M:
> PRINTRONC('v-M',degmaxprint,e);
```

$$(\frac{13750247892461}{502146957312000}\sin(8\,M) - \frac{7292605650809}{10042939146240}\sin(10\,M) + \frac{13476493246767}{1836843008000}\sin(12\,M) + \frac{1422154873037388137}{20487595858329600}\sin(16\,M) - \frac{454535614962827}{13937140039680}\sin(14\,M) - \frac{10261083179603471}{148325072896000}\sin(18\,M) + \frac{4027894135040576041}{1557057285233049600}\sin(20\,M) + \frac{35916051}{22960537600}\sin(6\,M) + \frac{426752653741}{115880067072000}\sin(2\,M) + \frac{72929638721}{30901351219200}\sin(4\,M))e^{20}, \; + \ldots + , (\frac{5}{96}\sin(M) + \frac{1097}{960}\sin(5\,M) - \frac{43}{64}\sin(3\,M))\,e^5 + (-\frac{11}{24}\sin(2\,M) + \frac{103}{96}\sin(4\,M))\,e^4 + (-\frac{1}{4}\sin(M) + \frac{13}{12}\sin(3\,M))\,e^3 + \frac{5}{4}\sin(2\,M)\,e^2 + 2\sin(M)\,e$$

On peut convertir ces développements en fonction de $X = e\,\exp(I M)$ et de son conjugué Xb par la procédure suivante :

```
> ENXXB:=proc(u,degmx) local k,ux;
> ux:=u:
> for k from 1 to degmx do:
> ux:=subs(cos(k*M)=(X^k+Xb^k)/2/e^k,sin(k*M)=(X^k-Xb^k)/2/I/e^k,ux):
> od:
> ux:=algsubs(e^2=X*Xb,expand(ux)):
> ux:=sort(ux,[X,Xb]) :
> end:
```

C(3.145)

Applications:

```
> ASR:=ENXXB(‘a/r‘,degmax): PRINTRONC(ASR,5,[X,Xb]);
```

$$\frac{610351562500}{14849255421}X^{20},\ +\ ...\ +\ ,\frac{625}{768}X^5-\frac{81}{256}X^4\,Xb+\frac{1}{384}X^3\,Xb^2+\frac{1}{384}X^2\,Xb^3-\frac{81}{256}X\,Xb^4+\frac{625}{768}Xb^5+\frac{2}{3}X^4-\frac{1}{6}X^3\,Xb-\frac{1}{6}X\,Xb^3+\frac{2}{3}Xb^4+\frac{9}{16}X^3-\frac{1}{16}X^2\,Xb-\frac{1}{16}X\,Xb^2+\frac{9}{16}Xb^3+\frac{1}{2}X^2+\frac{1}{2}Xb^2+\frac{1}{2}X+\frac{1}{2}Xb+1$$

```
> VMM:=ENXXB(‘v-M‘,degmax): PRINTRONC(VMM,5,[X,Xb]);
```

$$-\frac{4027894135040576041}{311411457046609920}I\,X^{20},\ +\ ...\ +\ ,-\frac{1097}{1920}I\,X^5+\frac{43}{128}I\,X^4\,Xb-\frac{5}{192}I\,X^3\,Xb^2+\frac{5}{192}I\,X^2\,Xb^3-\frac{43}{128}I\,X\,Xb^4+\frac{1097}{1920}I\,Xb^5-\frac{103}{192}I\,X^4+\frac{11}{48}I\,X^3\,Xb-\frac{11}{48}I\,X\,Xb^3+\frac{103}{192}I\,Xb^4-\frac{13}{24}I\,X^3+\frac{1}{8}I\,X^2\,Xb-\frac{1}{8}I\,X\,Xb^2+\frac{13}{24}I\,Xb^3-\frac{5}{8}I\,X^2+\frac{5}{8}I\,Xb^2-I\,X+I\,Xb$$

La conversion inverse (mais en exponentielles complexes de *M*) s'obtiendrait par la procédure suivante :

```
> ENeEXPM:=proc(u);
> subs(X=e*exp(I*M),Xb=e/exp(I*M),u): collect(combine(%,exp),exp):
> end:
```

Exemple :

```
> TRONC(ENeEXPM(ASR),5,e);
```

$$\left(\frac{1}{384}e^{(-I\,M)}-\frac{81}{256}e^{(3\,I\,M)}-\frac{81}{256}e^{(-3\,I\,M)}+\frac{1}{384}e^{(I\,M)}+\frac{625}{768}e^{(-5\,I\,M)}+\frac{625}{768}e^{(5\,I\,M)}\right)e^5+\left(\frac{2}{3}e^{(4\,I\,M)}-\frac{1}{6}e^{(2\,I\,M)}+\frac{2}{3}e^{(-4\,I\,M)}-\frac{1}{6}e^{(-2\,I\,M)}\right)e^4+\left(\frac{9}{16}e^{(-3\,I\,M)}-\frac{1}{16}e^{(-I\,M)}-\frac{1}{16}e^{(I\,M)}+\frac{9}{16}e^{(3\,I\,M)}\right)e^3+\left(\frac{1}{2}e^{(-2\,I\,M)}+\frac{1}{2}e^{(2\,I\,M)}\right)e^2+\left(\frac{1}{2}e^{(I\,M)}+\frac{1}{2}e^{(-I\,M)}\right)e+1$$

Pour vérification, on revient aux fonctions trigonométriques: Il suffit d'appliquer la fonction de conversion :

```
> convert(%,trig);
```

$$(\frac{1}{192}\cos(M)+\frac{625}{384}\cos(5\,M)-\frac{81}{128}\cos(3\,M))\,e^5+(\frac{4}{3}\cos(4\,M)-\frac{1}{3}\cos(2\,M))\,e^4+(-\frac{1}{8}\cos(M)+\frac{9}{8}\cos(3\,M))\,e^3$$
$$+\cos(2\,M)\,e^2+\cos(M)\,e+1$$

2.2.3. Calcul de r/a et de theta = exp(I(v-M))

Si *degmax* est assez petit, l'inverse de (a/r) peut encore s'obtenir simplement avec la procédure *poisson* , sinon pour limiter le temps de calcul, il vaut mieux utiliser les variables $X,\ Xb$ et utiliser la procédure *mtaylor*:

```
> if degmax<13 then 'r/a':=poisson(1/'a/r', [e], degmax+1):
> RSA:=ENXXB('r/a',degmax):
> else RSA:=sort(mtaylor(1/ASR,[X,Xb],degmax+1),[X,Xb]):
> print(nops(%));
> fi:  'r/a':=ENeEXPM(RSA): PRINTRONC(RSA,5,[X,Xb]);
```

$$222$$

$$-\frac{30517578125}{14849255421}X^{20},\ +\ldots+,-\frac{125}{768}X^5+\frac{45}{256}X^4\,Xb-\frac{5}{384}X^3\,Xb^2-\frac{5}{384}X^2\,Xb^3+\frac{45}{256}X\,Xb^4-\frac{125}{768}Xb^5-\frac{1}{6}X^4+\frac{1}{6}X^3\,Xb$$
$$+\frac{1}{6}X\,Xb^3-\frac{1}{6}Xb^4-\frac{3}{16}X^3+\frac{3}{16}X^2\,Xb+\frac{3}{16}X\,Xb^2-\frac{3}{16}Xb^3-\frac{1}{4}X^2+\frac{1}{2}X\,Xb-\frac{1}{4}Xb^2-\frac{1}{2}X-\frac{1}{2}Xb+1$$

La fonction : $\theta=\exp\left(I\,(v-M)\right)$ intervient souvent dans l'expression des perturbations. Pour la développer, on procède comme précédemment:

```
> if degmax <13 then Theta:=poisson(exp(I*'v-M'),[e],degmax+1):
```

```
> THETA:=ENXXB(Theta,degmax):
> else THETA:=sort(mtaylor(exp(I*VMM),[X,Xb],degmax+1),[X,Xb]):
> print(nops(%));
> fi:  Theta:=ENeEXPM(THETA):  TRONC(Theta,5,e);
```

$$222$$

$$\left(-\frac{1}{15}e^{(-5\,I\,M)}+\frac{17}{48}e^{(I\,M)}-\frac{7}{3}e^{(3\,I\,M)}+\frac{81}{40}e^{(5\,I\,M)}+\frac{1}{48}e^{(-3\,I\,M)}\right)e^5 + \left(\frac{625}{384}e^{(4\,I\,M)}+\frac{7}{64}+\frac{1}{48}e^{(-2\,I\,M)}-\frac{9}{128}e^{(-4\,I\,M)}-\frac{27}{16}e^{(2\,I\,M)}\right)e^4+\left(-\frac{5}{4}e^{(I\,M)}-\frac{1}{12}e^{(-3\,I\,M)}+\frac{4}{3}e^{(3\,I\,M)}\right)e^3 + \left(-\frac{1}{8}e^{(-2\,I\,M)}-1+\frac{9}{8}e^{(2\,I\,M)}\right)e^2+\left(-e^{(-I\,M)}+e^{(I\,M)}\right)e+1$$

On aura aussi besoin du conjugué *THETAB* de θ . On l'obtient simplement en inversant les roles de X et de Xb dans θ et on vérifie que l'on a bien $\theta\,\overline{\theta}=1$ (au degré *degmax* considéré).

```
> THETAB:=subs(X=XX,Xb=X,XX=Xb,THETA):
> PRINTRONC(THETA,3,[X,Xb]); PRINTRONC(THETAB,3,[X,Xb]);
> PRODPOL(THETA,THETAB,degmax,[X,Xb]);
```

$$\frac{86540575088712692709}{79352097773977600000}X^{20},\ +\ ...\ +\ ,\ \frac{4}{3}X^3-\frac{5}{4}X^2\,Xb-\frac{1}{12}Xb^3+\frac{9}{8}X^2-X\,Xb-\frac{1}{8}Xb^2+X-Xb+1$$

$$-\frac{104127350297911241532841}{134267508217148866560000}X^{20},\ +\ ...\ +\ ,\ -\frac{1}{12}X^3-\frac{5}{4}X\,Xb^2+\frac{4}{3}Xb^3-\frac{1}{8}X^2-X\,Xb+\frac{9}{8}Xb^2-X+Xb+1$$

$$1$$

On termine en reformulant (a/r) en exponentielles complexes de M :

```
> 'a/r':=ENeEXPM(ASR): TRONC('a/r',5,e);
```

$$\left(\frac{625}{768}\,e^{(-5\,I\,M)} - \frac{81}{256}\,e^{(-3\,I\,M)} + \frac{1}{384}\,e^{(I\,M)} - \frac{81}{256}\,e^{(3\,I\,M)} + \frac{1}{384}\,e^{(-I\,M)} + \frac{625}{768}\,e^{(5\,I\,M)}\right)e^5 + \left(-\frac{1}{6}\,e^{(2\,I\,M)} + \frac{2}{3}\,e^{(-4\,I\,M)} - \frac{1}{6}\,e^{(-2\,I\,M)} + \frac{2}{3}\,e^{(4\,I\,M)}\right)e^4 + \left(-\frac{1}{16}\,e^{(I\,M)} - \frac{1}{16}\,e^{(-I\,M)} + \frac{9}{16}\,e^{(-3\,I\,M)} + \frac{9}{16}\,e^{(3\,I\,M)}\right)e^3 + \left(\frac{1}{2}\,e^{(-2\,I\,M)} + \frac{1}{2}\,e^{(2\,I\,M)}\right)e^2 + \left(\frac{1}{2}\,e^{(I\,M)} + \frac{1}{2}\,e^{(-I\,M)}\right)e + 1$$

Sauvegarde des résultats :

```
> save 'a/r', 'r/a', Theta, ASR, RSA, THETA, THETAB,
> cat(chemin,'kepler'.degmax.'.m');
```

2.2.4. Exemples d'utilisation et propriété de d'Alembert

On suppose ici que le calcul de (r/a), (a/r) et θ ont déjà été réalisés au degré 20 et enregistrés :

```
> degmax:=20:  read  cat(chemin,'kepler20.m'):
```

On peut cependant éventuellement redéfinir *degmax* :

```
> degmax:=6;
```

$$degmax := 6$$

On a construit les développements du mouvement képlérien sous deux formes, exemples :

```
> TRONC(ASR,4,[X,Xb]);
```

$$\frac{2}{3}X^4-\frac{1}{6}X^3\,Xb-\frac{1}{6}X\,Xb^3+\frac{2}{3}Xb^4+\frac{9}{16}X^3-\frac{1}{16}X^2\,Xb-\frac{1}{16}X\,Xb^2+\frac{9}{16}Xb^3+\frac{1}{2}X^2+\frac{1}{2}Xb^2+\frac{1}{2}X+\frac{1}{2}Xb+1$$

```
> TRONC('a/r',4,e);
```

$$(-\frac{1}{6}e^{(-2\,I\,M)}+\frac{2}{3}e^{(-4\,I\,M)}-\frac{1}{6}e^{(2\,I\,M)}+\frac{2}{3}e^{(4\,I\,M)})\,e^4+(-\frac{1}{16}e^{(-I\,M)}+\frac{9}{16}e^{(-3\,I\,M)}+\frac{9}{16}e^{(3\,I\,M)}-\frac{1}{16}e^{(I\,M)})\,e^3$$
$$+(\frac{1}{2}e^{(-2\,I\,M)}+\frac{1}{2}e^{(2\,I\,M)})\,e^2+(\frac{1}{2}e^{(I\,M)}+\frac{1}{2}e^{(-I\,M)})\,e+1$$

```
> TRONC(RSA,4,[X,Xb]);
```

$$-\frac{1}{6}X^4+\frac{1}{6}X^3\,Xb+\frac{1}{6}X\,Xb^3-\frac{1}{6}Xb^4-\frac{3}{16}X^3+\frac{3}{16}X^2\,Xb+\frac{3}{16}X\,Xb^2-\frac{3}{16}Xb^3-\frac{1}{4}X^2+\frac{1}{2}X\,Xb$$
$$-\frac{1}{4}Xb^2-\frac{1}{2}X-\frac{1}{2}Xb+1$$

```
> TRONC('r/a',4,e);
```

$$(-\frac{1}{6}e^{(4\,I\,M)}+\frac{1}{6}e^{(-2\,I\,M)}+\frac{1}{6}e^{(2\,I\,M)}-\frac{1}{6}e^{(-4\,I\,M)})\,e^4+(-\frac{3}{16}e^{(-3\,I\,M)}+\frac{3}{16}e^{(I\,M)}-\frac{3}{16}e^{(3\,I\,M)}+\frac{3}{16}e^{(-I\,M)})\,e^3$$
$$+(\frac{1}{2}-\frac{1}{4}e^{(2\,I\,M)}-\frac{1}{4}e^{(-2\,I\,M)})\,e^2+(-\frac{1}{2}e^{(I\,M)}-\frac{1}{2}e^{(-I\,M)})\,e+1$$

On peut en déduire par exemple les développements de $(r/a)\cos(v)$ et $(r/a)\sin(v)$, ou d'autres puissances de ces expressions :

```
> e:='e':M:='M':
> '(r/a)*exp(I*v)':=convert(combine(expand(
> mtaylor('r/a'*Theta,[e],degmax+1)*exp(I*M)),exp),trig):
> assume(e,real):assume(M,real):
> '(r/a)*cos(v)':=subs(e='e',M='M',collect(Re('(r/a)*exp(I*v)'),cos)):
> '(r/a)*sin(v)':=subs(e='e',M='M',collect(Im('(r/a)*exp(I*v)'),sin)):
```

```
> e:='e':M:='M':
> PRINTRONC(`(r/a)*cos(v)`,4,e);
```

$$\left(-\frac{4375}{9216}\cos(5\,M)+\frac{16807}{46080}\cos(7\,M)+\frac{567}{5120}\cos(3\,M)-\frac{7}{9216}\cos(M)\right)e^6,\ +\ldots+\,,$$
$$\left(\frac{5}{192}\cos(M)+\frac{125}{384}\cos(5\,M)-\frac{45}{128}\cos(3\,M)\right)e^4+\left(\frac{1}{3}\cos(4\,M)-\frac{1}{3}\cos(2\,M)\right)e^3+\left(-\frac{3}{8}\cos(M)+\frac{3}{8}\cos(3\,M)\right)e^2$$
$$+\left(\frac{1}{2}\cos(2\,M)-\frac{3}{2}\right)e+\cos(M)$$

```
> PRINTRONC(`(r/a)*sin(v)`,4,e);
```

$$\left(-\frac{457}{9216}\sin(M)-\frac{4625}{9216}\sin(5\,M)+\frac{16807}{46080}\sin(7\,M)+\frac{543}{5120}\sin(3\,M)\right)e^6,\ +\ldots+\,,$$
$$\left(-\frac{11}{192}\sin(M)+\frac{125}{384}\sin(5\,M)-\frac{51}{128}\sin(3\,M)\right)e^4+\left(-\frac{5}{12}\sin(2\,M)+\frac{1}{3}\sin(4\,M)\right)e^3+\left(-\frac{5}{8}\sin(M)+\frac{3}{8}\sin(3\,M)\right)e^2$$
$$+\frac{1}{2}\sin(2\,M)\,e+\sin(M)$$

On prend un autre exemple avec les développements de $(a/r)^3 * \sin(2v)$ et $(a/r)^3 * \cos(2v)$:

```
> e:='e':  M:='M':
> `(a/r)^3*exp(2*I*v)`:=convert(combine(expand(
> mtaylor(`a/r`^3*Theta^2,[e],degmax+1)*exp(I*M)^2),exp),trig):
> assume(e,real):assume(M,real):
> `(a/r)^3*cos(2v)`:=subs(e='e',M='M',collect(Re(`(a/r)^3*exp(2*I*v)`),
> cos)):
> `(a/r)^3*sin(2v)`:=subs(e='e',M='M',collect(Im(`(a/r)^3*exp(2*I*v)`),s
> in)):
> e:='e':  M:='M':
```

```
> PRINTRONC(‘(a/r)^3*cos(2v)‘,4,e);
```

$$\left(\frac{73369}{720}\cos(8\,M) - \frac{13827}{160}\cos(6\,M) - \frac{133}{1440}\cos(2\,M) + \frac{9079}{720}\cos(4\,M)\right)e^6,\ +\ldots+\,,$$
$$\left(\frac{533}{16}\cos(6\,M) + \frac{41}{48}\cos(2\,M) - \frac{115}{6}\cos(4\,M)\right)e^4 + \left(\frac{1}{12}\cos(M) - \frac{123}{16}\cos(3\,M) + \frac{845}{48}\cos(5\,M)\right)e^3$$
$$+ \left(-\frac{5}{2}\cos(2\,M) + \frac{17}{2}\cos(4\,M)\right)e^2 + \left(-\frac{1}{2}\cos(M) + \frac{7}{2}\cos(3\,M)\right)e + \cos(2\,M)$$

```
> PRINTRONC(‘(a/r)^3*sin(2v)‘,4,e);
```

$$\left(-\frac{13827}{160}\sin(6\,M) - \frac{217}{1440}\sin(2\,M) + \frac{8951}{720}\sin(4\,M) + \frac{73369}{720}\sin(8\,M)\right)e^6,\ +\ldots+\,,$$
$$\left(\frac{37}{48}\sin(2\,M) - \frac{115}{6}\sin(4\,M) + \frac{533}{16}\sin(6\,M)\right)e^4 + \left(-\frac{123}{16}\sin(3\,M) + \frac{1}{24}\sin(M) + \frac{845}{48}\sin(5\,M)\right)e^3$$
$$+ \left(-\frac{5}{2}\sin(2\,M) + \frac{17}{2}\sin(4\,M)\right)e^2 + \left(-\frac{1}{2}\sin(M) + \frac{7}{2}\sin(3\,M)\right)e + \sin(2\,M)$$

ATTENTION: seules les fonctions (a/r), θ et leurs produits ou puissances son exprimables sous forme de polynômes en (X, Xb) car leurs développements vérifient la propriété de d'Alembert de rang 0 . Rappelons qu'une série $S_n(e,\,M)$ vérifie la propriété de d'Alembert de rang n si elle se met sous la forme : C13.5

$$S_n(e,\,M) = \sum_{k=-\infty}^{\infty} e^{|n-k|}\, f_k(e^2)\, \exp\left(I\,k\,M\right)$$

où $f_k(e^2)$ est d'ordre 0 en excentricité; le terme prépondérant d'une telle série (de degré 0 en excentricité) est en facteur de $\exp\left(I\,n\,M\right)$. Une série de rang zéro est alors exprimable en fonction de X et Xb uniquement puisque $e^{|k|}\,e^{(I\,k\,M)}$ vaut X^k ou Xb^k suivant le signe de k et puisque $f_k(e^2)$ vaut $f_k(X\,Xb)$.

Si on applique par exemple ENXXB à $(r/a)\cos(v)$ qui vérifie la propriété de d'Alembert de rang 1, on obtient une expression où subsistent encore les variables e et M:

```
> ENXXB:=proc(u,degmx) local k,ux;
```

```
> ux:=u:
> for k from 1 to degmx do:
> ux:=subs(cos(k*M)=(X^k+Xb^k)/2/e^k,sin(k*M)=(X^k-Xb^k)/2/I/e^k,ux):
> od:
> ux:=algsubs(e^2=X*Xb,expand(ux)):
> ux:=sort(ux,[X,Xb]) :
> end:
```

```
> ENXXB(TRONC(‘(r/a)*cos(v)‘,4,e),4);
```

$$\frac{125}{384}\cos(M)\,(16\cos(M)^4+5-20\cos(M)^2)\,X^2\,Xb^2$$
$$+\,(-\frac{45}{256}X^3+\frac{5}{384}X^2\,Xb+\frac{5}{384}X\,Xb^2-\frac{45}{256}Xb^3-\frac{1}{6}X^2-\frac{1}{6}Xb^2-\frac{3}{16}X-\frac{3}{16}Xb-\frac{3}{2})\,e$$
$$+\,\frac{\frac{1}{6}X^4+\frac{1}{6}Xb^4+\frac{3}{16}X^3+\frac{3}{16}Xb^3+\frac{1}{4}X^2+\frac{1}{4}Xb^2+\frac{1}{2}X+\frac{1}{2}Xb}{e}$$

2.2.5. Développements en longitude moyenne et définition des "inégalités"

On peut introduire la longitude moyenne L à la place de M en utilisant les variables x et xb (conjugué de x) :

$$x = e\ \exp(I\,\varpi) = e\ \exp(I(L-M)) = Xb\ \exp(I\,L)$$

Exemple sur la série représentant $\theta = \exp\left(I\,(v - M)\right)$:

```
> T:=collect(combine(
> subs(X=xb*exp(I*L),Xb=x*exp(-I*L),THETA),exp),exp):
> collect(TRONC(T,5,[x,xb]),exp);
```

$$\left(-\frac{5}{4}\,xb^2\,x + \frac{17}{48}\,xb^3\,x^2 + xb\right)e^{(I\,L)} - e^{(-I\,L)}\,x + \left(-\frac{1}{12}\,x^3 + \frac{1}{48}\,xb\,x^4\right)e^{(-3\,I\,L)} + \left(\frac{1}{48}\,xb\,x^3 - \frac{1}{8}\,x^2\right)e^{(-2\,I\,L)} + \frac{81}{40}\,e^{(5\,I\,L)}\,xb^5$$
$$+ \frac{625}{384}\,e^{(4\,I\,L)}\,xb^4 + \left(\frac{4}{3}\,xb^3 - \frac{7}{3}\,xb^4\,x\right)e^{(3\,I\,L)} + \left(-\frac{27}{16}\,xb^3\,x + \frac{9}{8}\,xb^2\right)e^{(2\,I\,L)} - \frac{1}{15}\,e^{(-5\,I\,L)}\,x^5 + 1 - x\,xb + \frac{7}{64}\,x^2\,xb^2$$
$$- \frac{9}{128}\,e^{(-4\,I\,L)}\,x^4$$

En mécanique céleste clasique, les combinaisons linéaires entières des longitudes moyennes sont appelées **inégalités**. Les multiples de L sont donc un exemple d'**inégalité**; par abus de langage, dans un développement de Fourier en longitudes moyennes, le coefficient d'une inégalité telle que $p\,L$ est le facteur de $\exp\left(I\,p\,L\right)$ dans ce développement, c'est-à-dire une série (généralement tronquée au degré d) de la forme:

$$\sum_{\{n_1,\,n_2\}\in \mathrm{N}(d)} C_{n_1,\,n_2}\, x^{n_1}\, xb^{n_2} \qquad \text{en facteur de} \qquad \exp\left(I\,p\,L\right)$$

où $\mathrm{N}(d)$ est l'ensemble des couples d'entiers positifs ou nuls de somme inférieure ou égale à d.

La propriété de d'Alembert de rang 0 se traduit alors par la relation:

$$n_2 - n_1 = p$$

pour chacun des termes de l'inégalité $p\,L$. Cela traduit aussi la propriété d'invariance de ces développements vis-à-vis des rotations du repère dans lequel sont définies les longitudes; la somme précédente est en effet équivalente à :

$$\sum_{\{n_1,\, n_2\}\in \mathrm{N}(d)} C_{n_1,\, n_2}\, e^{(n_1+n_2)} \exp(I\,p\,L + I(n_1 - n_2)\varpi)$$

qui conserve cette valeur quand, quelque soit f , on change L en $L+f$ et ϖ en $\varpi + f$ à condition que $p + n_1 - n_2$ soit nul.

Exemple de sélection d'inégalité : Pour isoler une inégalité, on peut utiliser la fonction *select* ou la fonction *coeff* (mais alors on n'obtient que le coefficient de l'inégalité):

```
> select(has,T,exp(-2*I*L)); coeff(%,exp(-2*I*L));
```

$$\left(\frac{226634433169}{59929893273600}\,xb^7\,x^9 + \frac{125923}{17694720}\,xb^4\,x^6 + \frac{857}{92160}\,xb^3\,x^5 + \frac{37}{3072}\,xb^2\,x^4 + \frac{2165563433}{475634073600}\,xb^6\,x^8 + \frac{20926946577002801}{7594316075630592000}\,xb^9\,x^{11} + \frac{1}{48}\,xb\,x^3 - \frac{1}{8}\,x^2 + \frac{122830157549671}{38355131695104000}\,xb^8\,x^{10} + \frac{6951913}{1238630400}\,xb^5\,x^7\right)e^{(-2\,I\,L)}$$

$$\frac{226634433169}{59929893273600}\,xb^7\,x^9 + \frac{125923}{17694720}\,xb^4\,x^6 + \frac{857}{92160}\,xb^3\,x^5 + \frac{37}{3072}\,xb^2\,x^4 + \frac{2165563433}{475634073600}\,xb^6\,x^8 + \frac{20926946577002801}{7594316075630592000}\,xb^9\,x^{11} + \frac{1}{48}\,xb\,x^3 - \frac{1}{8}\,x^2 + \frac{122830157549671}{38355131695104000}\,xb^8\,x^{10} + \frac{6951913}{1238630400}\,xb^5\,x^7$$

Une inégalité est dite *séculaire* si elle ne dépend pas des longitudes moyennes (donc correspondant ici à $p = 0$). Pour isoler l'inégalité séculaire avec Maple, on peut utiliser la fonction *remove* en lui demandant de supprimer les termes contenant une longitude:

```
> remove(has,T,L);
```

$$-\frac{38048993776368689}{138078474102374400000}\,xb^{10}\,x^{10} - \frac{112803205913}{35224100536320}\,xb^9\,x^9 - \frac{29607313}{6502809600}\,xb^7\,x^7 - \frac{805811317 87}{21308406497280}\,xb^8\,x^8 - \frac{11917541}{2123366400}\,xb^6\,x^6 - \frac{5}{288}\,xb^3\,x^3 + \frac{7}{64}\,x^2\,xb^2 - \frac{1355}{147456}\,xb^4\,x^4 - \frac{26239}{3686400}\,xb^5\,x^5 - x\,xb + 1$$

2.2.6. Développement des coefficients de Hansen (degrés modérés)

On suppose ici que le calcul de (r/a), (a/r) et θ ont déjà été réalisés au degré 20 et enregistrés. Cependant, ici, il faut redéfinir degmax avec une valeur plus faible (degré modéré).

Dans ce paragraphe, les coefficients de Hansen sont calculés directement en variables X et Xb au degré 10.

```
> read cat(chemin,‘kepler20.m‘):
```

```
> degmax:=10:
> RSAT:=TRONC(RSA,degmax,[X,Xb]):
> THETAT:=TRONC(THETA,degmax,[X,Xb]):
```

Pour calculer les coefficients de Hansen $X_{n,\,m,\,k}(e)$, on part de leur définition de coefficient de $\exp(I\,k\,M)$ dans le développement en série de Fourier complexe de $(r/a)^n \exp(I\,m\,v)$, c'est-à-dire qu'on a aussi : C(3.146)

$$(\frac{r}{a})^n \,\exp\left(I\,m\,(v-M)\right) = \sum_{k=-\infty}^{\infty} X_{n,\,m,\,k}(e)\,\exp\left(I\,(k-m)\,M\right) = \sum_{n_2=0}^{\mathrm{D}} \left(\sum_{n_1=0}^{\mathrm{D}-n_2} X_{n,\,m,\,n_1,\,n_2}\,X^{n_1}\,Xb^{n_2}\right)$$

où D représente le degré maximum *degmax* des polynômes en X, Xb.

On utilise pour cela les développements limités de (r/a) et de θ construits précédemment en fonction de X, Xb; la procédure *mtaylor* permet alors de construire les développements des coefficients de Hansen, limités au même degré, **en laissant** m **et** n **indéterminés**.

```
> deb:=time():
> Xnm:=sort(map(factor,mtaylor(RSAT^n*THETAT^m,[X,Xb],degmax+1)),[X,Xb]
> ): nops(%);
> print("durée=",time()-deb);
```

66

$$\text{“durée=”},\ 1.309$$

```
> PRINTRONC(Xnm,2,[X,Xb]);
```

$$\begin{aligned}&(-\frac{1}{725760}\,n\,m^9-\frac{1}{27525120}\,n^9+\frac{1}{64512}\,m^9+\frac{1}{3715891200}\,n^{10}-\frac{5719087}{5806080}\,n+\frac{7281587}{5806080}\,m-\frac{1023307}{307200}\,n\,m\\
&-\frac{6698509}{44236800}\,n\,m^5+\frac{222643}{3538944}\,n^4\,m^2-\frac{1353293}{8847360}\,n^3\,m^3+\frac{7384741}{35389440}\,n^2\,m^4-\frac{152263}{11059200}\,m\,n^5+\frac{2019649}{44236800}\,m^6\\
&+\frac{221653}{176947200}\,n^6-\frac{2870431}{3932160}\,n\,m^4-\frac{8327357}{17694720}\,n^3\,m^2+\frac{1567087}{11796480}\,m\,n^4-\frac{5016857}{6635520}\,m\,n^3+\frac{9043949}{35389440}\,m^5\\
&-\frac{529007}{35389440}\,n^5+\frac{2939239}{3538944}\,n^2\,m^3+\frac{10249691}{92897280}\,n^4-\frac{29733251}{61931520}\,n^3+\frac{170568581}{185794560}\,m^4+\frac{125549713}{61931520}\,m^3\\
&+\frac{39891203}{20643840}\,n^2\,m^2+\frac{85111457}{77414400}\,n^2+\frac{191511907}{77414400}\,m^2+\frac{29115271}{12386304}\,n^2\,m-\frac{202708397}{92897280}\,n\,m^3-\frac{235392817}{61931520}\,n\,m^2\\
&+\frac{1}{3628800}\,m^{10}-\frac{1}{3870720}\,n^7\,m^3-\frac{341}{215040}\,n\,m^7-\frac{659}{20643840}\,m\,n^7-\frac{419}{138240}\,n^3\,m^5+\frac{15251}{17694720}\,n^6\,m\\
&+\frac{3799}{17694720}\,n^6\,m^2-\frac{3647}{4423680}\,n^5\,m^3-\frac{3181}{655360}\,n^5\,m^2-\frac{6277}{221184}\,n^3\,m^4+\frac{583}{294912}\,n^4\,m^4+\frac{53663}{3538944}\,n^4\,m^3\\
&+\frac{3209}{1105920}\,n^2\,m^6+\frac{7819}{245760}\,n^2\,m^5-\frac{21877}{1105920}\,n\,m^6-\frac{1}{5760}\,n^3\,m^6-\frac{5}{73728}\,n^5\,m^4+\frac{17}{737280}\,n^6\,m^3-\frac{13}{2580480}\,n^7\,m^2\\
&+\frac{53}{82575360}\,m\,n^8-\frac{23}{322560}\,n\,m^8+\frac{49}{368640}\,n^4\,m^5+\frac{47}{322560}\,n^2\,m^7-\frac{1}{460800}\,n^5\,m^5+\frac{1}{322560}\,n^2\,m^8\\
&-\frac{1}{185794560}\,m\,n^9-\frac{1}{241920}\,n^3\,m^7+\frac{1}{20643840}\,n^8\,m^2+\frac{1}{1105920}\,n^6\,m^4+\frac{1}{276480}\,n^4\,m^6+\frac{5819}{1105920}\,m^7\\
&+\frac{257}{123863040}\,n^8+\frac{733}{1935360}\,m^8-\frac{1159}{17694720}\,n^7)X^{10},\ +\ldots+,(\frac{5}{8}\,m+\frac{1}{2}\,m^2-\frac{3}{8}\,n-\frac{1}{2}\,n\,m+\frac{1}{8}\,n^2)\,X^2\\
&+(\frac{1}{4}\,n-m^2+\frac{1}{4}\,n^2)\,X\,Xb+(-\frac{5}{8}\,m+\frac{1}{2}\,n\,m+\frac{1}{2}\,m^2-\frac{3}{8}\,n+\frac{1}{8}\,n^2)\,Xb^2+(m-\frac{1}{2}\,n)\,X+(-m-\frac{1}{2}\,n)\,Xb+1\end{aligned}$$

```
> save `Xnm`,cat(chemin,`Xnm`.degmax.`.m`);
```

Le coefficient de $\exp(I(q\,M))$ dans le développement de $(r/a)^n \exp(Im(v-M))$ est alors $X_{n,\,m,\,q+m}(e)$; on constate bien que ce coefficient est au moins de degré $|q|$ en excentricité.

Pour cela, on refait apparaître les puissances de $\exp(I\,M)$ dans le développement de *Xnm* :

Pour avoir les coefficients de Hansen sous la forme classique en fonction de l'excentricité, il suffit d'appliquer cette fonction de conversion :

```
> ENeEXPM:=proc(u);
> subs(X=e*exp(I*M),Xb=e/exp(I*M),u):  collect(combine(%,exp),exp):
> end:
```

```
> `Sum(X(n,m,q+m)exp(IqM)`:=ENeEXPM(Xnm):  nops(%);
```

$$26$$

```
> collect(TRONC(`Sum(X(n,m,q+m)exp(IqM)`,3,e),exp);
```

$$\begin{aligned}
&((m-\frac{1}{2}\,n)\,e+(-\frac{1}{8}\,m-\frac{1}{2}\,m^3-\frac{5}{8}\,m^2+\frac{1}{16}\,n^2+\frac{3}{16}\,n-\frac{1}{16}\,n^3+\frac{5}{16}\,n\,m+\frac{1}{8}\,n^2\,m+\frac{1}{4}\,n\,m^2)\,e^3)\,e^{(I\,M)}\\
&+(\frac{5}{8}\,m^2-\frac{13}{24}\,m+\frac{11}{16}\,n\,m-\frac{1}{8}\,n^2\,m-\frac{17}{48}\,n-\frac{1}{6}\,m^3-\frac{1}{48}\,n^3+\frac{3}{16}\,n^2-\frac{1}{4}\,n\,m^2)\,e^3\,e^{(-3\,I\,M)}\\
&+(-\frac{17}{48}\,n+\frac{13}{24}\,m-\frac{1}{48}\,n^3+\frac{3}{16}\,n^2+\frac{1}{6}\,m^3+\frac{5}{8}\,m^2-\frac{11}{16}\,n\,m-\frac{1}{4}\,n\,m^2+\frac{1}{8}\,n^2\,m)\,e^3\,e^{(3\,I\,M)}\\
&+((\frac{1}{2}\,m^3-\frac{5}{8}\,m^2+\frac{1}{8}\,m+\frac{3}{16}\,n+\frac{1}{16}\,n^2-\frac{5}{16}\,n\,m-\frac{1}{8}\,n^2\,m-\frac{1}{16}\,n^3+\frac{1}{4}\,n\,m^2)\,e^3+(-m-\frac{1}{2}\,n)\,e)\,e^{(-I\,M)}\\
&+(-\frac{5}{8}\,m+\frac{1}{2}\,n\,m+\frac{1}{2}\,m^2-\frac{3}{8}\,n+\frac{1}{8}\,n^2)\,e^2\,e^{(-2\,I\,M)}+(\frac{5}{8}\,m+\frac{1}{2}\,m^2-\frac{3}{8}\,n-\frac{1}{2}\,n\,m+\frac{1}{8}\,n^2)\,e^2\,e^{(2\,I\,M)}+1\\
&+(\frac{1}{4}\,n-m^2+\frac{1}{4}\,n^2)\,e^2
\end{aligned}$$

La procédure suivante : *Hansen(n,m,k)* isole le coefficient de $\exp(I((k-m)M))$:

```
> Hansen:=proc(nn,mm,k) local u,q;
> u:=`Sum(X(n,m,q+m)exp(IqM)`;q:=k-mm:
> if q<>0 then sort(coeff(u,exp(I*q*M)),e) else sort(remove(has,u,M),e)
> fi; subs(n=nn,m=mm,%);
> end:
```

Exemples de calculs (attention, il faut que $k-m$ soit une valeur numérique entière) :

```
> TRONC(Hansen(p,q,q+2),5,e);
```

$$(\frac{11}{48}p-\frac{11}{48}q+\frac{47}{96}pq-\frac{2}{3}q^2-\frac{1}{96}p^2+\frac{1}{2}pq^2+\frac{1}{6}pq^3+\frac{1}{32}p^2q-\frac{1}{24}qp^3-\frac{5}{8}q^3-\frac{1}{6}q^4-\frac{1}{16}p^3+\frac{1}{96}p^4)e^4$$
$$+(\frac{5}{8}q+\frac{1}{2}q^2-\frac{3}{8}p-\frac{1}{2}pq+\frac{1}{8}p^2)e^2$$

```
> Hansen(p,3,3);
```

$$\frac{1}{14745600}(393201p^2-35p^9+202310p^3+p^{10}+154790p^4-2635155p+330p^8+330p^7-13317p^6+4035p^5$$
$$-10529244)e^{10}$$
$$+\frac{1}{147456}(171756p+590949-48366p^2-26384p^3+43p^4+1168p^5+10p^6-20p^7+p^8)e^8$$
$$+\frac{1}{2304}(-3513p-32328+3025p^2+417p^3-83p^4-9p^5+p^6)e^6+\frac{1}{64}(2p+1215-73p^2-2p^3+p^4)e^4$$
$$+\frac{1}{4}(p-36+p^2)e^2+1$$

```
> Hansen(5,6,1);
```

$$-\frac{12581857}{5160960}e^9+\frac{2996329}{92160}e^7-\frac{169021}{3840}e^5$$

```
> Hansen(-3,m,m);
```

$$\frac{1}{14745600}\,(-1024\,m^{10}+39916800-26673336\,m^2+6582510\,m^4-760516\,m^6+42880\,m^8)\,e^{10}$$
$$+\frac{1}{147456}\,(362880-243396\,m^2+58441\,m^4-6112\,m^6+256\,m^8)\,e^8$$
$$+\frac{1}{2304}\,(5040-3394\,m^2+772\,m^4-64\,m^6)\,e^6+\frac{1}{64}\,(120-81\,m^2+16\,m^4)\,e^4+\frac{1}{4}\,(6-4\,m^2)\,e^2+1$$

```
> TRONC(Hansen(n,m,3+m),5,e);
```

$$(\frac{77}{256}\,n-\frac{43}{128}\,m+\frac{143}{192}\,n\,m-\frac{27}{32}\,m^2-\frac{17}{192}\,n^2+\frac{617}{768}\,n\,m^2+\frac{37}{96}\,n\,m^3-\frac{19}{192}\,n^2\,m-\frac{5}{64}\,n^2\,m^2-\frac{1}{48}\,n^2\,m^3-\frac{7}{128}\,m\,n^3$$
$$+\frac{1}{128}\,m\,n^4-\frac{1}{96}\,n^3\,m^2-\frac{307}{384}\,m^3-\frac{5}{16}\,m^4-\frac{41}{768}\,n^3+\frac{7}{384}\,n^4-\frac{1}{768}\,n^5-\frac{1}{24}\,m^5+\frac{1}{16}\,n\,m^4)e^5$$
$$+(-\frac{17}{48}\,n+\frac{13}{24}\,m-\frac{1}{48}\,n^3+\frac{3}{16}\,n^2+\frac{1}{6}\,m^3+\frac{5}{8}\,m^2-\frac{11}{16}\,n\,m-\frac{1}{4}\,n\,m^2+\frac{1}{8}\,n^2\,m)\,e^3$$

2.2.7. Développement des coefficients de Hansen (degrés élevés)

Au delà du degré 12 et jusqu'aux environs du degré 20, le calcul de $(\frac{r}{a})^n \exp(m\,I\,(v-M))$ est à réaliser autrement. On évite notamment au maximum la manipulation des $\sin(kM)$ et $\cos(kM)$, introduisant à leur place la variable $Z = \exp(I\,M)$ et son conjugué $1/Z$. Ainsi $2\cos(nM)$ est remplacé par $Z^n + 1/Z^n$, et l'intégration sur M revient à substituer $-I\,Z^k/k$ à Z^k et $I/k\,Z^k$ à $1/Z^k$. Dans ce paragraphe, les coefficients de Hansen sont calculés directement en fonction de l'excentricité e et de la variable Z.

```
> degmax:=20;
```

$$degmax := 20$$

On lit ici les développements de (r/a) et de θ obtenus précédemment à des fins de comparaisons avec ceux qu'on va construire par la méthode proposée dans ce paragraphe.

```
> read cat(chemin,‘kepler20.m‘):
```

Développement de (a/r) en fonction de (e, Z) et comparaison avec les résultats précédents (en (e, M)):

```
> ‘aSr‘:= collect(
> mtaylor(1+sum(BesselJ(n,n*e)*(Z^n+1/Z^n),n=1..degmax),[e],degmax+1),e):
```

```
> TRONC(‘aSr‘,4,e); TRONC(‘a/r‘,4,e);
```

$$\left(\frac{2}{3}Z^4 + \frac{2}{3}\frac{1}{Z^4} - \frac{1}{6}Z^2 - \frac{1}{6}\frac{1}{Z^2}\right)e^4 + \left(-\frac{1}{16}Z - \frac{1}{16}\frac{1}{Z} + \frac{9}{16}Z^3 + \frac{9}{16}\frac{1}{Z^3}\right)e^3 + \left(\frac{1}{2}Z^2 + \frac{1}{2}\frac{1}{Z^2}\right)e^2 + \left(\frac{1}{2}Z + \frac{1}{2}\frac{1}{Z}\right)e + 1$$

$$\left(-\frac{1}{6}e^{(-2\,I\,M)} + \frac{2}{3}e^{(4\,I\,M)} - \frac{1}{6}e^{(2\,I\,M)} + \frac{2}{3}e^{(-4\,I\,M)}\right)e^4 + \left(\frac{9}{16}e^{(-3\,I\,M)} + \frac{9}{16}e^{(3\,I\,M)} - \frac{1}{16}e^{(-I\,M)} - \frac{1}{16}e^{(I\,M)}\right)e^3$$
$$+ \left(\frac{1}{2}e^{(-2\,I\,M)} + \frac{1}{2}e^{(2\,I\,M)}\right)e^2 + \left(\frac{1}{2}e^{(-I\,M)} + \frac{1}{2}e^{(I\,M)}\right)e + 1$$

Développement de (r/a) en fonction de (e, Z) et comparaison avec celui en (e, M):

```
> 'rSa':=collect(expand(mtaylor(1/'aSr',e,degmax+1)),e):
> TRONC('rSa',4,e); TRONC('r/a',4,e);
```

$$\left(\frac{1}{6}Z^2-\frac{1}{6}Z^4+\frac{1}{6}\frac{1}{Z^2}-\frac{1}{6}\frac{1}{Z^4}\right)e^4+\left(\frac{3}{16}Z+\frac{3}{16}\frac{1}{Z}-\frac{3}{16}Z^3-\frac{3}{16}\frac{1}{Z^3}\right)e^3+\left(-\frac{1}{4}Z^2-\frac{1}{4}\frac{1}{Z^2}+\frac{1}{2}\right)e^2+\left(-\frac{1}{2}Z-\frac{1}{2}\frac{1}{Z}\right)e+1$$

$$\left(\frac{1}{6}e^{(2IM)}-\frac{1}{6}e^{(4IM)}+\frac{1}{6}e^{(-2IM)}-\frac{1}{6}e^{(-4IM)}\right)e^4+\left(\frac{3}{16}e^{(-IM)}-\frac{3}{16}e^{(-3IM)}+\frac{3}{16}e^{(IM)}-\frac{3}{16}e^{(3IM)}\right)e^3$$
$$+\left(\frac{1}{2}-\frac{1}{4}e^{(-2IM)}-\frac{1}{4}e^{(2IM)}\right)e^2+\left(-\frac{1}{2}e^{(IM)}-\frac{1}{2}e^{(-IM)}\right)e+1$$

Les calculs utilisant PRODPOL sont comparables à ceux réalisés avec *mtaylor* :

```
> p:=PRODPOL('aSr','aSr',degmax,e):
> sqe2:=mtaylor(sqrt(1-e*e),[e],degmax+1):
> p:=PRODPOL(p,sqe2,degmax,e)-1:  nops(%);
```

$$20$$

```
> s:=collect(expand(mtaylor('aSr'^2*sqrt(1-e*e),[e],degmax+1)-1),e):
> nops(%);
```

$$20$$

```
> simplify(p-s);
```

$$0$$

On fait maintenant "l'intégration par rapport à M " de cette série en (e, Z) pour obtenir directement $I(v\text{-}M)$, puis θ :

```
> s:=collect(p,Z):
> 'I(v-M)':=0:
```

```
> for k from 1 to nops(s) do:  d:=degree(op(k,s),Z):
> #print(d,op(k,s),I/d*op(k,s));
> ‘I(v-M)‘:=‘I(v-M)‘ + 1/d*op(k,s):
> od:
> ‘I(v-M)‘:=collect(‘I(v-M)‘,e):  nops(%);
```

$$20$$

```
> T:=collect(expand(mtaylor(exp(‘I(v-M)‘),e,degmax+1)),e):
> PRINTRONC(T,4,e);
```

$$
\begin{aligned}
&(\frac{475110161534664614 27}{86068915523813376000}\,\frac{1}{Z^{14}}+\frac{128840905889375}{131634811977596928}\,\frac{1}{Z^{6}}+\frac{1201605721749}{841813590016000}\,\frac{1}{Z^{4}}-\frac{1652211122481}{1157493686272000}\,Z^{2}\\
&+\frac{20926946577002801}{759431607563059 2000}\,\frac{1}{Z^{2}}+\frac{250520623630191}{376185448038 40000}\,\frac{1}{Z^{10}}-\frac{153397171881875}{162012076280119296}\,Z^{4}\\
&-\frac{2421451098 6328125}{16371590698631168}\,\frac{1}{Z^{16}}-\frac{38048993776368689}{13807847410237440000}+\frac{32500810053757901458 4599}{789808871865581568000}\,Z^{16}\\
&+\frac{666658328696636098 1897}{3751592141361512448000}\,\frac{1}{Z^{18}}+\frac{817842490296083}{1343212367118336000}\,\frac{1}{Z^{8}}-\frac{438970879898905 6477}{478672043554897920 00}\,\frac{1}{Z^{12}}\\
&+\frac{2894867431563851686969}{405030190700298240 00}\,Z^{12}-\frac{3868391217872531}{335803091779584000}\,Z^{6}-\frac{987316771728515625}{4092897674657792}\,Z^{14}\\
&-\frac{10412735029791124153 2841}{305153427766247424000}\,Z^{18}-\frac{759367184767227750 23}{747925068054528 0000}\,Z^{10}+\frac{179591813518438 89}{300948358430720 00}\,Z^{8}\\
&-\frac{104127350297911241532841}{1342675082171488665600 00}\,\frac{1}{Z^{20}}+\frac{8654057508871269270 09}{793520977739776 0000}\,Z^{20})e^{20},\ +\ldots+,\\
&(-\frac{9}{128}\,\frac{1}{Z^{4}}+\frac{625}{384}\,Z^{4}+\frac{7}{64}+\frac{1}{48}\,\frac{1}{Z^{2}}-\frac{27}{16}\,Z^{2})\,e^{4}+(\frac{4}{3}\,Z^{3}-\frac{1}{12}\,\frac{1}{Z^{3}}-\frac{5}{4}\,Z)\,e^{3}+(\frac{9}{8}\,Z^{2}-1-\frac{1}{8}\,\frac{1}{Z^{2}})\,e^{2}+(-\frac{1}{Z}+Z)\,e+1
\end{aligned}
$$

Vérification, en comparant avec le polynôme Theta calculé précédemment:

```
> TRONC(Theta,4,e);
```

$$(\frac{7}{64} - \frac{27}{16}\,e^{(2\,I\,M)} + \frac{1}{48}\,e^{(-2\,I\,M)} + \frac{625}{384}\,e^{(4\,I\,M)} - \frac{9}{128}\,e^{(-4\,I\,M)})\,e^4 + (\frac{4}{3}\,e^{(3\,I\,M)} - \frac{1}{12}\,e^{(-3\,I\,M)} - \frac{5}{4}\,e^{(I\,M)})\,e^3$$
$$+ (-1 + \frac{9}{8}\,e^{(2\,I\,M)} - \frac{1}{8}\,e^{(-2\,I\,M)})\,e^2 + (-e^{(-I\,M)} + e^{(I\,M)})\,e + 1$$

Pour obtenir le développement de $(\frac{r}{a})^n\ \exp(m\,I\,(v-M))$, on calcule séparément $(\frac{r}{a})^n$ et $\exp(m\,I\,(v-M))$ par la formule du binome pour n et m arbitraires (voir procédure POWERn) puis on en fait le produit par PRODPOL

```
> deb:=time():  RSAn:=POWERn(rSa,n,degmax,e):
> print("durée=",time()-deb):  TRONC(%,4,e);
```

“durée=”, 1.900

$$(-\frac{1}{64}\,n^2 - \frac{1}{16}\,n^3\,Z^2 - \frac{1}{32}\,n^3 + \frac{95}{384}\,n^2\,Z^4 - \frac{1}{96}\,n^2\,Z^2 - \frac{71}{192}\,n\,Z^4 - \frac{1}{16}\,\frac{n^3}{Z^2} + \frac{1}{96}\,n^4\,Z^2 + \frac{1}{384}\,\frac{n^4}{Z^4} - \frac{3}{64}\,\frac{n^3}{Z^4} + \frac{95}{384}\,\frac{n^2}{Z^4} - \frac{1}{96}\,\frac{n^2}{Z^2}$$
$$+ \frac{11}{48}\,n\,Z^2 + \frac{1}{96}\,\frac{n^4}{Z^2} - \frac{71}{192}\,\frac{n}{Z^4} + \frac{1}{32}\,n + \frac{1}{384}\,n^4\,Z^4 - \frac{3}{64}\,n^3\,Z^4 + \frac{11}{48}\,\frac{n}{Z^2} + \frac{1}{64}\,n^4)e^4 +$$
$$(\frac{1}{16}\,\frac{n^2}{Z} + \frac{1}{16}\,n^2\,Z + \frac{3}{16}\,n^2\,Z^3 + \frac{3}{16}\,\frac{n^2}{Z^3} - \frac{17}{48}\,\frac{n}{Z^3} - \frac{1}{48}\,\frac{n^3}{Z^3} - \frac{17}{48}\,n\,Z^3 - \frac{1}{16}\,n^3\,Z + \frac{3}{16}\,\frac{n}{Z} - \frac{1}{48}\,n^3\,Z^3 + \frac{3}{16}\,n\,Z - \frac{1}{16}\,\frac{n^3}{Z})\,e^3$$
$$+ (-\frac{3}{8}\,n\,Z^2 - \frac{3}{8}\,\frac{n}{Z^2} + \frac{1}{4}\,n^2 + \frac{1}{8}\,\frac{n^2}{Z^2} + \frac{1}{4}\,n + \frac{1}{8}\,n^2\,Z^2)\,e^2 + (-\frac{1}{2}\,\frac{n}{Z} - \frac{1}{2}\,n\,Z)\,e + 1$$

```
> deb:=time():  Tm:=POWERn(T,m,degmax,e):  print("durée=",time()-deb):
> TRONC(%,4,e);
```

“durée=”, 2.540

$$(-\frac{1}{6}\frac{m^4}{Z^2}+\frac{103}{192}m\,Z^4+\frac{283}{384}\frac{m^2}{Z^4}-\frac{103}{192}\frac{m}{Z^4}-\frac{2}{3}m^2\,Z^2-\frac{1}{6}m^4\,Z^2-\frac{5}{16}\frac{m^3}{Z^4}+\frac{1}{24}m^4\,Z^4+\frac{1}{4}m^4+\frac{5}{8}\frac{m^3}{Z^2}+\frac{5}{16}m^3\,Z^4-\frac{11}{48}m\,Z^2$$
$$+\frac{11}{48}\frac{m}{Z^2}-\frac{5}{8}m^3\,Z^2-\frac{9}{64}m^2+\frac{1}{24}\frac{m^4}{Z^4}+\frac{283}{384}m^2\,Z^4-\frac{2}{3}\frac{m^2}{Z^2})e^4$$
$$+(\frac{1}{8}\frac{m}{Z}+\frac{1}{6}m^3\,Z^3+\frac{5}{8}m^2\,Z^3-\frac{5}{8}\frac{m^2}{Z}+\frac{5}{8}\frac{m^2}{Z^3}-\frac{13}{24}\frac{m}{Z^3}+\frac{1}{2}\frac{m^3}{Z}-\frac{1}{2}m^3\,Z-\frac{5}{8}m^2\,Z-\frac{1}{6}\frac{m^3}{Z^3}-\frac{1}{8}m\,Z+\frac{13}{24}m\,Z^3)\,e^3$$
$$+(-\frac{5}{8}\frac{m}{Z^2}-m^2+\frac{5}{8}m\,Z^2+\frac{1}{2}m^2\,Z^2+\frac{1}{2}\frac{m^2}{Z^2})\,e^2+(m\,Z-\frac{m}{Z})\,e+1$$

```
> deb:=time(): Hans:=PRODPOL(RSAn,Tm,degmax,e):
> print("durée=",time()-deb): TRONC(%,3,e);
```

"durée=", 28.640

$$(\frac{11}{16}\frac{n\,m}{Z^3}+\frac{3}{16}n\,Z+\frac{3}{16}\frac{n}{Z}+\frac{1}{2}\frac{m^3}{Z}-\frac{17}{48}n\,Z^3-\frac{17}{48}\frac{n}{Z^3}-\frac{1}{48}n^3\,Z^3-\frac{1}{48}\frac{n^3}{Z^3}+\frac{1}{4}\frac{n\,m^2}{Z}-\frac{1}{4}n\,Z^3\,m^2+\frac{1}{6}m^3\,Z^3+\frac{3}{16}n^2\,Z^3$$
$$+\frac{3}{16}\frac{n^2}{Z^3}+\frac{1}{16}\frac{n^2}{Z}+\frac{1}{16}n^2\,Z-\frac{1}{2}m^3\,Z-\frac{1}{6}\frac{m^3}{Z^3}-\frac{1}{16}\frac{n^3}{Z}-\frac{1}{16}n^3\,Z-\frac{5}{8}\frac{m^2}{Z}+\frac{5}{8}\frac{m^2}{Z^3}+\frac{5}{8}m^2\,Z^3-\frac{5}{8}m^2\,Z-\frac{1}{8}m\,Z$$
$$-\frac{13}{24}\frac{m}{Z^3}+\frac{13}{24}m\,Z^3+\frac{1}{8}\frac{m}{Z}+\frac{1}{8}n^2\,Z\,m-\frac{1}{4}\frac{n\,m^2}{Z^3}+\frac{1}{8}n^2\,Z^3\,m+\frac{1}{4}n\,Z\,m^2-\frac{1}{8}\frac{n^2\,m}{Z}-\frac{1}{8}\frac{n^2\,m}{Z^3}+\frac{5}{16}n\,m\,Z-\frac{5}{16}\frac{n\,m}{Z}$$
$$-\frac{11}{16}n\,Z^3\,m)e^3$$
$$+(-m^2+\frac{1}{2}\frac{m^2}{Z^2}-\frac{3}{8}\frac{n}{Z^2}-\frac{1}{2}n\,Z^2\,m+\frac{1}{8}\frac{n^2}{Z^2}+\frac{1}{8}n^2\,Z^2+\frac{1}{2}m^2\,Z^2+\frac{1}{4}n^2+\frac{1}{4}n+\frac{5}{8}m\,Z^2-\frac{5}{8}\frac{m}{Z^2}-\frac{3}{8}n\,Z^2+\frac{1}{2}\frac{n\,m}{Z^2})\,e^2$$
$$+(-\frac{m}{Z}-\frac{1}{2}\frac{n}{Z}+m\,Z-\frac{1}{2}n\,Z)\,e+1$$

```
> nops(Hans); save Hans, cat(chemin,`Hansen`.degmax.`.m`):
```

21

On peut aussi exprimer $(\frac{r}{a})^n \exp(m\,I\,(v-M))$ en série de Laurent de $Z=\exp(I\,M)$. D'après ce qu'on a vu précédemment, le facteur de $Z^{(k-m)}$ dans cette série est le coefficient de Hansen $X_{n,\,m,\,k}(e)$

```
> deb:=time():  HansZ:=collect(sort(map(factor,Hans),e),Z):
> nops(HansZ); print("durée=",time()-deb):
```

$$51$$
$$\text{“durée=”},\ 7.539$$

Exemples de coefficient d'une puissance de Z (ici puissances 3 et 0):

```
> TRONC(coeff(HansZ,Z,3),5,e);
```

$$(\frac{77}{256}\,n-\frac{43}{128}\,m+\frac{143}{192}\,n\,m-\frac{27}{32}\,m^2-\frac{17}{192}\,n^2+\frac{617}{768}\,n\,m^2+\frac{37}{96}\,n\,m^3-\frac{19}{192}\,n^2\,m-\frac{5}{64}\,n^2\,m^2-\frac{1}{48}\,n^2\,m^3-\frac{7}{128}\,m\,n^3$$
$$+\frac{1}{128}\,m\,n^4-\frac{1}{96}\,n^3\,m^2-\frac{307}{384}\,m^3-\frac{5}{16}\,m^4-\frac{41}{768}\,n^3+\frac{7}{384}\,n^4-\frac{1}{768}\,n^5-\frac{1}{24}\,m^5+\frac{1}{16}\,n\,m^4)e^5$$
$$+(-\frac{17}{48}\,n+\frac{13}{24}\,m-\frac{1}{48}\,n^3+\frac{3}{16}\,n^2+\frac{1}{6}\,m^3+\frac{5}{8}\,m^2-\frac{11}{16}\,n\,m-\frac{1}{4}\,n\,m^2+\frac{1}{8}\,n^2\,m)\,e^3$$

```
> TRONC(coeff(HansZ,Z,0),5,e);
```

$$(-\frac{1}{32}\,n^3-\frac{9}{64}\,m^2-\frac{1}{8}\,n^2\,m^2+\frac{1}{32}\,n-\frac{1}{64}\,n^2+\frac{1}{64}\,n^4+\frac{1}{4}\,m^4)\,e^4+(\frac{1}{4}\,n^2-m^2+\frac{1}{4}\,n)\,e^2+1$$

Sauvegarde

```
> save HansZ, cat(chemin,‘HansenZ‘.degmax.‘.m‘):
```

On peut enfin convertir ce développement en variables X, Xb, pour le rendre comparable au polynôme *Xnm* construit au degré 10 au paragraphe précédent; Pour limiter l'usage de la mémoire, il faut fractionner les calculs, en partant de l'expression *HansZ* où on a factorisé chaque puissance de Z. On peut ainsi calculer sans problème un développement de degré 20 en variables X, Xb à coefficients dépendant explicitement de n et de m:

```
> read cat(chemin,‘HansenZ‘.degmax.‘.m‘): read
> cat(chemin,‘Xnm10.m‘):

> deb0:=time():Xnm20:=0:
> for k from 1 to nops(HansZ) do # deb:=time():
> algsubs(1/Z=Xb/e,op(k,HansZ)):
> algsubs(Z=X/e,%):
> H20:=collect(algsubs(e^2=X*Xb,collect(%,e)),[X,Xb],distributed):
> Xnm20:=Xnm20+H20
> # print(k,time()-deb,degree(H20,{X,Xb})):
> od:
> nops(Xnm20); duree:=time()-deb0;
```

$$231$$

$$duree := 18.731$$

```
> save Xnm20, cat(chemin,‘Xnm‘.degmax.‘.m‘):
```

Vérification par comparaison avec le résultat au degré 10 :

```
> simplify(TRONC(Xnm20,10,[X,Xb])-Xnm);
```

$$0$$

Procédure d’extraction des coefficients de Hansen à partir du développement *Hans*Z :

```
> HansenZ:=proc(nn,mm,k) local u,q;
> u:=HansZ; q:=k-mm:
> sort(coeff(u,Z,q),e);
> subs(n=nn,m=mm,%);
> end:
```

Quelques exemples :

```
> HansenZ(5,2,2);
```

$$-\frac{3263010373559}{210691031040000}e^{20}-\frac{19114387847}{1053455155200}e^{18}-\frac{20077951}{928972800}e^{16}-\frac{4289101}{162570240}e^{14}-\frac{247661}{8294400}e^{12}-\frac{2227}{28800}e^{10}+\frac{47}{288}e^{8}-\frac{11}{288}e^{6}-\frac{55}{16}e^{4}+\frac{7}{2}e^{2}+1$$

```
> HansenZ(5,1,1);
```

$$-\frac{4153153310943462 89}{1380784741023744 0000}e^{20}-\frac{124876358389351}{3451961852559360}e^{18}-\frac{951870285227}{21308406497280}e^{16}-\frac{11891446571}{208089907200}e^{14}-\frac{163162421}{2123366400}e^{12}-\frac{824203}{7372800}e^{10}-\frac{24635}{147456}e^{8}-\frac{995}{1152}e^{6}+\frac{167}{64}e^{4}+\frac{13}{2}e^{2}+1$$

```
> TRONC(HansenZ(n,m,m-1),4,e);
```

$$\left(\frac{1}{2}m^3-\frac{5}{8}m^2+\frac{1}{8}m+\frac{3}{16}n+\frac{1}{16}n^2-\frac{5}{16}n\,m-\frac{1}{8}n^2\,m-\frac{1}{16}n^3+\frac{1}{4}n\,m^2\right)e^3+\left(-m-\frac{1}{2}n\right)e$$

2.2.8. Calcul des coefficients de Hansen par le formulaire de Brumberg (degrés très élevés)

C14

Le calcul direct des coefficients $X_{n,\,m,\,k}(e)$ est réalisable par le formulaire suivant (Brumberg, 1967):

- Si $k = 0$, en posant $C_{n,\,m} = \dfrac{(-1)^{|m|}\,(n+|m|+1)!}{(n+1)!\;|m|\,!}$ on a , avec F fonction hypergéométrique :

* pour $n < -1$:

$$X_{n,\,m,\,0}(e) = C_{n,\,m}\,(\frac{e}{2})^{|m|}\,(1-e^2)^{(n+\frac{3}{2})}\,\mathrm{F}(\frac{n+|m|+2}{2},\,\frac{n+|m|+3}{2},\,1+|m|\,,\,e^2)$$

* pour $n \geq -1$:

$$X_{n,\,m,\,0}(e) = C_{n,\,m}\,\beta^{|m|}\,(1+\beta^2)^{(-n-1)}\,\mathrm{F}(-n-1,\,-n+|m|-1,\,1+|m|\,,\,\beta^2) \qquad \text{où} \qquad \beta = \frac{e}{1+\sqrt{1-e^2}}$$

- Si $k \neq 0$, en posant $\nu = \dfrac{k}{1+\beta^2}$ on a :

$$X_{n,\,m,\,k}(e) = \beta^{|m-m|}\,(1+\beta^2)^{(-n-1)}\left(\sum_{s=0}^{\infty} P_{n,\,s+s_1}(m,\,\nu)\,P_{n,\,s+s_2}(-m,\,-\nu)\,\beta^{(2\,s)}\right)$$

avec $s_1 = \max(0,\,k-m)$ et $P_{n,\,s}(m,\,x) = \sum_{r=0}^{r_s} \frac{(-n+m-1)_r\,x^{(s-r)}}{r!\,(s-r)!}$ où $r_s = s$ si $n-m+1 < 0$ et $r_s = \min(s,\,n-m+1)$ sinon.

Les développements limités en excentricité des coefficients de Hansen sont alors calculables par les procédures suivantes. ATTENTION , les indices n, m, k doivent être NUMERIQUES. L'excentricité e est arbitraire.

```
> Pnms:=proc(n,m,s,x) option remember:  local  k,rs;
> k:=n-m+1:  if k<0 then rs:=s else rs:=min(s,k) fi:
> sum(pochhammer(-n+m-1,r)*x^(s-r)/r!/(s-r)!,r=0..rs):
```

```
> end:

> Xnmk:=proc(n,m,k,dmx) local c,am,res,q,s1,s2,nu,t1,t2,t3,d,s;
> if k = 0 then
> am:=abs(m):
> c:=(-1)^am*pochhammer(n+2,am)/am!  :
> if n<-1 then
> res:=c*mtaylor( (e/2)^am*(1-e^2)^(n+3/2)*
> hypergeom([(n+am+2)/2,(n+am+3)/2],[am+1],e^2),e,dmx+1):
> else
> q:=e/(1+sqrt(1-e^2)):
> res:=c*mtaylor(q^am*(1+q^2)^(-n-1)*
> hypergeom([-n-1,-n+am-1],[am+1],q^2),e,dmx+1):
> fi:
> else
> q:=e/(1+sqrt(1-e^2)):  d:=dmx+1:
> nu:=k/(1+q^2):  s1:=max(0,k-m):  s2:=max(0,m-k):
> res:=mtaylor(q^abs(m-k)*(1+q^2)^(-n-1),e,d):
> t3:=0:
> for s from 0 to d/2 do:
> t1:=mtaylor(Pnms(n,m,s+s1,nu),e,d):
> t2:=mtaylor(Pnms(n,-m,s+s2,-nu),e,d):
> t3:=t3+mtaylor(t1*t2*q^(2*s),e,d):
> od:
> res:=mtaylor(res*t3,e,d):
> fi:
```

```
> RETURN(res):
> end:
```

Le calcul numérique des coefficients de Hansen pour une excentricité *e* donnée, à une précision e donnée, peut se faire par la procédure suivante où tous les paramètres sont donnés avec des valeurs numériques (*dg* est en particulier le nombre de chiffres significatifs à conserver dans l'utilisation de la fonction *evalf* ; on pose alors e = 10^(-*dg*))

```
> Xnmke:=proc(n,m,k,e,dg) local c,am,res,q,s1,s2,nu,t1,t2,t3,s,eps;
> if k = 0 then
> am:=abs(m):
> c:=evalf((-1)^am*pochhammer(n+2,am)/am!,dg) :
> if n<-1 then
> res:=evalf(c*(e/2)^am*(1-e^2)^(n+3/2)*
> hypergeom([(n+am+2)/2,(n+am+3)/2],[am+1],e^2),dg):
> else
> q:=evalf(e/(1+sqrt(1-e^2)),dg):
> res:=evalf(c*q^am*(1+q^2)^(-n-1)*
> hypergeom([-n-1,-n+am-1],[am+1],q^2),dg):
> fi:
> else
> q:=evalf(e/(1+sqrt(1-e^2)),dg):  eps:=evalf(10^(-dg)):
> nu:=evalf(k/(1+q^2),dg):  s1:=max(0,k-m):  s2:=max(0,m-k):
> res:=evalf(q^abs(m-k)*(1+q^2)^(-n-1),dg):
> t3:=0:
> for s from 0 to 100 do:
> t1:=evalf(Pnms(n,m,s+s1,nu)*Pnms(n,-m,s+s2,-nu)*q^(2*s),dg):
> t3:=evalf(t3+t1,dg):
```

```
> if abs(t1/t3) < eps then  break fi:
> od:  if s>=100 then print(`ATTENTION non convergence`) fi:
> res:=evalf(res*t3,dg):
> fi:
> RETURN(res):
> end:
```

A des fins de comparaisons, on lit l'ensemble des développements des coefficients de Hansen construits au degré 20 et la procédure d'extraction d'un coefficient de cet ensemble:

```
> read cat(chemin,`HansenZ20.m`);
> HansenZ:=proc(nn,mm,k) local q;
> q:=k-mm:
> sort(coeff(HansZ,Z,q),e);
> subs(n=nn,m=mm,%);
> end:
```

Quelques exemples, pour une excentricité assez forte :

```
> e0:=5/10;
```

$$e0 := \frac{1}{2}$$

Développement en excentricités au degré 20, issu du développement *Hans* calculé au paragraphe précédent :

```
> HansenZ(1,3,1); evalf(subs(e=e0,%),20);
```

$$-\frac{53739053640528409}{3797158037815296000}e^{20}-\frac{31051227754949}{1826434842624000}e^{18}-\frac{1961513551}{93640458240}e^{16}-\frac{12686925827}{475634073600}e^{14}-\frac{2442451}{68812800}e^{12}-\frac{884149}{17694720}e^{10}-\frac{1997}{23040}e^{8}+\frac{155}{1024}e^{6}-\frac{77}{24}e^{4}+\frac{31}{8}e^{2}$$

$$.77019621609167023467$$

valeur exacte, issue du formulaire de Brumberg, sans effectuer de développement :

```
> Xnmke(1,3,1,e0,20);
```

$$.77019621243399430166$$

valeur issue du développement en excentricité au degré 30 par la formule de Brumberg :

```
> Xnmk(1,3,1,30); evalf(subs(e=e0,%),20);
```

$$\frac{31}{8}e^{2}-\frac{77}{24}e^{4}+\frac{155}{1024}e^{6}-\frac{1997}{23040}e^{8}-\frac{884149}{17694720}e^{10}-\frac{2442451}{68812800}e^{12}-\frac{12686925827}{475634073600}e^{14}-\frac{1961513551}{93640458240}e^{16}-\frac{31051227754949}{1826434842624000}e^{18}-\frac{53739053640528409}{3797158037815296000}e^{20}-\frac{87627349608780683959}{7290543432605368320000}e^{22}-\frac{257507120191044203791}{24822564544346849280000}e^{24}-\frac{4202852857680309495601}{463201363107203619225600 0}e^{26}-\frac{43851884884077360717309 77879}{546461808125723469781401600000}e^{28}-\frac{155903434324644737320672071 8131}{2176132355913992128551714816000 00}e^{30}$$

$$.77019621243593639503$$

valeurs issues du développement en excentricité aux degrés 40 et 50 par la formule de Brumberg :

```
> Xnmk(1,3,1,40); evalf(subs(e=e0,%),20);
```

$$\frac{31}{8}e^{2}-\frac{77}{24}e^{4}+\frac{155}{1024}e^{6}-\frac{1997}{23040}e^{8}-\frac{884149}{17694720}e^{10}-\frac{2442451}{68812800}e^{12}-\frac{12686925827}{475634073600}e^{14}-\frac{1961513551}{93640458240}e^{16}$$
$$-\frac{31051227754949}{1826434842624000}e^{18}-\frac{53739053640528409}{3797158037815296000}e^{20}-\frac{87627349608780683959}{7290543432605368320000}e^{22}$$
$$-\frac{25750712019104420 3791}{24822564544346849280000}e^{24}-\frac{42028528576803094956001}{4632013631072036192256000}e^{26}$$
$$-\frac{4385188488407736071730977879}{546461808125723469781401600000}e^{28}-\frac{155903434324644737320672 0718131}{21761323559139921285517148 1600000}e^{30}$$
$$-\frac{115008442997028009382055423707033}{1783651341722361405366494822 4000000}e^{32}-\frac{12226393830214024316379739 03982183027}{209212573376714507241824207241216000000}e^{34}$$
$$-\frac{26409192176904576673007722644859 8810251}{495554848804977762820134272218693632 00000}e^{36}$$
$$-\frac{29886328505907711270744784034817136 33232777}{611656270525001124395137158852787568640000000}e^{38}$$
$$-\frac{3845060427007541768627846249194005460112256457}{854177981788164070217809042337917839605760000000}e^{40}$$

$$.77019621243399553535$$

rappel de la valeur exacte

```
> Xnmke(1,3,1,e0,20);
```

$$.77019621243399430166$$

Au degré 50, On obtient 17 chiffres exacts:

```
> Xnmk(1,3,1,50); evalf(subs(e=e0,%),20);
```

$$-\frac{38450604270075417686278462491940054601122564 57}{854177981788164070217809042337917839605760000000}e^{40}+\frac{31}{8}e^{2}-\frac{31051227754949}{1826434842624000}e^{18}-\frac{1961513551}{93640458240}e^{16}$$
$$-\frac{12686925827}{475634073600}e^{14}-\frac{2442451}{68812800}e^{12}-\frac{53739053640528409}{3797158037815296000}e^{20}$$
$$-\frac{115008442997028009382055423707033}{1783651341722361405366494822400 0000}e^{32}-\frac{1559034343246447373206720718131}{217613235591399212855171481600000}e^{30}$$
$$-\frac{438518848840773607173097 7879}{546461808125723469781401600000}e^{28}-\frac{42028528576803094956001}{4632013631072036192256000}e^{26}$$
$$-\frac{2575071201910442037 91}{248225645443468492800 00}e^{24}-\frac{166971454504410449866920740732101504847082985 24183}{400894199452578336955558377203929439388303360000 0000}e^{42}$$
$$+\frac{155}{1024}e^{6}-\frac{101117111130074364322681841867108259808368439443616 4317}{280423247509561612261726133727236300847257625821184000000}e^{46}$$
$$-\frac{4004294430628878602977262095455540254184625995163225 5239}{118760897645831806539714613662892057124390987366400000000000}e^{48}$$
$$-\frac{224723925710934906388579110077431670532063849581904 67}{580895695006786010248604088568493757673651568640 0000000}e^{44}$$
$$-\frac{890801640617175213628555478878133252665647164651000018262455609}{281724411378005978142620542987730677283188901204994293760000000000}e^{50}$$
$$-\frac{876273496087806839 59}{729054343260536832000 0}e^{22}-\frac{77}{24}e^{4}-\frac{12226393830214024316379739039821830 27}{209212573376714507241824207241216000000}e^{34}$$
$$-\frac{264091921769045766730077226448598810251}{495554848804977762820134272218693632 00000}e^{36}$$
$$-\frac{298863285059077112707447840348171363323277 7}{611656270525001124395137158852787568640000000}e^{38}-\frac{884149}{17694720}e^{10}-\frac{1997}{23040}e^{8}$$

$$.77019621243399430241$$

2.3. Fonctions des inclinaisons intervenant dans les perturbations de 2 planètes

On considère 2 planètes orbitant dans des plans non confondus, repérés chacun par son inclinaison et sa longitude du nœud par rapport à un même repère de référence $\mathrm{O}\, x_0\, y_0\, z_0$. On développe ici les fonctions de ces inclinaisons qui interviennent dans le calcul de la distance mutuelle de ces 2 planètes. C25.1.3

2.3.1. Calcul du cosinus de l'angle entre les directions des 2 planètes

Soit S l'angle entre les vecteurs unitaires des directions des 2 planètes. On calcule ici $\cos(S)$ en fonction des éléments d'orbite de ces planètes.

Le résultat est exprimé en fonction de la variable complexe $Y = \sin(\frac{i}{2})\, \exp\left(I\,(L - \Omega)\right)$ de chaque planète et de son conjugué noté Yb , où i est l'inclinaison de l'orbite de cette planète sur un plan extérieur fixe, L sa longitude moyenne, et Ω sa longitude du nœud. On aura aussi besoin de $\theta = \exp\left(I\,(v - M)\right)$, la fonction de l'équation du centre déjà rencontrée précédemment. Comme on l'a vu à cette occasion, cette fonction se développe en fonction de l'excentricité e par l'intermédiaire des variables (X, Xb) , ou (x, xb, L) avec $X = xb\, \exp\left(I\, L\right)$ où $x = eexp(I\,\varpi)$. De manière analogue en définissant

$$y = \gamma\, \exp\left(I\,\Omega\right) \qquad \text{avec} \qquad \gamma = \sin(\frac{i}{2})$$

et yb conjugué de y on a encore $Y = yb\, \exp\left(I\, L\right)$. On pourra alors obtenir finalement des développements fonctions uniquement des variables (x, xb, y, yb, L) associées à chaque planète (les deux planètes seront dans la suite indicées par i et j , sans confusion avec l'inclinaison i qui ne sera plus utilisée autrement que dans la variable γ).

Si ce n'est pas déjà fait, commencer par exécuter la phase d' initialisations , puis :

```
> with(linalg,multiply,dotprod,vector,matrix):
```

Définition de l'opérateur matrice de rotation autour du premier axe :

```
> R1:=proc(ang) local u;
> u:=matrix(3,3,[1,0,0,0,cos(ang),sin(ang),0,-sin(ang),cos(ang)]);
> end:
```

Définission de l'opérateur matrice de rotation autour du troisième axe :

```
> R3:=proc(ang) local u;
> u:=matrix(3,3,[cos(ang),sin(ang),0,-sin(ang),cos(ang),0,0,0,1]);
> end:
```

Les vecteurs unitaires des planètes i et j sont d'abord donnés dans deux repères dont les plans de base sont situés chacun dans le plan de chaque orbite et dont les premiers axes sont suivant le nœud de chaque orbite avec le plan extérieur fixe (pour chacune, w est la longitude vraie dans l'orbite et Ω la longitude du nœud). L'angle que fait chaque rayon vecteur avec le nœud de l'orbite correspondante est alors $w - \Omega$.

```
> OP[i]:=vector(3,[cos(w[i]-Omega[i]),sin(w[i]-Omega[i]),0]);
> OP[j]:=vector(3,[cos(w[j]-Omega[j]),sin(w[j]-Omega[j]),0]);
```

$$OP_i := [\cos(w_i - \Omega_i),\ \sin(w_i - \Omega_i),\ 0]$$

$$OP_j := [\cos(-w_j + \Omega_j),\ -\sin(-w_j + \Omega_j),\ 0]$$

Par 2 rotations (d'angles $-i$ et $-\Omega$ autour des premier et troisième axes), on ramène ces 2 vecteurs unitaires dans le même repère extérieur:

```
> OP[i]:=multiply(R1(-i[i]),OP[i]):
> OP[i]:=multiply(R3(-Omega[i]),OP[i]);
> OP[j]:=multiply(R1(-i[j]),OP[j]):
> OP[j]:=multiply(R3(-Omega[j]),OP[j]);
```

$$OP_i := [\cos(\Omega_i)\cos(w_i-\Omega_i)-\sin(\Omega_i)\cos(i_i)\sin(w_i-\Omega_i),\ \sin(\Omega_i)\cos(w_i-\Omega_i)+\cos(\Omega_i)\cos(i_i)\sin(w_i-\Omega_i),\ \sin(i_i)\sin(w_i-\Omega_i)]$$

$$OP_j := [\cos(\Omega_j)\cos(w_j-\Omega_j)-\sin(\Omega_j)\cos(i_j)\sin(w_j-\Omega_j),\ \sin(\Omega_j)\cos(w_j-\Omega_j)+\cos(\Omega_j)\cos(i_j)\sin(w_j-\Omega_j),\ \sin(i_j)\sin(w_j-\Omega_j)]$$

Leur produit scalaire donne $\cos(S)$:

```
> cosS:=dotprod(OP[i],OP[j],'orthogonal');
```

$$cosS := (\cos(\Omega_i)\cos(w_i-\Omega_i)-\sin(\Omega_i)\cos(i_i)\sin(w_i-\Omega_i))\,(\cos(\Omega_j)\cos(-w_j+\Omega_j)-\sin(\Omega_j)\cos(i_j)\sin(w_j-\Omega_j))$$
$$+\,(\sin(\Omega_i)\cos(w_i-\Omega_i)+\cos(\Omega_i)\cos(i_i)\sin(w_i-\Omega_i))\,(\sin(\Omega_j)\cos(-w_j+\Omega_j)+\cos(\Omega_j)\cos(i_j)\sin(w_j-\Omega_j))$$
$$+\,\sin(i_i)\sin(w_i-\Omega_i)\sin(i_j)\sin(w_j-\Omega_j)$$

On isole ensuite les sinus et cosinus des inclinaisons pour que ces dernières ne soient pas mélangées aux longitudes lorsqu'on fera des combinaisons d'angles. Pour cela, on introduit les sinus des demi-inclinaisons (notés γ) et leurs cosinus (notés C) :

```
> u:=subs(cos(i[i])=1-2*gamma[i]^2,cos(i[j])=1-2*gamma[j]^2,sin(i[i])=
> 2*gamma[i]*C[i],sin(i[j])=2*gamma[j]*C[j],expand(cosS)):
> cosS:=collect(combine(u,trig),cos);
```

$$cosS := 2\,\gamma_i\,C_i\,\gamma_j\,C_j\cos(w_i-\Omega_i-w_j+\Omega_j)-2\,\gamma_i\,C_i\,\gamma_j\,C_j\cos(w_i-\Omega_i+w_j-\Omega_j)+(-{\gamma_i}^2{\gamma_j}^2+{\gamma_j}^2)\cos(w_j-2\,\Omega_j+w_i)$$
$$+\,{\gamma_i}^2{\gamma_j}^2\cos(-2\,\Omega_i-w_j+2\,\Omega_j+w_i)+({\gamma_i}^2-{\gamma_i}^2{\gamma_j}^2)\cos(w_i-2\,\Omega_i+w_j)+(-{\gamma_i}^2+1+{\gamma_i}^2{\gamma_j}^2-{\gamma_j}^2)\cos(w_i-w_j)$$

On aura besoin par la suite de la différence $\Delta_{\cos}=\cos(S)-\cos(w_i-w_j)$. On l'obtient dès maintenant; c'est une fonction qui est au moins de l'ordre du carré des inclinaisons :

```
> DeltacosS:=collect(cosS-cos(w[i]-w[j]),cos);
```

$$DeltacosS := 2\,\gamma_i\,C_i\,\gamma_j\,C_j\cos(w_i-\Omega_i-w_j+\Omega_j)-2\,\gamma_i\,C_i\,\gamma_j\,C_j\cos(w_i-\Omega_i+w_j-\Omega_j)+(-{\gamma_i}^2\,{\gamma_j}^2+{\gamma_j}^2)\cos(w_j-2\,\Omega_j+w_i)$$
$$+{\gamma_i}^2\,{\gamma_j}^2\cos(-2\,\Omega_i-w_j+2\,\Omega_j+w_i)+({\gamma_i}^2-{\gamma_i}^2\,{\gamma_j}^2)\cos(w_i-2\,\Omega_i+w_j)+(-{\gamma_i}^2+{\gamma_i}^2\,{\gamma_j}^2-{\gamma_j}^2)\cos(w_i-w_j)$$

On substitue enfin aux variables γ et Ω les variables Y et Yb, et aux longitudes w les fonctions θ de $v-M$ (en identifiant w à $\varpi+v=\varpi+M+v-M=L+v-M$) Ceci est réalisé par cette procédure de conversion:

```
> ENYYBTTA:=proc(s,i) local ss:
> ss:=subs(gamma[i]=Y[i]*exp(I*Omega[i])/exp(I*L[i]),s):
> algsubs(exp(I*w[i])=theta[i]*exp(I*L[i]),%):
> algsubs(1/exp(I*w[i])=1/theta[i]/exp(I*L[i]),%):
> algsubs(exp(I*Omega[i])^2=Yb[i]/Y[i]*exp(I*L[i])^2,%):
> algsubs(1/exp(I*Omega[i])^2=Y[i]/Yb[i]/exp(I*L[i])^2,%):
> # subs(C[i]=sqrt(1-Y[i]*Yb[i]),%):
> end:

> expand(convert(cosS,exp)):  ENYYBTTA(%,i):  ENYYBTTA(%,j):
> COSS:=collect(combine(expand(%),exp),exp);
```

$$COSS := \left(\frac{1}{2}\,\frac{\theta_i}{\theta_j}-\frac{1}{2}\,\frac{{Yb_i}^2\,Y_j\,Yb_j}{\theta_i\,\theta_j}-\frac{1}{2}\,\frac{\theta_i\,Y_i\,Yb_i}{\theta_j}+\frac{1}{2}\,\frac{{Yb_i}^2}{\theta_i\,\theta_j}-\frac{1}{2}\,\frac{\theta_i\,Y_j\,Yb_j}{\theta_j}+\frac{1}{2}\,\frac{\theta_j\,{Yb_i}^2\,{Y_j}^2}{\theta_i}+\frac{1}{2}\,\theta_i\,\theta_j\,{Y_j}^2-\frac{1}{2}\,\theta_i\,\theta_j\,Y_i\,{Y_j}^2\,Yb_i\right.$$
$$\left.+\frac{1}{2}\,\frac{\theta_i\,Y_i\,Y_j\,Yb_i\,Yb_j}{\theta_j}\right)e^{(I\,L_i-I\,L_j)}+\left(\frac{1}{2}\,\frac{\theta_i\,{Yb_j}^2\,{Y_i}^2}{\theta_j}+\frac{1}{2}\,\frac{{Yb_j}^2}{\theta_i\,\theta_j}-\frac{1}{2}\,\frac{\theta_j\,Y_j\,Yb_j}{\theta_i}-\frac{1}{2}\,\frac{{Yb_j}^2\,Y_i\,Yb_i}{\theta_i\,\theta_j}-\frac{1}{2}\,\theta_i\,\theta_j\,{Y_i}^2\,Y_j\,Yb_j\right.$$
$$\left.+\frac{1}{2}\,\theta_i\,\theta_j\,{Y_i}^2+\frac{1}{2}\,\frac{\theta_j\,Y_i\,Y_j\,Yb_i\,Yb_j}{\theta_i}-\frac{1}{2}\,\frac{\theta_j\,Y_i\,Yb_i}{\theta_i}+\frac{1}{2}\,\frac{\theta_j}{\theta_i}\right)e^{(-I\,L_i+I\,L_j)}-\frac{C_i\,C_j\,Yb_i\,Yb_j}{\theta_i\,\theta_j}+\frac{\theta_j\,C_i\,Y_j\,C_j\,Yb_i}{\theta_i}$$
$$+\frac{\theta_i\,Y_i\,C_i\,C_j\,Yb_j}{\theta_j}-\theta_i\,\theta_j\,Y_i\,C_i\,Y_j\,C_j$$

```
> expand(convert(DeltacosS*2,exp)):  ENYYBTTA(%,i):  ENYYBTTA(%,j):
> DDELTACOSS:=collect(combine(expand(%),exp),exp);
```

Pour une utilisation ultérieure, notons qu'on a calculé ici l'expression du <u>double</u> de $\Delta_{\cos}$.

$$DDELTACOSS := \left(-\frac{\theta_i\,Y_i\,Yb_i}{\theta_j}+\theta_i\,\theta_j\,{Y_j}^2-\frac{{Yb_i}^2\,Y_j\,Yb_j}{\theta_i\,\theta_j}+\frac{\theta_j\,{Yb_i}^2\,{Y_j}^2}{\theta_i}-\theta_i\,\theta_j\,Y_i\,{Y_j}^2\,Yb_i-\frac{\theta_i\,Y_j\,Yb_j}{\theta_j}+\frac{\theta_i\,Y_i\,Y_j\,Yb_i\,Yb_j}{\theta_j}+\frac{{Yb_i}^2}{\theta_i\,\theta_j}\right)e^{(I\,L_i-I\,L_j)}$$
$$+\left(-\frac{{Yb_j}^2\,Y_i\,Yb_i}{\theta_i\,\theta_j}+\frac{{Yb_j}^2}{\theta_i\,\theta_j}+\frac{\theta_i\,{Yb_j}^2\,{Y_i}^2}{\theta_j}-\theta_i\,\theta_j\,{Y_i}^2\,Y_j\,Yb_j-\frac{\theta_j\,Y_i\,Yb_i}{\theta_i}+\frac{\theta_j\,Y_i\,Y_j\,Yb_i\,Yb_j}{\theta_i}+\theta_i\,\theta_j\,{Y_i}^2-\frac{\theta_j\,Y_j\,Yb_j}{\theta_i}\right)e^{(-I\,L_i+I\,L_j)}$$
$$+2\,\frac{\theta_i\,Y_i\,C_i\,C_j\,Yb_j}{\theta_j}-2\,\theta_i\,\theta_j\,Y_i\,C_i\,Y_j\,C_j-2\,\frac{C_i\,C_j\,Yb_i\,Yb_j}{\theta_i\,\theta_j}+2\,\frac{\theta_j\,C_i\,Y_j\,C_j\,Yb_i}{\theta_i}$$

On observe qu'avec ces variables Y, Yb on obtient des expressions où n'interviennent que les arguments $\pm\{L_i - L_j\}$.

Sauvegrade des résultats principaux :

```
> save cosS, DeltacosS, COSS, DDELTACOSS, cat(chemin,`cosS.m`);
```

REMARQUE : Quand on remplace les variables Y et Yb par les expressions suivantes, analogues à celles introduites pour les excentricités:

$$Y = yb\,\exp\,(I\,L) \qquad \text{et} \qquad Yb = y\,\exp\,(-I\,L)$$

(tout comme on avait défini $X = xb\,\exp\,(I\,L)$ et $Xb = x\,\exp\,(-I\,L)$), on obtient une expression où interviennent 4 arguments :

sommes et différences des longitudes moyennes

$$\pm\{L_i + L_j\} \qquad \text{et} \qquad \pm\{L_i - L_j\}$$

```
> COSSyij:=collect(combine(expand( subs(
> X[i]=xb[i]*exp(I*L[i]),Xb[i]=x[i]/exp(I*L[i]),
> Y[i]=yb[i]*exp(I*L[i]),Yb[i]=y[i]/exp(I*L[i]),
> X[j]=xb[j]*exp(I*L[j]),Xb[j]=x[j]/exp(I*L[j]),
> Y[j]=yb[j]*exp(I*L[j]),Yb[j]=y[j]/exp(I*L[j]),COSS)),exp),exp):
> nops(%);
```

```
> varxyij:=[x[i],xb[i],y[i],yb[i],x[j],xb[j],y[j],yb[j]]:
> varXYij:=[X[i],Xb[i],Y[i],Yb[i],X[j],Xb[j],Y[j],Yb[j]]:
> collect(TRONC(COSSyij,4,varxyij),exp);
```

$$4$$

$$\begin{aligned}&(\frac{1}{2}\frac{{y_j}^2}{\theta_i\,\theta_j} - \frac{1}{2}\frac{{y_i}^2\,yb_j\,y_j}{\theta_i\,\theta_j} - \frac{1}{2}\frac{{y_j}^2\,yb_i\,y_i}{\theta_i\,\theta_j} - \frac{C_i\,C_j\,y_i\,y_j}{\theta_i\,\theta_j} + \frac{1}{2}\frac{{y_i}^2}{\theta_i\,\theta_j})\,e^{(-I\,L_i - I\,L_j)}\\
&+ (\frac{1}{2}\theta_i\,\theta_j\,{yb_j}^2 - \frac{1}{2}\theta_i\,\theta_j\,{yb_i}^2\,yb_j\,y_j + \frac{1}{2}\theta_i\,\theta_j\,{yb_i}^2 - \frac{1}{2}\theta_i\,\theta_j\,yb_i\,{yb_j}^2\,y_i - \theta_i\,\theta_j\,yb_i\,C_i\,yb_j\,C_j)\,e^{(I\,L_i + I\,L_j)}\\
&+ (\frac{1}{2}\frac{\theta_i}{\theta_j} - \frac{1}{2}\frac{\theta_i\,yb_j\,y_j}{\theta_j} + \frac{1}{2}\frac{\theta_i\,yb_i\,yb_j\,y_i\,y_j}{\theta_j} + \frac{1}{2}\frac{\theta_i\,{y_j}^2\,{yb_i}^2}{\theta_j} - \frac{1}{2}\frac{\theta_i\,yb_i\,y_i}{\theta_j} + \frac{\theta_i\,yb_i\,C_i\,C_j\,y_j}{\theta_j})\,e^{(I\,L_i - I\,L_j)}\\
&+ (\frac{1}{2}\frac{\theta_j}{\theta_i} - \frac{1}{2}\frac{\theta_j\,yb_j\,y_j}{\theta_i} - \frac{1}{2}\frac{\theta_j\,yb_i\,y_i}{\theta_i} + \frac{1}{2}\frac{\theta_j\,{y_i}^2\,{yb_j}^2}{\theta_i} + \frac{\theta_j\,C_i\,yb_j\,C_j\,y_i}{\theta_i} + \frac{1}{2}\frac{\theta_j\,yb_i\,yb_j\,y_i\,y_j}{\theta_i})\,e^{(-I\,L_i + I\,L_j)}\end{aligned}$$

On remarque qu'en étendant la notion d'inégalité aux combinaisons entières de longitudes moyennes $p\,L_i + q\,L_j$, les termes de *COSSyij* dépendent de 4 inégalités, avec p et q égaux à +1 ou -1, que l'on peut écrire sous la forme :

$$C_{\{p,\,q,\,n_5,\,n_6,\,n_7,\,n_8\}}\,{\theta_i}^{\tau_i}\,{\theta_j}^{\tau_j}\,{y_i}^{n_5}\,{yb_i}^{n_6}\,{y_j}^{n_7}\,{yb_j}^{n_8}\,\exp\left(I\,(p\,L_i + q\,L_j)\right)$$

Chaque terme vérifie la propriété suivante :

$$n_6 - n_5 + n_8 - n_7 = p + q$$

comparable à la propriété de d'Alembert de rang 0 vue pour les séries en excentricités. Cela traduit en fait la même propriété d'invariance vis-à-vis des rotations du repère dans lequel sont mesurées les longitudes moyennes et celles des nœuds: Dans toute rotation du repère $\mathrm{O}\, x_0\, y_0\, z_0$ autour de l'axe $\mathrm{O}\, z_0$, les inclinaisons sont inchangées tandis que les longitudes moyennes et celles des nœuds augmentent toutes de la même quantité ϕ, mais dans chaque expression :

$${\gamma_i}^{(n_5+n_6)}\, {\gamma_j}^{(n_7+n_8)}\, \exp\left(I\left(p\, L_i + q\, L_j\right) + (n_5 - n_6)\, \Omega_i + (n_7 - n_8)\, \Omega_j\right)$$

l'argument est augmenté de $(p + q + n_5 - n_6 + n_7 - n_8)\, \phi$ c'est-à-dire zéro. C(6.70)

Notons encore que, dans $\cos(S)$, le degré global $n_6 + n_5 + n_8 + n_7$ des termes en inclinaisons est un entier pair .

Evidemment, la même propriété s'applique aux termes de l'expression analogue qu'on obtiendrait pour *DDELTACOSSyij* .

Enfin, si on remplaçait les fonctions θ_i et θ_j par leur développement en (x, xb, L), on obtiendrait de la même façon des termes de la forme :

$$C_{\{p,\, q,\, n_1,\, n_2,\, n_3,\, n_4,\, n_5,\, n_6,\, n_7,\, n_8\}}\, {x_i}^{n_1}\, {xb_i}^{n_2}\, {x_j}^{n_3}\, {xb_j}^{n_4}\, {y_i}^{n_5}\, {yb_i}^{n_6}\, {y_j}^{n_7}\, {yb_j}^{n_8}\, \exp\left(I\left(p\, L_i + q\, L_j\right)\right)$$

où p et q deviennent quelconques, mais on a toujours la propriété analogue due à la même invariance par rotation :

$$n_2 - n_1 + n_4 - n_3 + n_6 - n_5 + n_8 - n_7 = p + q$$

Pour la suite, il sera intéressant de rester en variables (Y, Yb) , mais en substituant une variable notée σ à $\exp\left(I\left(L_i - L_j\right)\right)$. On obtient :

```
>  COSSsig:=expand(subs(exp(I*L[i]-I*L[j])=sigma,exp(-I*L[i]+I*L[j])=1/s
>  igma,COSS));
```

$$COSSsig := \frac{1}{2}\frac{\theta_i\,\sigma}{\theta_j} - \frac{1}{2}\frac{\sigma\,{Yb_i}^2\,Y_j\,Yb_j}{\theta_i\,\theta_j} - \frac{1}{2}\frac{\sigma\,\theta_i\,Y_i\,Yb_i}{\theta_j} + \frac{1}{2}\frac{\sigma\,{Yb_i}^2}{\theta_i\,\theta_j} - \frac{1}{2}\frac{\sigma\,\theta_i\,Y_j\,Yb_j}{\theta_j} + \frac{1}{2}\frac{\theta_j\,\sigma\,{Yb_i}^2\,{Y_j}^2}{\theta_i} + \frac{1}{2}\theta_i\,\theta_j\,\sigma\,{Y_j}^2$$
$$- \frac{1}{2}\theta_i\,\theta_j\,\sigma\,Y_i\,Yb_i\,{Y_j}^2 + \frac{1}{2}\frac{\sigma\,\theta_i\,Y_i\,Y_j\,Yb_i\,Yb_j}{\theta_j} + \frac{1}{2}\frac{\theta_i\,{Y_i}^2\,{Yb_j}^2}{\theta_j\,\sigma} + \frac{1}{2}\frac{{Yb_j}^2}{\theta_i\,\theta_j\,\sigma} - \frac{1}{2}\frac{\theta_j\,Y_j\,Yb_j}{\sigma\,\theta_i} - \frac{1}{2}\frac{Y_i\,Yb_i\,{Yb_j}^2}{\theta_i\,\theta_j\,\sigma}$$
$$- \frac{1}{2}\frac{\theta_i\,\theta_j\,{Y_i}^2\,Y_j\,Yb_j}{\sigma} + \frac{1}{2}\frac{\theta_i\,\theta_j\,{Y_i}^2}{\sigma} + \frac{1}{2}\frac{\theta_j\,Y_i\,Y_j\,Yb_i\,Yb_j}{\sigma\,\theta_i} - \frac{1}{2}\frac{\theta_j\,Y_i\,Yb_i}{\sigma\,\theta_i} + \frac{1}{2}\frac{\theta_j}{\theta_i\,\sigma} - \frac{C_i\,C_j\,Yb_i\,Yb_j}{\theta_i\,\theta_j} + \frac{\theta_j\,C_i\,C_j\,Yb_i\,Y_j}{\theta_i}$$
$$+ \frac{\theta_i\,C_i\,C_j\,Y_i\,Yb_j}{\theta_j} - \theta_i\,\theta_j\,C_i\,C_j\,Y_i\,Y_j$$

Pour substituer à θ_i et θ_j leur développement en fonction des variables X, Xb, il faut prendre des précautions pour ne pas engendrer les nombreux termes de degré trop élevé qui ne manqueraient pas d'arriver dans leurs produits. Pour obtenir une expression développée à un degré maximum donné, on commence par éliminer les fonctions θ des dénominateurs (en donnant à leur inverse les noms *Tbi* ou *Tbj*).

```
> read cat(chemin,‘kepler20.m‘):
> THETA10:=TRONC(THETA,10,[X,Xb]):
> THETAB10:=TRONC(THETAB,10,[X,Xb]):
> THETAi:=subs(X=X[i],Xb=Xb[i],THETA10):
> THETAj:=subs(i=j,THETAi):
> THETABi:=subs(X=X[i],Xb=Xb[i],THETAB10):
> THETABj:=subs(i=j,THETABi):
> varXYij:=[X[i],Xb[i],Y[i],Yb[i],X[j],Xb[j],Y[j],Yb[j]]:
> varXij:=[X[i],Xb[i],X[j],Xb[j]]:
> varXYije:=[X[i],Xb[i],Y[i],Yb[i],X[j],Xb[j],Y[j],Yb[j],Y[e],Yb[e]]:
```

```
> CONVTHETA:=proc(p,i,j) local s:
> s:=collect(combine(expand(p),exp),exp):
> s:=algsubs(1/theta[i]='Tbi',s):
> s:=algsubs(1/theta[j]='Tbj',s):
> s:=subs(theta[i]='Ti',theta[j]='Tj',s):
> s:=subs(exp(I*L[i]-I*L[j])=sigma,exp(-I*L[i]+I*L[j])=1/sigma,s):
> expand(sort(collect(s,sigma),varXYije));
> end:
> COSSXij:= CONVTHETA(COSS,i,j); nops(%);
```

$$COSSXij := -C_i\,C_j\,Ti\,Tbj\,Y_i\,Tj^2\,Y_j + C_i\,C_j\,Ti\,Tbj\,Y_i\,Yb_j + C_i\,C_j\,Tbi\,Tbj\,Yb_i\,Tj^2\,Y_j - C_i\,C_j\,Tbi\,Tbj\,Yb_i\,Yb_j$$
$$+\frac{1}{2}\sigma\,Tbi\,Tbj\,Yb_i^2\,Tj^2\,Y_j^2 - \frac{1}{2}\sigma\,Tbi\,Tbj\,Yb_i^2\,Y_j\,Yb_j + \frac{1}{2}\sigma\,Tbi\,Tbj\,Yb_i^2 + \frac{1}{2}\sigma\,Ti\,Tbj\,Y_i\,Yb_i\,Y_j\,Yb_j - \frac{1}{2}\sigma\,Ti\,Tbj\,Y_i\,Yb_i$$
$$-\frac{1}{2}\sigma\,Ti\,Tbj\,Y_j\,Yb_j + \frac{1}{2}\sigma\,Ti\,Tbj - \frac{1}{2}\sigma\,Ti\,Tbj\,Tj^2\,Y_i\,Yb_i\,Y_j^2 + \frac{1}{2}\sigma\,Ti\,Tbj\,Tj^2\,Y_j^2 + \frac{1}{2}\,\frac{Ti\,Tbj\,Y_i^2\,Yb_j^2}{\sigma}$$
$$-\frac{1}{2}\,\frac{Ti\,Tbj\,Y_i^2\,Tj^2\,Y_j\,Yb_j}{\sigma} + \frac{1}{2}\,\frac{Ti\,Tbj\,Y_i^2\,Tj^2}{\sigma} - \frac{1}{2}\,\frac{Tbi\,Tbj\,Y_i\,Yb_i\,Yb_j^2}{\sigma} + \frac{1}{2}\,\frac{Tbi\,Tbj\,Yb_j^2}{\sigma} + \frac{1}{2}\,\frac{Tbi\,Tbj\,Tj^2\,Y_i\,Yb_i\,Y_j\,Yb_j}{\sigma}$$
$$-\frac{1}{2}\,\frac{Tbi\,Tbj\,Tj^2\,Y_i\,Yb_i}{\sigma} - \frac{1}{2}\,\frac{Tbi\,Tbj\,Tj^2\,Y_j\,Yb_j}{\sigma} + \frac{1}{2}\,\frac{Tbi\,Tbj\,Tj^2}{\sigma}$$

$$22$$

```
> DDELTACOSSXij:= CONVTHETA(DDELTACOSS,i,j); nops(%);
```

$$DDELTACOSSXij := -2\,C_i\,C_j\,Ti\,Tbj\,Y_i\,Tj^2\,Y_j + 2\,C_i\,C_j\,Ti\,Tbj\,Y_i\,Yb_j + 2\,C_i\,C_j\,Tbi\,Tbj\,Yb_i\,Tj^2\,Y_j - 2\,C_i\,C_j\,Tbi\,Tbj\,Yb_i\,Yb_j + \sigma\,Tbi\,Tbj\,{Yb_i}^2\,Tj^2\,{Y_j}^2 - \sigma\,Tbi\,Tbj\,{Yb_i}^2\,Y_j\,Yb_j + \sigma\,Tbi\,Tbj\,{Yb_i}^2 + \sigma\,Ti\,Tbj\,Y_i\,Yb_i\,Y_j\,Yb_j - \sigma\,Ti\,Tbj\,Y_i\,Yb_i - \sigma\,Ti\,Tbj\,Y_j\,Yb_j - \sigma\,Ti\,Tbj\,Tj^2\,Y_i\,Yb_i\,{Y_j}^2 + \sigma\,Ti\,Tbj\,Tj^2\,{Y_j}^2 + \frac{Ti\,Tbj\,{Y_i}^2\,{Yb_j}^2}{\sigma} - \frac{Ti\,Tbj\,{Y_i}^2\,Tj^2\,Y_j\,Yb_j}{\sigma} + \frac{Ti\,Tbj\,{Y_i}^2\,Tj^2}{\sigma} - \frac{Tbi\,Tbj\,Y_i\,Yb_i\,{Yb_j}^2}{\sigma} + \frac{Tbi\,Tbj\,{Yb_j}^2}{\sigma} + \frac{Tbi\,Tbj\,Tj^2\,Y_i\,Yb_i\,Y_j\,Yb_j}{\sigma} - \frac{Tbi\,Tbj\,Tj^2\,Y_i\,Yb_i}{\sigma} - \frac{Tbi\,Tbj\,Tj^2\,Y_j\,Yb_j}{\sigma}$$

$$20$$

Ces formulations sont ainsi utilisables par la procédure *SUBS* qui substitue des séries entières à des variables, en ne conservant que les termes de degré inférieur à un degré donné (ici égal à 10):

```
> deb:=time():
> COSSXij10:=
> SUBS([Ti=THETAi,Tj=THETAj,Tbi=THETABi,Tbj=THETABj],COSSXij,10,varXYij)
> : print("durée=",time()-deb): nops(%); TRONC(%%,2,varXYij);
> save COSSXij10, cat(chemin,‘COSSXij10.m‘);
```

"durée=", 308.540

$$8374$$

$$(\frac{9}{16}\,\sigma - \frac{1}{16}\,\frac{1}{\sigma})\,{X_i}^2 - (\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,X_i\,Xb_i - (\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,X_i\,X_j + (\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,X_i\,Xb_j + (\frac{9}{16}\,\frac{1}{\sigma} - \frac{1}{16}\,\sigma)\,{Xb_i}^2 + (\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,Xb_i\,X_j$$
$$+ \frac{1}{2}\,\frac{{Y_i}^2}{\sigma} - (\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,Y_i\,Yb_i - C_i\,C_j\,Y_i\,Y_j + C_i\,C_j\,Y_i\,Yb_j + \frac{1}{2}\,\sigma\,{Yb_i}^2 + C_i\,C_j\,Yb_i\,Y_j - C_i\,C_j\,Yb_i\,Yb_j + (\frac{9}{16}\,\frac{1}{\sigma} - \frac{1}{16}\,\sigma)\,{X_j}^2$$
$$- (\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,X_j\,Xb_j + (\frac{9}{16}\,\sigma - \frac{1}{16}\,\frac{1}{\sigma})\,{Xb_j}^2 + \frac{1}{2}\,\sigma\,{Y_j}^2 - (\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,Y_j\,Yb_j + \frac{1}{2}\,\frac{{Yb_j}^2}{\sigma} - (\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,Xb_i\,Xb_j$$
$$+ (-\frac{1}{2}\,\frac{1}{\sigma} + \frac{1}{2}\,\sigma)\,X_i + (-\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,Xb_i + (-\frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma})\,X_j + (-\frac{1}{2}\,\frac{1}{\sigma} + \frac{1}{2}\,\sigma)\,Xb_j + \frac{1}{2}\,\sigma + \frac{1}{2}\,\frac{1}{\sigma}$$

```
> deb:=time():
> DDELTACOSSXij10:=
> SUBS([Ti=THETAi,Tj=THETAj,Tbi=THETABi,Tbj=THETABj],DDELTACOSSXij,10,va
> rXYij):  print("durée=",time()-deb):  nops(%);
> TRONC(%%,2,varXYij);
> save DDELTACOSSXij10, cat(chemin,`DDELTACOSSXij10.m`);
```

$$\text{“durée=”},\ 192.531$$

$$6640$$

$$\frac{{Y_i}^2}{\sigma} + (-\sigma - \frac{1}{\sigma})\,Y_i\,Yb_i - 2\,C_i\,C_j\,Y_i\,Y_j + 2\,C_i\,C_j\,Y_i\,Yb_j + \sigma\,{Yb_i}^2 + 2\,C_i\,C_j\,Yb_i\,Y_j - 2\,C_i\,C_j\,Yb_i\,Yb_j + \sigma\,{Y_j}^2 + (-\sigma - \frac{1}{\sigma})\,Y_j\,Yb_j$$
$$+ \frac{{Yb_j}^2}{\sigma}$$

2.3.2. Calcul des puissances de $\cos(S)$ et de $2\,\Delta_{\cos}$ $C_{25.1.6}$

Ces puissances interviennent dans le développement de l'inverse de la distance de 2 planètes, c'est-à-dire dans la fonction perturbatrice représentant leurs perturbations mutuelles. Les puissances de $\cos(S)$ sont nécessaires pour le développement en polynômes de Legendre (voir en 3.4.1) de cette fonction perturbatrice, tandis que celles de $2\,\Delta_{\cos} = 2(\cos(S) - \cos(w - w'))$ servent dans son développement en coefficients de Laplace (voir en 3.4.4) . On effectue le calcul de ces puissances à partir de *COSS* et de 2**DELTACOSS* obtenus précédemment.

Si ce n'est pas déjà fait, commencer par exécuter la phase d' initialisations , puis

```
> read cat(chemin,‘cosS.m‘);
```

On commence par remplacer les exponentielles complexes de la différence des londitudes moyennes par une variable σ:

```
> COSSsig:=subs(exp(I*L[i]-I*L[j])=sigma,exp(-I*L[i]+I*L[j])=1/sigma,CO
> SS):
> DDELTACOSSsig:=subs(exp(I*L[i]-I*L[j])=sigma,exp(-I*L[i]+I*L[j])=1/si
> gma,DDELTACOSS):
```

On veut mettre en facteur les monômes des variables Y et Yb afin de pouvoir facilement extraire le coefficient de chacun de ces monômes.

```
> var:=[Y[i],Y[j],Yb[i],Yb[j]]:  degmax:=10;
```

$$degmax := 10$$

On initialise la puissance 1 pour l'utiliser ensuite dans une boucle de calcul des puissances successives. On remarque que les 2 derniers termes de ce développement sont de degré 0 en inclinaisons. Pour limiter le nombre de termes calculés, on tronque le résultat au degré *degmax* par la procédure *PRODPOL* . Ces développements sont assez "creux" (beaucoup de monômes possibles n'y sont pas), ce qui explique le nombre limité de termes à chaque puissance. Notons que les variables C qui restent dans ces développements sont des fonctions non développées de YYb dont la valeur principale vaut 1 ($C = \sqrt{1 - YYb}$).

```
> COSSY1:=collect(COSSsig,var,distributed);
```

$$COSSY1 := -\frac{1}{2}\frac{\theta_i\,\theta_j\,{Y_i}^2\,Y_j\,Yb_j}{\sigma} + \frac{1}{2}\frac{\theta_i\,{Y_i}^2\,{Yb_j}^2}{\theta_j\,\sigma} + \frac{1}{2}\frac{\theta_i\,\theta_j\,{Y_i}^2}{\sigma} - \frac{1}{2}\theta_i\,\theta_j\,\sigma\,Y_i\,Yb_i\,{Y_j}^2 - \theta_i\,\theta_j\,C_i\,C_j\,Y_i\,Y_j$$
$$+ (\frac{1}{2}\frac{\theta_j}{\theta_i\,\sigma} + \frac{1}{2}\frac{\theta_i\,\sigma}{\theta_j})\,Yb_j\,Yb_i\,Y_i\,Y_j - \frac{1}{2}\frac{Y_i\,Yb_i\,{Yb_j}^2}{\theta_i\,\theta_j\,\sigma} + (-\frac{1}{2}\frac{\theta_j}{\theta_i\,\sigma} - \frac{1}{2}\frac{\theta_i\,\sigma}{\theta_j})\,Y_i\,Yb_i + \frac{\theta_i\,C_i\,C_j\,Y_i\,Yb_j}{\theta_j} + \frac{1}{2}\frac{\theta_j\,\sigma\,{Yb_i}^2\,{Y_j}^2}{\theta_i}$$
$$+ \frac{1}{2}\theta_i\,\theta_j\,\sigma\,{Y_j}^2 - \frac{1}{2}\frac{\sigma\,{Yb_i}^2\,Y_j\,Yb_j}{\theta_i\,\theta_j} + \frac{\theta_j\,C_i\,C_j\,Yb_i\,Y_j}{\theta_i} + (-\frac{1}{2}\frac{\theta_j}{\theta_i\,\sigma} - \frac{1}{2}\frac{\theta_i\,\sigma}{\theta_j})\,Yb_j\,Y_j + \frac{1}{2}\frac{\sigma\,{Yb_i}^2}{\theta_i\,\theta_j} - \frac{C_i\,C_j\,Yb_i\,Yb_j}{\theta_i\,\theta_j}$$
$$+ \frac{1}{2}\frac{{Yb_j}^2}{\theta_i\,\theta_j\,\sigma} + \frac{1}{2}\frac{\theta_j}{\theta_i\,\sigma} + \frac{1}{2}\frac{\theta_i\,\sigma}{\theta_j}$$

```
> for k  from 2 to degmax do:  deb:=time():
> COSSY.k:=PRODPOL(COSSY1,COSSY.(k-1),degmax,var):
> print(k,nops(%),time()-deb); od:
```

$$2,\ 107,\ .051$$
$$3,\ 326,\ 2.559$$
$$4,\ 509,\ 12.971$$
$$5,\ 586,\ 28.980$$
$$6,\ 587,\ 43.150$$
$$7,\ 588,\ 55.939$$
$$8,\ 589,\ 71.090$$
$$9,\ 590,\ 83.260$$
$$10,\ 591,\ 96.400$$

A la place de ce calcul des puissances successives, on peut aussi effectuer le développement de $(\cos(S))^m$ pour m quelconque, c'est-à-dire calculer une puissance quelconque de *COSS*, à un degré maxi donné (c'est finalement beaucoup plus rapide aussi) :

46 termes au degré 4, 130 termes au degré 6, 295 termes au degré 8, 581 termes au degré 10

```
> deb:=time(): COSSYm:=mtaylor(COSSsig^m,var,degmax+1):
> nops(%);time()-deb;
```

$$581$$
$$4.681$$

Il reste à sustituer une valeur numérique à *m* et à simplifier l'expression pour obtenir une forme équivalente aux résultats obtenus avec PRODPOL

```
> for k from 1 to degmax do: deb:=time():
> COSSYm.k:=sort(collect(expand(simplify(subs(m=k,COSSYm))),var,distribu
> ted),var): print(k,nops(%),time()-deb);od:
```

$$1,\ 19,\ .270$$
$$2,\ 107,\ .140$$
$$3,\ 326,\ .679$$
$$4,\ 509,\ 1.290$$
$$5,\ 586,\ 4.510$$
$$6,\ 587,\ 12.261$$
$$7,\ 588,\ 5.379$$
$$8,\ 589,\ 5.490$$
$$9,\ 590,\ 9.221$$
$$10,\ 591,\ 8.829$$

vérification:

```
> COSSYm10-COSSY10;
```

$$0$$

```
> save
> COSSYm1,COSSYm2,COSSYm3,COSSYm4,COSSYm5,COSSYm6,COSSYm7,COSSYm8,COSSYm
> 9,COSSYm10, cat(chemin,`COSSYm1a`.degmax.`.m`) :
> save COSSYm, cat(chemin,`COSSYpuissm.m`) :
```

Pour 2*$DELTACOSS$, qui est de degré 2 au moins en inclinaisons, il suffit d'atteindre la puissance *degmax*/2 pour obtenir les puissances de cette expression jusqu'au degré *degmax* en inclinaisons.

```
> DDELTACOSSY1:=collect(DDELTACOSSsig,var,distributed);
```

$$DDELTACOSSY1 := -\frac{\theta_i\,\theta_j\,{Y_i}^2\,Y_j\,Yb_j}{\sigma} + \frac{\theta_i\,{Y_i}^2\,{Yb_j}^2}{\theta_j\,\sigma} + \frac{\theta_i\,\theta_j\,{Y_i}^2}{\sigma} - \theta_i\,\theta_j\,\sigma\,Y_i\,{Y_j}^2\,Yb_i - 2\,\theta_i\,\theta_j\,C_i\,C_j\,Y_i\,Y_j$$
$$+ \left(\frac{\theta_j}{\theta_i\,\sigma} + \frac{\theta_i\,\sigma}{\theta_j}\right) Yb_i\,Yb_j\,Y_j\,Y_i - \frac{Y_i\,Yb_i\,{Yb_j}^2}{\theta_i\,\theta_j\,\sigma} + \left(-\frac{\theta_i\,\sigma}{\theta_j} - \frac{\theta_j}{\theta_i\,\sigma}\right) Y_i\,Yb_i + 2\,\frac{\theta_i\,C_i\,C_j\,Y_i\,Yb_j}{\theta_j} + \frac{\theta_j\,\sigma\,{Y_j}^2\,{Yb_i}^2}{\theta_i} + \theta_i\,\theta_j\,\sigma\,{Y_j}^2$$
$$- \frac{\sigma\,Y_j\,{Yb_i}^2\,Yb_j}{\theta_i\,\theta_j} + 2\,\frac{\theta_j\,C_i\,C_j\,Y_j\,Yb_i}{\theta_i} + \left(-\frac{\theta_i\,\sigma}{\theta_j} - \frac{\theta_j}{\theta_i\,\sigma}\right) Yb_j\,Y_j + \frac{\sigma\,{Yb_i}^2}{\theta_i\,\theta_j} - 2\,\frac{C_i\,C_j\,Yb_i\,Yb_j}{\theta_i\,\theta_j} + \frac{{Yb_j}^2}{\theta_i\,\theta_j\,\sigma}$$

```
> for k  from 2 to degmax/2 do:  deb:=time():
> DDELTACOSSY.k:=PRODPOL(DDELTACOSSY1,DDELTACOSSY.(k-1),degmax,var):
> print(k,nops(%),time()-deb); od:
```

$$2,\ 94,\ .020$$
$$3,\ 277,\ 1.639$$
$$4,\ 375,\ 5.980$$
$$5,\ 286,\ 5.980$$

L'autre méthode, avec m quelconque, nécessite d'ajouter un terme a de degré 0 (sinon divisions par zéro dans *mtaylor*), puis d'y faire $a = 0$. De ce fait, le calcul est moins performant; il faut même éviter de simplifier l'expression pour m quelconque (non numérique):

```
> deb:=time():  DDELTACOSSYm:=mtaylor((a+DDELTACOSSsig)^m,var,degmax+1):
> nops(%); time()-deb;
```

$$581$$
$$4.631$$

```
> for k  from 1 to degmax/2 do:  deb:=time():
> s:=simplify(subs(m=k,DDELTACOSSYm)):
> DDELTACOSSYm.k:=sort(collect(expand(subs(a=0,s)),var,distributed),var
> ):
> print(k,time()-deb,nops(%));
> od:
```

$$1,\ .280,\ 17$$
$$2,\ .169,\ 94$$
$$3,\ .651,\ 277$$
$$4,\ 1.339,\ 375$$
$$5,\ 3.391,\ 286$$

vérification:

```
> DDELTACOSSYm5-DDELTACOSSY5;
```

$$0$$

```
> save
> DDELTACOSSYm1,DDELTACOSSYm2,DDELTACOSSYm3,DDELTACOSSYm4,DDELTACOSSYm5,
> cat(chemin,`DELTACOSSYm1a`.(degmax/2).`.m` ):
> save  DDELTACOSSYm, cat(chemin,`DELTACOSSYpuissm.m`) :
```

Exemple de résultat (polynômes des variables Y_i , Yb_i , Y_j et Yb_j , à coefficients fonctions des variables σ , θ_i et θ_j):

```
> PRINTRONC(DDELTACOSSYm1,2,var);
```

$$-\frac{\theta_i\,\theta_j\,{Y_i}^2\,Y_j\;Yb_j}{\sigma},\ +\ldots+,\ \frac{\theta_i\,\theta_j\,{Y_i}^2}{\sigma}-2\,\theta_i\,\theta_j\,C_i\,C_j\,Y_i\,Y_j+(-\frac{\theta_i\,\sigma}{\theta_j}-\frac{\theta_j}{\theta_i\,\sigma})\,Y_i\;Yb_i+2\,\frac{\theta_i\,C_i\,C_j\,Y_i\;Yb_j}{\theta_j}+\theta_i\,\theta_j\,\sigma\,{Y_j}^2$$
$$+2\,\frac{\theta_j\,C_i\,C_j\,Y_j\;Yb_i}{\theta_i}+(-\frac{\theta_i\,\sigma}{\theta_j}-\frac{\theta_j}{\theta_i\,\sigma})\,Y_j\;Yb_j+\frac{\sigma\;{Yb_i}^2}{\theta_i\,\theta_j}-2\,\frac{C_i\,C_j\;Yb_i\;Yb_j}{\theta_i\,\theta_j}+\frac{{Yb_j}^2}{\theta_i\,\theta_j\,\sigma}$$

2.4. Développement des coordonnées sphériques équatoriales d'un satellite

Soit un satellite S orbitant une planète suivant un mouvement osculateur képlérien.

On calcule ici les 2 coordonnées sphériques δ et β (latitude et longitude) du satellite S dans le repère équatorial tournant $Oxyz$ lié à la planète (Oxy est son plan équatorial) , de façon qu'elles soient exprimées en fonction des éléments orbitaux (Ω_i, i_i, L_i, $v_i - M_i$) du satellite relatifs à un repère extérieur $Ox_0\, y_0\, z_0$, et en fonction des variables (angles d'Euler, ou variables d'Andoyer) représentant l'orientation de $Oxyz$ par rapport au même repère extérieur. Les variables associées au satellite seront affectées d'un indice i tandis que celles qui caractérisent notamment l'équateur de la planète dépendront d'un indice e.

2.4.1. Calcul en fonction des angles d'Euler de la planète

La planète est repérée par les angles d'Euler classiques (Ω_e, I_e, Φ) du repère équatorial $Oxyz$ mesurés par rapport au repère extérieur $Ox_0\, y_0\, z_0$ (à ces angles correspondent les repères intermédiaires $Ou\, v\, z_0$ et $Ouwz$ où Ou est le nœud de l'équateur sur le plan $Ox_0\, y_0$. Pour mesurer la rotation propre de la planète autour de son axe Oz par rapport au repère extérieur, on introduit l'angle L_e tel que $L_e = \Omega_e + \Phi$ d'où l'on tire $\Phi = L_e - \Omega_e$.

Si ce n'est pas déjà fait, commencer par exécuter la phase d' initialisations , puis :

```
>  with(linalg,vector,matrix,multiply,dotprod):
```

et définition des matrices de rotation autour des premier et troisième axes :

```
>  R1:=proc(ang) local u;
>  u:=matrix(3,3,[1,0,0,0,cos(ang),sin(ang),0,-sin(ang),cos(ang)]);
>  end:
```

```
> R3:=proc(ang) local u;
> u:=matrix(3,3,[cos(ang),sin(ang),0,-sin(ang),cos(ang),0,0,0,1]);
> end:
> i:='i': e:='e':
```

On part des coordonnées du satellite S dans le repère $Ox'y'z'$ où $Ox'y'$ est le plan orbital et où Ox' est suivant le nœud de l'orbite dans le plan $Ox_0\,y_0$. Interviennent donc w_i la longitude vraie de S dans l'orbite et Ω_i sa longitude du nœud :

```
> OS:=vector([cos(w[i]-Omega[i]),sin(w[i]-Omega[i]),0]);
```

$$OS := [\cos(w_i - \Omega_i),\, \sin(w_i - \Omega_i),\, 0]$$

On ramène alors *OS* dans le repère équatorial $Oxyz$ en effectuant les 4 rotations successives suivantes :

```
> OS:=multiply(R1(-i[i]),OS): OS:=multiply(R3(-Omega[i]+Omega[e]),OS):
> OS:=multiply(R1(I[e]),OS); OS:=multiply(R3(L[e]-Omega[e]),OS);
```

$$OS := \Big[\cos(\%2)\cos(\%1) - \sin(\%2)\cos(i_i)\sin(\%1),$$
$$\cos(I_e)\,(\sin(\%2)\cos(\%1) + \cos(\%2)\cos(i_i)\sin(\%1)) + \sin(I_e)\sin(i_i)\sin(\%1),$$
$$-\sin(I_e)\,(\sin(\%2)\cos(\%1) + \cos(\%2)\cos(i_i)\sin(\%1)) + \cos(I_e)\sin(i_i)\sin(\%1)\Big]$$
$$\%1 := w_i - \Omega_i$$
$$\%2 := \Omega_i - \Omega_e$$

$$OS := \left[\cos(\%4)\,(\cos(\%2)\cos(\%1) - \sin(\%2)\cos(i_i)\sin(\%1)) - \sin(\%4)\,(\cos(I_e)\,\%3 + \sin(I_e)\sin(i_i)\sin(\%1)),\right.$$
$$\sin(\%4)\,(\cos(\%2)\cos(\%1) - \sin(\%2)\cos(i_i)\sin(\%1)) + \cos(\%4)\,(\cos(I_e)\,\%3 + \sin(I_e)\sin(i_i)\sin(\%1)),$$
$$\left.-\sin(I_e)\,\%3 + \cos(I_e)\sin(i_i)\sin(\%1)\right]$$
$$\%1 := w_i - \Omega_i$$
$$\%2 := \Omega_i - \Omega_e$$
$$\%3 := \sin(\%2)\cos(\%1) + \cos(\%2)\cos(i_i)\sin(\%1)$$
$$\%4 := -L_e + \Omega_e$$

On isole ensuite les sinus et cosinus des inclinaisons pour que ces dernières ne soient pas mélangées aux longitudes lorsqu'on fera des combinaisons d'angles. On introduit pour cela les sinus et cosinus des demi-inclinaisons (notés γ et C respectivement).

Les 3 composantes de *OS* dans le repère $Oxyz$ sont alors égales à $\cos(\delta)\cos(\beta)$, $\cos(\delta)\sin(\beta)$, $\sin(\delta)$:

```
>  for k to 3 do :
>  subs(cos(i[i])=1-2*gamma[i]^2,cos(I[e])=1-2*gamma[e]^2,sin(i[i])=2*gam
>  ma[i]*C[i],sin(I[e])=2*gamma[e]*C[e],expand(OS[k])):
>  combine(%,trig):  if k=1 then OS[k]:= collect(%,cos) else OS[k]:=
>  collect(%,sin) fi:  od:
>  `cos(delta)*cos(beta)`:=OS[1]; `cos(delta)*sin(beta)`:=OS[2];
>  `sin(delta)`:=OS[3];
```

$$cos(delta) * cos(beta) := 2\,\gamma_e\,C_e\,\gamma_i\,C_i\cos(-L_e + \Omega_e + w_i - \Omega_i) - 2\,\gamma_e\,C_e\,\gamma_i\,C_i\cos(L_e - \Omega_e + w_i - \Omega_i)$$
$$+ ({\gamma_e}^2{\gamma_i}^2 - {\gamma_i}^2 - {\gamma_e}^2 + 1)\cos(-L_e + w_i) + ({\gamma_i}^2 - {\gamma_e}^2{\gamma_i}^2)\cos(-2\,\Omega_i + L_e + w_i) + (-{\gamma_e}^2{\gamma_i}^2 + {\gamma_e}^2)\cos(-2\,\Omega_e + L_e + w_i)$$
$$+ {\gamma_e}^2{\gamma_i}^2\cos(-L_e + w_i + 2\,\Omega_e - 2\,\Omega_i)$$

$$cos(delta)*sin(beta) := 2\,\gamma_e\,C_e\,\gamma_i\,C_i\sin(L_e-\Omega_e+w_i-\Omega_i)+2\,\gamma_e\,C_e\,\gamma_i\,C_i\sin(-L_e+\Omega_e+w_i-\Omega_i)$$
$$+({\gamma_e}^2\,{\gamma_i}^2-{\gamma_i}^2-{\gamma_e}^2+1)\sin(-L_e+w_i)+(-{\gamma_i}^2+{\gamma_e}^2\,{\gamma_i}^2)\sin(-2\,\Omega_i+L_e+w_i)+{\gamma_e}^2\,{\gamma_i}^2\sin(-L_e+w_i+2\,\Omega_e-2\,\Omega_i)$$
$$+(-{\gamma_e}^2+{\gamma_e}^2\,{\gamma_i}^2)\sin(-2\,\Omega_e+L_e+w_i)$$

$$sin(delta) := (-4\,\gamma_i\,C_i\,{\gamma_e}^2+2\,\gamma_i\,C_i)\sin(w_i-\Omega_i)+(-2\,\gamma_e\,C_e+2\,\gamma_e\,C_e\,{\gamma_i}^2)\sin(-\Omega_e+w_i)+2\,\gamma_e\,C_e\,{\gamma_i}^2\sin(-2\,\Omega_i+\Omega_e+w_i)$$

Vérification :

```
> simplify(expand(
> subs(C[e]=sqrt(1-gamma[e]^2),C[i]=sqrt(1-gamma[i]^2),
> `cos(delta)*cos(beta)`^2+`cos(delta)*sin(beta)`^2+`sin(delta)`^2)));
```

$$1$$

On transforme enfin ces composantes pour les exprimer en fonction des variables θ , Y , Yb et L du satellite, et de celles analogues définies pour la planète: $Y_e := \gamma_e\ \exp\left(I\ \Phi\right) = \gamma_e\ \exp\left(I\ (L_e-\Omega_e)\right)$ et son conjugué Yb_e.

Grace à la procédure suivante, on calcule d'abord le sinus de la latitude :

```
> ENYYBTTA:=proc(s,i) local ss:
> ss:=subs(gamma[i]=Y[i]*exp(I*Omega[i])/exp(I*L[i]),s):
> algsubs(exp(I*w[i])=theta[i]*exp(I*L[i]),%):
> algsubs(1/exp(I*w[i])=1/theta[i]/exp(I*L[i]),%):
> algsubs(exp(I*Omega[i])^2=Yb[i]/Y[i]*exp(I*L[i])^2,%):
> algsubs(1/exp(I*Omega[i])^2=Y[i]/Yb[i]/exp(I*L[i])^2,%):
> # subs(C[i]=sqrt(1-Y[i]*Yb[i]),%):
> end:
```

```
> co:=expand(convert(`sin(delta)`,exp)):
```

```
> sind:=collect(combine(co,exp),exp,factor);
> ENYYBTTA(co,i): ENYYBTTA(%,e):
> SIND:=collect(combine(expand(%),exp),exp,factor);
```

$$sind := -I\,\gamma_i\,C_i\,(-1+2\,\gamma_e{}^2)\,e^{(-I\,w_i+I\,\Omega_i)} + I\,\gamma_i\,C_i\,(-1+2\,\gamma_e{}^2)\,e^{(I\,w_i-I\,\Omega_i)} + I\,\gamma_e\,C_e\,\gamma_i{}^2\,e^{(2\,I\,\Omega_i-I\,\Omega_e-I\,w_i)} - I\,\gamma_e\,C_e\,\gamma_i{}^2\,e^{(-2\,I\,\Omega_i+I\,\Omega_e+I\,w_i)} + I\,(\gamma_i-1)\,(\gamma_i+1)\,C_e\,\gamma_e\,e^{(I\,\Omega_e-I\,w_i)} - I\,(\gamma_i-1)\,(\gamma_i+1)\,C_e\,\gamma_e\,e^{(-I\,\Omega_e+I\,w_i)}$$

$$SIND := -\frac{I\,Y_e\,C_e\,(\theta_i{}^2\,Y_i\,Yb_i-\theta_i{}^2-Yb_i{}^2)\,e^{(I\,L_i-I\,L_e)}}{\theta_i} - \frac{I\,C_e\,Yb_e\,(1+\theta_i{}^2\,Y_i{}^2-Y_i\,Yb_i)\,e^{(-I\,L_i+I\,L_e)}}{\theta_i} + \frac{I\,(-Yb_i+\theta_i{}^2\,Y_i)\,C_i\,(2\,Y_e\,Yb_e-1)}{\theta_i}$$

On calcule ensuite les deux autres coordonnées, rassemblées en un nombre complexe : $\cos(\delta)\ \exp(I\,\beta)$; c'est une variable complexe représentant les coordonnées équatoriales de S dans le repère équatorial lié à la planète .

On exprime enfin cette variable complexe et son conjugué dans les mêmes variables que $sin(\delta)$:

```
> co:=expand(convert(
> 'cos(delta)*cos(beta)'+I*'cos(delta)*sin(beta)',exp)):
> cosdexpb:=collect(combine(co,exp),exp,factor):
> ENYYBTTA(co,i): ENYYBTTA(%,e):
> COSDEXPB:=collect(combine(expand(%),exp),exp,factor);
```

```
> co:=expand(convert(
> 'cos(delta)*cos(beta)'-I*'cos(delta)*sin(beta)',exp)):
> cosdexpbb:=collect(combine(co,exp),exp,factor):
> ENYYBTTA(co,i): ENYYBTTA(%,e):
> COSDEXPBB:=collect(combine(expand(%),exp),exp,factor);
```

$$COSDEXPB := \frac{({\theta_i}^2\,Y_i\,Yb_i - {\theta_i}^2 - {Yb_i}^2)\,(Y_e\,Yb_e - 1)\,e^{(I\,L_i - I\,L_e)}}{\theta_i} + \frac{{Yb_e}^2\,(1 + {\theta_i}^2\,{Y_i}^2 - Y_i\,Yb_i)\,e^{(-I\,L_i + I\,L_e)}}{\theta_i} + 2\,\frac{C_e\,C_i\,Yb_e\,(-Yb_i + {\theta_i}^2\,Y_i)}{\theta_i}$$

$$COSDEXPBB := -\frac{{Y_e}^2\,({\theta_i}^2\,Y_i\,Yb_i - {\theta_i}^2 - {Yb_i}^2)\,e^{(I\,L_i - I\,L_e)}}{\theta_i} - \frac{(1 + {\theta_i}^2\,{Y_i}^2 - Y_i\,Yb_i)\,(Y_e\,Yb_e - 1)\,e^{(-I\,L_i + I\,L_e)}}{\theta_i} - 2\,\frac{C_e\,C_i\,Y_e\,(-Yb_i + {\theta_i}^2\,Y_i)}{\theta_i}$$

Vérification: on a bien $\cos^2\delta + \sin^2\delta = 1$.

```
> simplify(expand(
> subs(C[e]=sqrt(1-Y[e]*Yb[e]),C[i]=sqrt(1-Y[i]*Yb[i]),
> COSDEXPB*COSDEXPBB+SIND^2)));
```

$$1$$

```
> simplify(expand(
> subs(C[e]=sqrt(1-gamma[e]^2),C[i]=sqrt(1-gamma[i]^2),
> cosdexpb*cosdexpbb+sind^2)));
```

$$1$$

Sauvegarde des résultats :

```
> save `sin(delta)`,sind,cosdexpb,cosdexpbb,SIND,COSDEXPB,COSDEXPBB,
> cat(chemin,`sincosd.m`);
```

Pour obtenir un développement complètement exprimé en fonction des excentricités et inclinaisons, il faut y substituer θ_i par son développement en fonction des variables *X*, *Xb* , mais il faut prendre des précautions pour ne pas engendrer les nombreux termes de degré trop élevé qui ne manqueraient pas d'arriver dans leurs produits. Pour obtenir une expression développée à un degré maximum donné, on commence par éliminer les fonctions θ_i des dénominateurs (en donnant à leur inverse le nom *Tbi*).

```
> read cat(chemin,‘sincosd.m‘);
> read cat(chemin,‘kepler20.m‘):
> Dmax:=10;
```

$$Dmax := 10$$

```
> THETA10:=TRONC(THETA,Dmax,[X,Xb]):
> THETAB10:=TRONC(THETAB,Dmax,[X,Xb]):
> THETAi:=subs(X=X[i],Xb=Xb[i],THETA10):
> THETABi:=subs(X=X[i],Xb=Xb[i],THETAB10):
> varXYij:=[X[i],Xb[i],Y[i],Yb[i],X[j],Xb[j],Y[j],Yb[j]]:
> varXij:=[X[i],Xb[i],X[j],Xb[j]]:
> varXYie:=[X[i],Xb[i],Y[i],Yb[i],Y[e],Yb[e]]:

> CONVTHETA:=proc(p,i,j) local s:
> s:=collect(combine(expand(p),exp),exp):
> s:=algsubs(1/theta[i]='Tbi',s):
> s:=algsubs(1/theta[j]='Tbj',s):
> s:=subs(theta[i]='Ti',theta[j]='Tj',s):
> s:=subs(exp(I*L[i]-I*L[j])=sigma,exp(-I*L[i]+I*L[j])=1/sigma,s):
> expand(sort(collect(s,sigma),varXYije));
> end:
> SINDi:=CONVTHETA(SIND,i,e):  nops(%):

> COSDEXPBi:=CONVTHETA(COSDEXPB,i,e):  nops(%):

> COSDEXPBBi:=CONVTHETA(COSDEXPBB,i,e):  nops(%):
```

Ces formulations sont ainsi utilisables par la procédure SUBS qui substitue des séries entières à des variables, en ne conservant que les termes de degré inférieur à un degré donné (ici égal à 10):

```
> SINDXYie10:= SUBS([Ti=THETAi,Tbi=THETABi],SINDi,Dmax,varXYie):
> nops(%); TRONC(%%,2,varXYie);
```

$$402$$

$$-I\,C_i\,X_i\,Y_i - I\,C_i\,X_i\,Yb_i + I\,\sigma\,C_e\,X_i\,Y_e + \frac{I\,C_e\,X_i\,Yb_e}{\sigma} + I\,C_i\,Xb_i\,Y_i + I\,C_i\,Xb_i\,Yb_i - I\,\sigma\,C_e\,Xb_i\,Y_e - \frac{I\,C_e\,Xb_i\,Yb_e}{\sigma} - I\,C_i\,Y_i + I\,C_i\,Yb_i + I\,\sigma\,C_e\,Y_e - \frac{I\,C_e\,Yb_e}{\sigma}$$

```
> COSDEXPBXYie10:=
> SUBS([Ti=THETAi,Tbi=THETABi],COSDEXPBi,Dmax,varXYie):  nops(%);
> TRONC(%%,2,varXYie);
```

$$418$$

$$\frac{9}{8}\,\sigma\,{X_i}^2 - \sigma\,X_i\,Xb_i - \frac{1}{8}\,\sigma\,{Xb_i}^2 - \sigma\,Y_i\,Yb_i + 2\,C_e\,C_i\,Y_i\,Yb_e + \sigma\,{Yb_i}^2 - 2\,C_e\,C_i\,Yb_i\,Yb_e - \sigma\,Y_e\,Yb_e + \frac{{Yb_e}^2}{\sigma} + \sigma\,X_i - \sigma\,Xb_i + \sigma$$

```
> COSDEXPBBXYie10:=
> SUBS([Ti=THETAi,Tbi=THETABi],COSDEXPBBi,Dmax,varXYie):  nops(%);
> TRONC(%%,2,varXYie);
```

$$418$$

$$-\frac{1}{8}\,\frac{{X_i}^2}{\sigma} - \frac{X_i\,Xb_i}{\sigma} + \frac{9}{8}\,\frac{{Xb_i}^2}{\sigma} + \frac{{Y_i}^2}{\sigma} - \frac{Y_i\,Yb_i}{\sigma} - 2\,C_e\,C_i\,Y_i\,Y_e + 2\,C_e\,C_i\,Yb_i\,Y_e + \sigma\,{Y_e}^2 - \frac{Y_e\,Yb_e}{\sigma} - \frac{X_i}{\sigma} + \frac{Xb_i}{\sigma} + \frac{1}{\sigma}$$

```
> save SINDXYie10, COSDEXPBXYie10, COSDEXPBBXYie10,
> cat(chemin,`SINDCOSDXYie10.m`);
```

On transforme enfin ces fonctions pour leur usage dans le développement, par inégalités, des fonctions harmoniques sphériques en 4.3.2 (on y remplace simplement $\exp\left(I\left(L_i - L_e\right)\right)$ par la variable σ):

```
> SINDsig:=collect(subs(exp(I*L[i]-I*L[e])=sigma,exp(-I*L[i]+I*L[e])=1/
> sigma,SIND),varYie,distributed);
```

$$SINDsig := -\frac{I\,Y_e\,C_e\,({\theta_i}^2\,Y_i\,Yb_i - {\theta_i}^2 - {Yb_i}^2)\,\sigma}{\theta_i} - \frac{I\,C_e\,Yb_e\,(1 + {\theta_i}^2\,{Y_i}^2 - Y_i\,Yb_i)}{\theta_i\,\sigma} + \frac{I\,(-Yb_i + {\theta_i}^2\,Y_i)\,C_i\,(2\,Yb_e\,Y_e - 1)}{\theta_i}$$

```
> COSDEXPBsig:=collect(subs(exp(I*L[i]-I*L[e])=sigma,exp(-I*L[i]+I*L[e]
> )=1/sigma,COSDEXPB),varYie,distributed);
```

$$COSDEXPBsig := \frac{({\theta_i}^2\,Y_i\,Yb_i - {\theta_i}^2 - {Yb_i}^2)\,(Yb_e\,Y_e - 1)\,\sigma}{\theta_i} + \frac{{Yb_e}^2\,(1 + {\theta_i}^2\,{Y_i}^2 - Y_i\,Yb_i)}{\theta_i\,\sigma} + 2\,\frac{C_e\,C_i\,Yb_e\,(-Yb_i + {\theta_i}^2\,Y_i)}{\theta_i}$$

```
> COSDEXPBBsig:=collect(subs(exp(I*L[i]-I*L[e])=sigma,exp(-I*L[i]+I*L[e
> ])=1/sigma,COSDEXPBB),varYie,distributed);
```

$$COSDEXPBBsig := -\frac{{Y_e}^2\,({\theta_i}^2\,Y_i\,Yb_i - {\theta_i}^2 - {Yb_i}^2)\,\sigma}{\theta_i} - \frac{(1 + {\theta_i}^2\,{Y_i}^2 - Y_i\,Yb_i)\,(Yb_e\,Y_e - 1)}{\theta_i\,\sigma} - 2\,\frac{C_e\,C_i\,Y_e\,(-Yb_i + {\theta_i}^2\,Y_i)}{\theta_i}$$

```
> save SINDsig,COSDEXPBsig, COSDEXPBBsig,
> cat(chemin,`SINDCOSDEXPBsig.m`):
```

3. Développements de fonctions perturbatrices de planètes

3.1. Introduction

On se propose de construire le développement analytique de la fonction perturbatrice d'une planète d'un système planétaire. On suppose cependant que l'on recherche ces développements en fonction des éléments osculateurs des orbites de ces planètes. On utilise pour cela les développements du mouvement képlérien et ceux en inclinaisons introduits dans la première partie.

Dans le cas du mouvement héliocentrique des planètes, la fonction perturbatrice de la planète P_i par la planète P_j de masse m_j est de la forme :

$$K\, m_j \, (\frac{1}{\Delta} - \frac{r_i \cos(S)}{{r_j}^2})$$

où Δ est la distance mutuelle des 2 planètes et S l'angle entre leurs deux rayons vecteurs r_i et r_j . La fonction $\frac{1}{\Delta}$ contribue à la partie directe de la perturbation et l'autre fonction à la partie indirecte. Dans ce qui suit, on propose de développer chacune de ces deux parties en fonction des éléments orbitaux des deux planètes. C25.1

3.2. Initialisations

On relit les développements en X, Xb, Y et Yb construits dans la première partie, et on les prépare à pouvoir être utilisés pour deux corps d'indices i et j . On demande ci dessous un degré maxi égal à 10 pour pouvoir utiliser les développements construits à ce degré dans les sections précédentes:

Si ce n'est pas déjà fait, commencer par exécuter la phase d' initialisations générales, puis :

```
> read cat(chemin,'cosS.m'):
> read cat(chemin,'sincosd.m'):
> read cat(chemin,'kepler20.m'):
> read cat(chemin,'COSSYm1a10.m'):
> read cat(chemin,'DELTACOSSYm1a5.m'):
> read cat(chemin,'Xnm10.m');
```

```
> read cat(chemin,`COSSXij10.m`);
> Dmax:=10;
```

$$Dmax := 10$$

```
> ASR:=TRONC(ASR,Dmax,[X,Xb]):  PRINTRONC(ASR,3,[X,Xb]);
> RSA:=TRONC(RSA,Dmax,[X,Xb]):  PRINTRONC(RSA,3,[X,Xb]);
> THETA:=TRONC(THETA,Dmax,[X,Xb]):  PRINTRONC(THETA,3,[X,Xb]);
> THETAB:=TRONC(THETAB,Dmax,[X,Xb]):  PRINTRONC(THETAB,3,[X,Xb]);
> PRODPOL(THETA,THETAB,Dmax,[X,Xb]);
```

$$\frac{390625}{145152}\,X^{10},\ + \dots + ,\ \frac{9}{16}\,X^3 - \frac{1}{16}\,X^2\,Xb - \frac{1}{16}\,X\,Xb^2 + \frac{9}{16}\,Xb^3 + \frac{1}{2}\,X^2 + \frac{1}{2}\,Xb^2 + \frac{1}{2}\,X + \frac{1}{2}\,Xb + 1$$

$$-\frac{78125}{290304}\,X^{10},\ + \dots + ,\ -\frac{3}{16}\,X^3 + \frac{3}{16}\,X^2\,Xb + \frac{3}{16}\,X\,Xb^2 - \frac{3}{16}\,Xb^3 - \frac{1}{4}\,X^2 + \frac{1}{2}\,X\,Xb - \frac{1}{4}\,Xb^2 - \frac{1}{2}\,X - \frac{1}{2}\,Xb + 1$$

$$\frac{25937424601}{3715891200}\,X^{10},\ + \dots + ,\ \frac{4}{3}\,X^3 - \frac{5}{4}\,X^2\,Xb - \frac{1}{12}\,Xb^3 + \frac{9}{8}\,X^2 - X\,Xb - \frac{1}{8}\,Xb^2 + X - Xb + 1$$

$$-\frac{4782969}{45875200}\,X^{10},\ + \dots + ,\ -\frac{1}{12}\,X^3 - \frac{5}{4}\,X\,Xb^2 + \frac{4}{3}\,Xb^3 - \frac{1}{8}\,X^2 - X\,Xb + \frac{9}{8}\,Xb^2 - X + Xb + 1$$

$$1$$

```
> ASRj:=subs(X=X[j],Xb=Xb[j],ASR): ASRi:=subs(j=i,ASRj):
> RSAi:=subs(X=X[i],Xb=Xb[i],RSA): RSAj:=subs(i=j,RSAi):
> THETAi:=subs(X=X[i],Xb=Xb[i],THETA):
> THETAj:=subs(i=j,THETAi):
> THETABi:=subs(X=X[i],Xb=Xb[i],THETAB):
> THETABj:=subs(i=j,THETABi):
> varXYij:=[X[i],Xb[i],X[j],Xb[j],Y[i],Yb[i],Y[j],Yb[j]]:
> varXij:=[X[i],Xb[i],X[j],Xb[j]]:
> varxyij:=[x[i],xb[i],x[j],xb[j],y[i],yb[i],y[j],yb[j]]:
```

3.3. Développement de la partie indirecte planétaire classique (cas héliocentrique)

Dans un problème planétaire en variables héliocentriques, la partie indirecte de la fonction perturbatrice de la planète P_i perturbée par P_j, fait intervenir l'expression

$$\frac{r_i \cos(S)}{{r_j}^2}$$

. On la développe ici directement sous plusieurs formes, jusqu'à un degré maximum *degmx* donné en excentricités et inclinaisons : Tant que ce degré est faible, il est possible d'effectuer un calcul complet; sinon, on propose une méthode de calcul où une quelconque "inégalité" donnée peut être développée à un degré élevé. C25.1.3

3.3.1. Calcul complet (degrés faibles)

(Il faut avoir exécuté au préalable la partie " Initialisations ")

```
> degmx:=4;
```

$$degmx := 4$$

```
> ASR:=TRONC(ASR,degmx,[X,Xb]):  PRINTRONC(ASR,2,[X,Xb]);
> RSA:=TRONC(RSA,degmx,[X,Xb]):  PRINTRONC(RSA,2,[X,Xb]);
> THETA:=TRONC(THETA,degmx,[X,Xb]):  PRINTRONC(THETA,2,[X,Xb]);
> THETAB:=TRONC(THETAB,degmx,[X,Xb]):  PRINTRONC(THETAB,2,[X,Xb]);
```

$$\frac{2}{3}X^4,\ + ... + ,\ \frac{1}{2}X^2 + \frac{1}{2}Xb^2 + \frac{1}{2}X + \frac{1}{2}Xb + 1$$

$$-\frac{1}{6}X^4,\ + ... + ,\ -\frac{1}{4}X^2 + \frac{1}{2}X\,Xb - \frac{1}{4}Xb^2 - \frac{1}{2}X - \frac{1}{2}Xb + 1$$

$$\frac{625}{384} X^4, \ + \ldots + , \ \frac{9}{8} X^2 - X\,Xb - \frac{1}{8} Xb^2 + X - Xb + 1$$

$$-\frac{9}{128} X^4, \ + \ldots + , \ -\frac{1}{8} X^2 - X\,Xb + \frac{9}{8} Xb^2 - X + Xb + 1$$

```
> ASRj:=subs(X=X[j],Xb=Xb[j],ASR): ASRi:=subs(j=i,ASRj):
> RSAi:=subs(X=X[i],Xb=Xb[i],RSA): RSAj:=subs(i=j,RSAi):
> THETAi:=subs(X=X[i],Xb=Xb[i],THETA):
> THETAj:=subs(i=j,THETAi):
> THETABi:=subs(X=X[i],Xb=Xb[i],THETAB):
> THETABj:=subs(i=j,THETABi):
```

On obtient d'abord le développement de la partie indirecte au degré *degmx* en variables X, Xb, Y, Yb des 2 planètes et en fonction de la variable σ (introduite dans *COSSXij10*):

```
> Uind:=mtaylor(RSAi*ASRj^2*COSSXij10,varXYij,degmx+1):  nops(%);
```

$$202$$

```
> UindXYij:=collect(combine(expand(Uind),exp),exp):  nops(%);
```

$$262$$

```
> collect(TRONC(UindXYij,2,varXYij),sigma);
```

$$(\frac{1}{2} X_i\,Xb_j - \frac{1}{4} X_i\,Xb_i - \frac{1}{4} X_j\,Xb_j + \frac{1}{16} {X_j}^2 + \frac{1}{16} {Xb_i}^2 + \frac{1}{2} {Yb_i}^2 - \frac{1}{2} Y_i\,Yb_i + \frac{3}{16} {X_i}^2 + \frac{1}{2} {Y_j}^2 + \frac{1}{4} X_i - \frac{1}{2} Y_j\,Yb_j - \frac{3}{4} Xb_i + \frac{27}{16} {Xb_j}^2$$
$$+ Xb_j + \frac{1}{2} - \frac{3}{2} Xb_i\,Xb_j)\sigma + C_i\,C_j\,Y_i\,Yb_j - C_i\,C_j\,Y_i\,Y_j - C_i\,C_j\,Yb_i\,Yb_j + C_i\,C_j\,Yb_i\,Y_j + (\frac{1}{16} {X_i}^2 + \frac{1}{16} {Xb_j}^2 - \frac{1}{4} X_i\,Xb_i$$
$$+ \frac{1}{2} {Y_i}^2 - \frac{3}{2} X_i\,X_j + \frac{1}{2} + \frac{3}{16} {Xb_i}^2 - \frac{3}{4} X_i + \frac{1}{2} Xb_i\,X_j - \frac{1}{2} Y_i\,Yb_i + \frac{27}{16} {X_j}^2 + X_j - \frac{1}{4} X_j\,Xb_j - \frac{1}{2} Y_j\,Yb_j + \frac{1}{2} {Yb_j}^2 + \frac{1}{4} Xb_i)/\sigma$$

On peut transformer ce développement en l'exprimant en fonction des variables x, xb, y, yb définies par $X = xb\ \exp(I\,L)$, $Xb = x\ \exp(-I\,L)$ et $Y = yb\ \exp(I\,L)$, $Yb = y\ \exp(-I\,L)$:

```
>  Uindxyij:=collect(combine(expand( subs(sigma=exp(I*L[i])/exp(I*L[j]),
>  X[i]=xb[i]*exp(I*L[i]),Xb[i]=x[i]/exp(I*L[i]),
>  Y[i]=yb[i]*exp(I*L[i]),Yb[i]=y[i]/exp(I*L[i]),
>  X[j]=xb[j]*exp(I*L[j]),Xb[j]=x[j]/exp(I*L[j]),
>  Y[j]=yb[j]*exp(I*L[j]),Yb[j]=y[j]/exp(I*L[j]),UindXYij)),exp),exp):
>  nops(%);
```

$$50$$

```
>  collect(TRONC(Uindxyij,2,varxyij),exp);
```

$$\begin{aligned}&\frac{1}{4}\,xb_i\,e^{(2\,I\,L_i-I\,L_j)}+\frac{3}{16}\,{xb_i}^2\,e^{(3\,I\,L_i-I\,L_j)}+\frac{1}{2}\,e^{(2\,I\,L_i-2\,I\,L_j)}\,xb_i\,x_j-\frac{3}{2}\,e^{(-2\,I\,L_j)}\,x_i\,x_j+\frac{27}{16}\,{x_j}^2\,e^{(I\,L_i-3\,I\,L_j)}+e^{(I\,L_i-2\,I\,L_j)}\,x_j\\
&-\frac{3}{2}\,e^{(2\,I\,L_j)}\,xb_i\,xb_j+\frac{1}{2}\,e^{(-2\,I\,L_i+2\,I\,L_j)}\,x_i\,xb_j+\frac{27}{16}\,{xb_j}^2\,e^{(-I\,L_i+3\,I\,L_j)}+e^{(-I\,L_i+2\,I\,L_j)}\,xb_j+\frac{3}{16}\,{x_i}^2\,e^{(-3\,I\,L_i+I\,L_j)}\\
&+\frac{1}{4}\,x_i\,e^{(-2\,I\,L_i+I\,L_j)}-\frac{3}{4}\,e^{(I\,L_j)}\,xb_i+(\frac{1}{2}\,{y_j}^2+\frac{1}{16}\,{x_i}^2+\frac{1}{16}\,{x_j}^2-C_i\,C_j\,y_i\,y_j+\frac{1}{2}\,{y_i}^2)\,e^{(-I\,L_i-I\,L_j)}\\
&+(-C_i\,C_j\,yb_i\,yb_j+\frac{1}{16}\,{xb_i}^2+\frac{1}{2}\,{yb_j}^2+\frac{1}{2}\,{yb_i}^2+\frac{1}{16}\,{xb_j}^2)\,e^{(I\,L_i+I\,L_j)}-\frac{3}{4}\,e^{(-I\,L_j)}\,x_i\\
&+(-\frac{1}{2}\,yb_i\,y_i+C_i\,C_j\,yb_i\,y_j-\frac{1}{4}\,xb_j\,x_j+\frac{1}{2}-\frac{1}{4}\,xb_i\,x_i-\frac{1}{2}\,yb_j\,y_j)\,e^{(I\,L_i-I\,L_j)}\\
&+(\frac{1}{2}-\frac{1}{4}\,xb_i\,x_i-\frac{1}{2}\,yb_j\,y_j-\frac{1}{2}\,yb_i\,y_i+C_i\,C_j\,y_i\,yb_j-\frac{1}{4}\,xb_j\,x_j)\,e^{(-I\,L_i+I\,L_j)}\end{aligned}$$

Si l'on veut revenir à une forme explicite des excentricités et inclinaisons, on utilise: $x = e\ \exp(I\,\varpi)$ et $y = \gamma\,e^{(I\,\Omega)}$ (on notera que dans les instructions suivantes, ω représente la longitude du périhélie à la place de ϖ).

```
> Uindei:=combine(
> subs(x[i]=e[i]*exp(I*omega[i]),xb[i]=e[i]/exp(I*omega[i]),
> y[i]=gamma[i]*exp(I*Omega[i]),yb[i]=gamma[i]/exp(I*Omega[i]),
> x[j]=e[j]*exp(I*omega[j]),xb[j]=e[j]/exp(I*omega[j]),
> y[j]=gamma[j]*exp(I*Omega[j]),yb[j]=gamma[j]/exp(I*Omega[j]),
> expand(Uindxyij)),exp):  nops(%);
```

$$262$$

La forme suivante, trigonométrique, permet éventuellement la comparaison avec les développements qu'on trouve de manière classique dans les manuels (cf. Cours de Mécanique Céleste) C(6.69)

```
> varegammaij:=[e[i],e[j],gamma[i],gamma[j]]:
> TRONC(Uindei,4,varegammaij):
> Uindei:=collect(convert(%,trig),cos):  nops(%);
> collect( TRONC(Uindei,2,varegammaij),cos);
```

$$75$$

$$-2\,C_i\,C_j\,\gamma_i\,\gamma_j\cos(-L_i-L_j+\Omega_i+\Omega_j)+2\,C_i\,C_j\,\gamma_i\,\gamma_j\cos(-L_i+L_j+\Omega_i-\Omega_j)+(-\frac{1}{2}\,{e_j}^2-{\gamma_i}^2+1-\frac{1}{2}\,{e_i}^2-{\gamma_j}^2)\cos(-L_i+L_j)$$
$$-3\,e_i\,e_j\cos(2\,L_j-\omega_i-\omega_j)+\frac{3}{8}\,{e_i}^2\cos(-3\,L_i+L_j+2\,\omega_i)+\frac{1}{8}\cos(L_i+L_j-2\,\omega_i)\,{e_i}^2+{\gamma_i}^2\cos(-L_i-L_j+2\,\Omega_i)$$
$$+{\gamma_j}^2\cos(-L_i-L_j+2\,\Omega_j)-\frac{3}{2}\,e_i\cos(L_j-\omega_i)+2\cos(-L_i+2\,L_j-\omega_j)\,e_j+\frac{1}{2}\cos(-2\,L_i+L_j+\omega_i)\,e_i$$
$$+\cos(-2\,L_i+2\,L_j+\omega_i-\omega_j)\,e_i\,e_j+\frac{1}{8}\,{e_j}^2\cos(L_i+L_j-2\,\omega_j)+\frac{27}{8}\,{e_j}^2\cos(-L_i+3\,L_j-2\,\omega_j)$$

Pour rendre compatibles le développement de cette partie indirecte avec celui de l'inverse de la distance on procède aux transformations suivantes :

On substitue les variables x et y aux variables X et Y (on a $x = Xb\,\exp(I\,L)$...) , mais on y remplace $\exp(I\,L)$ par une variable eL, ceci pour permettre des manipulations de polynômes généralisés (polynômes dont certaines variables peuvent être à des degrés négatifs). On y remplace alors aussi $\sigma = \exp\left(I\left(L_i - L_j\right)\right)$ par $\frac{eL_i}{eL_j}$:

```
> Uindxyij1:=subs( X[i]=xb[i]*eL[i],Xb[i]=x[i]/eL[i],
> Y[i]=yb[i]*eL[i],Yb[i]=y[i]/eL[i], X[j]=xb[j]*eL[j],Xb[j]=x[j]/eL[j],
> Y[j]=yb[j]*eL[j],Yb[j]=y[j]/eL[j],expand(UindXYij)):  nops(%);
> Uindxyij2:=expand(algsubs(1/sigma=eL[j]/eL[i],Uindxyij1)):
> nops(%);
> Uindxyij:=expand(algsubs(sigma=eL[i]/eL[j],Uindxyij2)):nops(%);
> op(2,%%);
```

$$262$$

$$262$$

$$262$$

$$\frac{27}{16}\,\frac{{x_j}^2\,eL_i}{{eL_j}^3}$$

```
> save UindXYij,Uindxyij, cat(chemin,‘Uind‘.degmx.‘.m‘):
```

3.3.2. Calcul par inégalités

Quand deux ou plusieurs corps sont en interaction, une **inégalité** est par définition une combinaison linéaire entière de leurs longitudes moyennes, par exemple $3\,L_i - 4\,L_j$. Comme on le voit dans le développement de *Uindxy* du paragraphe précédent , les inégalités y interviennent par leur exponentielle complexe , sous la forme:

$$\sum_{\{p,\,q,\,n_{1..8}\}\in\{Z^2,\,N^8\}} C_{\{p,\,q,\,n_1,\,n_2,\,n_3,\,n_4,\,n_5,\,n_6,\,n_7,\,n_8\}}\,{x_i}^{n_1}\,{xb_i}^{n_2}\,{x_j}^{n_3}\,{xb_j}^{n_4}\,{y_i}^{n_5}\,{yb_i}^{n_6}\,{y_j}^{n_7}\,{yb_j}^{n_8}\,\exp\left(I\left(p\,L_i + q\,L_j\right)\right)$$

L'ensemble des monômes qui peuvent se mettre en facteur de l'exponentielle d'une même inégalité est appelé coefficient de cette inégalité. On peut en fait calculer directement ce coefficient, sans avoir à effectuer le développement complet. On utilise pour cela la propriété de d'Alembert qui permet de déterminer quels sont les monômes des variables $\{x_i,\ xb_i,\ x_j,\ xb_j,\ y_i,\ yb_i,\ y_j,\ yb_j\}$ qui intervinnent dans le coefficient d'une inégalité. On part pour cela de la façon de construire $\frac{r_i \cos(S)}{{r_j}^2}$ à partir des développements du mouvement képlérien et de $\cos(S)$, aboutissant à la forme générale des développements en excentricités et inclinaisons où, lorsqu'on utilise les variables X , Xb , Y et Yb , les longitudes moyennes n'interviennent que par leur différence (présente ci-dessous dans la variable σ):

$$\sum_{\{p,\,q,\,n_{1..8}\}\in\{Z^2,\,N^8\}} C_{\{p,\,q,\,n_1,\,n_2,\,n_3,\,n_4,\,n_5,\,n_6,\,n_7,\,n_8\}}\ {Xb_i}^{n_1}\ {X_i}^{n_2}\ {Xb_j}^{n_3}\ {X_j}^{n_4}\ {Yb_i}^{n_5}\ {Y_i}^{n_6}\ {Yb_j}^{n_7}\ {Y_j}^{n_8}\ \sigma^k$$

Le passage aux variables $\{x_i,\ xb_i,\ x_j,\ xb_j,\ y_i,\ yb_i,\ y_j,\ yb_j\}$ montre qu'on a alors les relations :

$$n_2 - n_1 + n_6 - n_5 + k = p \qquad \text{et} \qquad n_4 - n_3 + n_8 - n_7 - k = q$$

d'où l'on tire encore

$$n_2 - n_1 + n_4 - n_3 + n_6 - n_5 + n_8 - n_7 = p + q \qquad \text{avec} \qquad n_6 + n_5 + n_8 + n_7$$

Dans le développement de $\cos(S)$ on a seulement $k = -1$, $k = 0$ et $k = 1$, mais dans celui de $[\cos(S)]^n$ on aurait k compris entre $-n$ et $+n$.

La quantité $C_I = p + q$ est appelée **caractéristique de l'inégalité** , celle $C_M = n_2 - n_1 + n_4 - n_3 + n_6 - n_5 + n_8 - n_7$ est appelée **caractéristique du monôme**; la propriété de d'Alembert s'écrit donc encore : $C_I = C_M$. C(6.71)

On se sert de ces relations pour déterminer quels sont les monômes présents dans le coefficient d'une inégalité donnée. Le programme suivant : "initmo" , réalise la détermination des $\{n_i\}_{i=1..8}$ dont la combinaison précédente vaut $C_I = C_M$, et correspondant à des monômes de degré compris entre *idmin* et *idmax*: Anx2.2

```
> read cat(chemin,"initmo.pr"):

#
```

initmo:=proc(idmin,idmax,icara,parite,code)
#
Ce programme cree la liste de tous les monômes a deux planetes pour des degres
inferieurs ou egaux a idmax, ayant une parite en inclinaison donnee
et une caracteristique donnee (cf. [3]).
#
idmin : degre minimum en excentricite-inclinaison (in)
idmax : degre maximum en excentricite-inclinaison (in)
parite: parite en inclinaison (0 : pair, 1 : impair) (in)
icara : caracteristique des monômes (in)
tmon : liste des monômes (deg,expos des 8 variables) (out)
nbp : nombre d'elements de tmon (out)
sortie codée (si code=1) ou explicite (si code=0) ou les deux (si code=2)
#

Exemple, liste des monômes de degrés entre 0 et 3 d'une inégalité de caractéristique 2 paire en inclinaisons (par exemple $2L_i$ ou $2L_e$ ou $(p+2)\,L_i - p\,L_e$ quelque soit p) :

```
> initmo(0,3,2,0,0); nops(%);
```

$$[[1,\ 2,\ {yb_i}^2],\ [1,\ 2,\ {xb_i}^2],\ [3,\ 2,\ yb_j\ yb_i],\ [3,\ 2,\ xb_i\ xb_j],\ [2,\ 2,\ {yb_j}^2],\ [2,\ 2,\ {xb_j}^2]]$$

$$6$$

Les indices *i et j indiquent les variables des planètes intérieure et extérieure respectivement;*

le premier chiffre indique quels corps sont impliqués dans le monôme (1 ⇒ corps intérieur seul, 2 ⇒ corps extérieur seul et 3 ⇒ deux corps), le deuxième donne le degré du monôme. Avec le code 1, les monômes sont donnés par les exposants, **dans l'ordre** des variables $\{xb_j,\ x_j,\ xb_i,\ x_i,\ yb_j,\ y_j,\ yb_i,\ y_i\}$.

```
> initmo(0,3,2,0,1); nops(%);
```

$$[[1, 2, 0, 0, 0, 0, 0, 0, 2, 0], [1, 2, 0, 0, 2, 0, 0, 0, 0, 0], [3, 2, 0, 0, 0, 0, 1, 0, 1, 0], [3, 2, 1, 0, 1, 0, 0, 0, 0, 0], [2, 2, 0, 0, 0, 0, 2, 0, 0, 0], [2, 2, 2, 0, 0, 0, 0, 0, 0, 0]]$$

$$6$$

```
> initmo(0,3,2,0,2);
```

$$[[1, 2, 0, 0, 0, 0, 0, 0, 2, 0, {yb_i}^2], [1, 2, 0, 0, 2, 0, 0, 0, 0, 0, {xb_i}^2], [3, 2, 0, 0, 0, 0, 1, 0, 1, 0, yb_j\, yb_i], [3, 2, 1, 0, 1, 0, 0, 0, 0, 0, xb_i\, xb_j], [2, 2, 0, 0, 0, 0, 2, 0, 0, 0, {yb_j}^2], [2, 2, 2, 0, 0, 0, 0, 0, 0, 0, {xb_j}^2]]$$

Autre exemple : Les monômes présents dans l'inégalité séculaire (indépendante des longitudes moyennes) ou des inégalités de caractéristique nulle sont, jusqu'au degré 2 : 1 terme de degré 0 et 8 termes de degré 2 :

```
> initmo(0,3,0,0,2); nops(%);
```

$$[[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1], [1, 2, 0, 0, 0, 0, 0, 0, 1, 1, yb_i\, y_i], [1, 2, 0, 0, 1, 1, 0, 0, 0, 0, xb_i\, x_i], [3, 2, 0, 0, 0, 0, 0, 1, 1, 0, y_j\, yb_i], [3, 2, 0, 1, 1, 0, 0, 0, 0, 0, xb_i\, x_j], [3, 2, 0, 0, 0, 0, 1, 0, 0, 1, yb_j\, y_i], [3, 2, 1, 0, 0, 1, 0, 0, 0, 0, x_i\, xb_j], [2, 2, 0, 0, 0, 0, 1, 1, 0, 0, yb_j\, y_j], [2, 2, 1, 1, 0, 0, 0, 0, 0, 0, xb_j\, x_j]]$$

$$9$$

Dans une série, les diverses inégalités de même caractéristique correspondent chacune à une valeur précise de l'exposant k de la variable σ dans son développement en variables X, Xb, Y et Yb. Pour une inégalité $p\,L_i + q\,L_j$ de caractéristique $p+q$ donnée, on obtient des 8-uplets $\{n_1,\, n_2,\, n_3,\, n_4,\, n_5,\, n_6,\, n_7,\, n_8\}$ pour lesquels on a $k = p + n_2 - n_1 + n_6 - n_5$ ou $k = n_4 - n_3 + n_8 - n_7 - q$. Les inégalités trouvées à un degré donné dans le développement de $\cos(S)$ doivent satisfaire à $k = 0,\ 1$, ou -1.

L'expression de $\frac{r_i \cos(S)}{{r_j}^2}$ est, au facteur $\frac{{a_j}^2}{a_i}$ près, le produit de $\frac{r_i}{a_i}$, de $(\frac{a_j}{r_j})^2$ et de $\cos(S)$, lui-même dépendant des 4 variables $Y_i,\ Yb_i,\ Y_j,\ Yb_j$, de $\theta_i,\ \theta_j$ et de σ . Pour calculer le coefficient d'un monôme défini par son 8-uplet, il suffit donc de rechercher d'abord si en facteur des variables d'inclinaison dans l'expression de $\cos(S)$, la variable σ se trouve à la bonne puissance (valeur de k tirée de l'une des expressions précédentes pour p et q donnés); ensuite, avec les puissances α et β de θ_i et θ_j (qui valent soit $+1$ soit -1 dans $\cos(S)$) trouvées dans ce monôme, on n'a plus qu'à rechercher le coefficient du monôme en $X_i,\ Xb_i$ dans le développement de $\frac{r_i}{a_i}\theta_i{}^{\alpha}$ et celui en $X_j,\ Xb_j$ dans le développement de $(\frac{a_j}{r_j})^2\,\theta_j{}^{\beta}$. Ces deux coefficients sont en fait des coefficients de Hansen, calculables par l'une ou l'autre des méthodes vues précédemment. Le coefficient final est le produit de ces deux coefficients et de celui trouvé dans $\cos(S)$.

Pour extraire le coefficient d'un monôme donné par les exposants des 4 variables Y et Yb , on peut utiliser la procédure COEFTTS suivante, où *s* est un développement exprimé en fonction des quantités $\theta_i,\ \theta_j$ et de σ (comme $\cos(S)$ ou une de ses puissances). On donne le développement *s* et les 4 exposants *ni*, *nbi*, *nj*, *nbj* des variables $Y_i,\ Y_j,\ Yb_i,\ Yb_j$; on en tire la liste des exposants de θ_i, θ_j et σ et le coefficient de ce monôme.

```
> COEFTTS:=proc(s,ni,nbi,nj,nbj) local c,r,k,km:
> r:=table():  km:=1:
> c:=coeff(coeff(coeff(coeff(s,Y[i],ni),Y[j],nj),Yb[i],nbi),Yb[j],nbj);
> if type(c,'+') then
> km:=nops(c):
> for k from 1 to km do
> r[k]:=COEF(op(k,c)):
> od
> else r[1]:=COEF(c):
> fi:
> RETURN([seq(r[k],k=1..km)]):
> end:

> COEF:=proc(u) local ui,uj,us,cf:
```

```
> ui:=degree(u,theta[i]); uj:=degree(u,theta[j]);us:=degree(u,sigma);
> cf:=coeff(coeff(coeff(u,theta[i],ui),theta[j],uj),sigma,us);
> RETURN([ui,uj,us,cf]):
> end:
```

exemples :

le coefficient de ${Y_i}^2\ Yb_i\ Yb_j$ dans le développement de $\cos^2(S)$ a deux termes : $\theta_i^0 \times \theta_j^0 \times \sigma^{-1} \times (-2C_i\,C_j)$ et $\theta_i^2 \times \theta_j^{-2} \times \sigma \times (-C_i\,C_j)$:

```
> COEFTTS(COSSYm2,2,1,0,1);
```

$$[[0,\ 0,\ -1,\ -2\,C_i\,C_j],\ [2,\ -2,\ 1,\ -C_i\,C_j]]$$

```
> COEFTTS(DDELTACOSSYm2,2,1,0,1);
```

$$[[2,\ -2,\ 1,\ -4\,C_i\,C_j],\ [0,\ 0,\ -1,\ -8\,C_i\,C_j]]$$

Le monôme suivant n'existe pas dans le développement de $\cos(S)$, d'où un coefficient obtenu égal à 0 et les autres exposants indéterminés:

```
> COEFTTS(COSSYm1,1,2,0,1);
```

$$[[-\infty,\ -\infty,\ -\infty,\ 0]]$$

monôme indépendant des variables d'inclinaison dans $\cos(S)$:

```
> COEFTTS(COSSYm1,0,0,0,0);
```

$$[[-1,\ 1,\ -1,\ \frac{1}{2}],\ [1,\ -1,\ 1,\ \frac{1}{2}]]$$

```
> COEFTTS(COSSYm10,0,0,0,0);
```

$$[[-10,\ 10,\ -10,\ \frac{1}{1024}],\ [-8,\ 8,\ -8,\ \frac{5}{512}],\ [6,\ -6,\ 6,\ \frac{45}{1024}],\ [-6,\ 6,\ -6,\ \frac{45}{1024}],\ [10,\ -10,\ 10,\ \frac{1}{1024}],\ [8,\ -8,\ 8,\ \frac{5}{512}],\ [-4,\ 4,\ -4,\ \frac{15}{128}],$$
$$[0,\ 0,\ 0,\ \frac{63}{256}],\ [4,\ -4,\ 4,\ \frac{15}{128}],\ [-2,\ 2,\ -2,\ \frac{105}{512}],\ [2,\ -2,\ 2,\ \frac{105}{512}]]$$

```
> nops(%);
```

$$11$$

Pour déterminer le coefficient de $X^{n_1}\,Xb^{nb_1}$ dans le développement de $(\frac{r}{a})^{nn}\,\theta^{mm}$, on se sert du calcul de $Xnm = (\frac{r}{a})^{n}\,\theta^{m}$ réalisé plus haut . La procédure suivante y remplace *n* et *m* par les exposants utiles désignés par *nn* et *mm* .

```
> COEFXXb:=proc(Xnm,nn,mm,n1,nb1) local cf:
> cf:=coeff(coeff(Xnm,X,n1),Xb,nb1);
> # print(cf):
> cf:=subs(n=nn,m=mm,cf):
> RETURN(cf):
> end:
```

Exemples pour $X^2\,Xb^4$ dans $\frac{r}{a}\,\theta$, puis pour $X\,Xb^4$ dans $(\frac{r}{a})^{(-2)}\,\theta^{(-1)}$:

```
> COEFXXb(Xnm,1,1,2,4);
```

$$\frac{25}{1024}$$

```
> COEFXXb(Xnm,-2,-1,1,4);
```

$$\frac{-20}{3}$$

Voici finalement la procédure pour calculer le coefficient de l'inégalité (*p*,*q*) au degrés compris entre *degmin* et *degmax* dans le développement du produit $(\frac{r_i}{a_i})^{n_i}\,(\frac{r_j}{a_j})^{n_j}\,SERCOS$ où *SERCOS* est supposé être une puissance de $\cos(S)$.

```
> INEG:=proc(p,q,degmin,degmax,ni,nj,SERCOS,Xnm) local
> s,k,kk,r,ic,ip,mon,mono,m1,m2,n1,n2,n3,n4,n5,n6,n7,n8,moncos,monoc,nti
> ,ntj,ns,c1,c2,c3,c,nmon,nmoncos:
> ic:=p+q:  r:=0:    ip:=0:
> mon:=initmo(degmin,degmax,ic,ip,2):  nmon:=nops(mon):  #
> print("mon=",mon):
```

```
> for m1 from 1 to nmon do:
> mono:=op(m1,mon):  # print("mono=",mono):
> n1:=op(3,mono):n2:=op(4,mono):n3:=op(5,mono):n4:=op(6,mono):
> n5:=op(7,mono):n6:=op(8,mono):n7:=op(9,mono):n8:=op(10,mono):
> moncos:=COEFTTS(SERCOS,n7,n8,n5,n6); nmoncos:=nops(moncos):  #
> print("moncos=",moncos):
> k:= p +n4-n3+n8-n7:  c:=0:  # kk:=-q-(n2-n1+n6-n5):
> print("k=",k,kk):
> for m2 from 1 to nmoncos do:
> monoc:=op(m2,moncos):  c3:=op(4,monoc):  # print("monoc=",monoc):
> if c3<>0 then
> nti:=op(1,monoc):  ntj:=op(2,monoc):    ns:=op(3,monoc):
> c1:=COEFXXb(Xnm,ni,nti,n3,n4):
> c2:=COEFXXb(Xnm,nj,ntj,n1,n2);  # print(nti,ntj,c3,c1,c2):
> c:=c+c1*c2*c3*piecewise(k=ns,1,0):  #
> print(nti,ntj,c3,c1,c2,"c=",c):
> fi:
> od:  # print(c,op(11,mono)):
> r:=r+c*op(11,mono):    # print("r=",r):
> od:
> print("inégalité :",p*L[i]+q*L[j],"  coeff=",r):
> RETURN(r):
> end:
```

Exemple pour des valeurs numériques de p et q :

```
> pp:=1:  qq:=-2:  ing:=INEG(pp,qq,0,4,1,-2,COSSYm1,Xnm):
```

$$\text{“inégalité :”},\ L_i - 2\,L_j,\ \text{“ coeff=”},\ x_j - x_j\,yb_i\,y_i - \frac{1}{2}\,x_j\,xb_i\,x_i + 2\,C_i\,C_j\,x_j\,y_j\,yb_i - x_j\,yb_j\,y_j - \frac{3}{4}\,xb_j\,{x_j}^2$$

Exemple de calcul des termes du développement ayant une caractéristique d'inégalité donnée (on laisse p quelconque et on fixe la caractéristique):

```
> ic:=-1:  pp:='pp':  qq:=ic-pp:
> ing:=INEG(pp,qq,0,4,1,-2,COSSYm1,Xnm):
```

$$\text{“inégalité :”},\ pp\,L_i + (-1-pp)\,L_j,\ \text{“ coeff=”},\ (\frac{1}{4}\,\%3 - \frac{3}{4}\,\%2)\,x_i + \%1\,x_j + (-\frac{1}{4}\,\%3 + \frac{3}{4}\,\%2)\,x_i\,yb_i\,y_i - \frac{3}{4}\,\%2\,xb_i\,{y_i}^2$$
$$- \frac{3}{16}\,\%3\,xb_i\,{x_i}^2 - \frac{3}{2}\,C_i\,C_j\,(\begin{cases}1 & pp=0\\ 0 & otherwise\end{cases})\,x_i\,y_j\,yb_i + \frac{3}{2}\,C_i\,C_j\,(\begin{cases}1 & pp=0\\ 0 & otherwise\end{cases})\,xb_i\,y_j\,y_i - \%1\,x_j\,yb_i\,y_i$$
$$- \frac{1}{2}\,\%1\,x_j\,xb_i\,x_i - \frac{3}{4}\,(\begin{cases}1 & pp-1=-1\\ 0 & otherwise\end{cases})\,xb_i\,{y_j}^2 + 2\,C_i\,C_j\,(\begin{cases}1 & pp-1=0\\ 0 & otherwise\end{cases})\,x_j\,y_j\,yb_i$$
$$+ (-\frac{3}{32}\,(\begin{cases}1 & pp-1=-1\\ 0 & otherwise\end{cases}) + \frac{27}{32}\,(\begin{cases}1 & pp-1=1\\ 0 & otherwise\end{cases}))\,{x_j}^2\,xb_i + \frac{1}{2}\,C_i\,C_j\,(\begin{cases}1 & pp+2=0\\ 0 & otherwise\end{cases})\,x_i\,yb_j\,y_i$$
$$+ \frac{3}{8}\,(\begin{cases}1 & pp+2=-1\\ 0 & otherwise\end{cases})\,xb_j\,{x_i}^2 + (-\frac{1}{4}\,\%3 + \frac{3}{4}\,\%2)\,x_i\,yb_j\,y_j + (-\frac{1}{8}\,\%3 + \frac{3}{8}\,\%2)\,xb_j\,x_j\,x_i - \%1\,x_j\,yb_j\,y_j$$
$$- \frac{3}{4}\,\%1\,xb_j\,{x_j}^2$$
$$\%1 := \begin{cases}1 & pp=1\\ 0 & otherwise\end{cases}$$
$$\%2 := \begin{cases}1 & pp+1=1\\ 0 & otherwise\end{cases}$$
$$\%3 := \begin{cases}1 & pp+1=-1\\ 0 & otherwise\end{cases}$$

```
> eval(subs(pp=1,ing));
```

$$x_j - x_j\,yb_i\,y_i - \frac{1}{2}\,x_j\,xb_i\,x_i + 2\,C_i\,C_j\,x_j\,y_j\,yb_i - x_j\,yb_j\,y_j - \frac{3}{4}\,xb_j\,{x_j}^2$$

```
> pp:=1:  ing;
```

$$x_j - x_j\,yb_i\,y_i - \frac{1}{2}\,x_j\,xb_i\,x_i + 2\,C_i\,C_j\,x_j\,y_j\,yb_i - x_j\,yb_j\,y_j - \frac{3}{4}\,xb_j\,{x_j}^2$$

Comparaison avec un calcul direct par sélection d'une inégalité dans le développement complet (mais degré modéré) de la partie indirecte calculé précédemment:

```
> read cat(chemin,‘Uind4.m‘):
> Uind4coll:=TRONC(Uindxyij,4,varxyij):
> ingc:=coeff(coeff(Uind4coll,eL[i],pp),eL[j],qq); ing-ingc;
```

$$ingc := x_j - x_j\,yb_i\,y_i - \frac{1}{2}\,x_j\,xb_i\,x_i + 2\,C_i\,C_j\,x_j\,y_j\,yb_i - x_j\,yb_j\,y_j - \frac{3}{4}\,xb_j\,{x_j}^2$$

$$0$$

3.4. Développement de l'inverse de la distance de 2 planètes

Soient 2 planètes P_i et P_j repérées par rapport au Soleil par leurs rayons vecteurs de modules r_i et r_j et faisant entre eux l'angle S. Leur distance est donnée par $\Delta = \sqrt{{r_i}^2 + {r_j}^2 - 2\,r_i\,r_j\cos(S)}$. Si $r_i < r_j$, en notant $\rho = \frac{r_i}{r_j}$, on a alors son inverse :

$$\frac{1}{\Delta} = \frac{1}{r_j\,\sqrt{1+\rho^2-2\,\rho\cos(S)}}$$

On propose les deux méthodes classiques, l'une par les polynômes de Legendre et l'autre par les coefficients de Laplace :

3.4.1. Développement en polynômes de Legendre

On peut développer l'inverse de la distance en polynômes de Legendre: $\mathrm{C}_{25.1.1}$

$$\frac{r_j}{\Delta} = \sum_{n=0}^{\infty} \rho^n \, P_n(\cos(S)) \qquad \text{soit} \qquad \frac{a_j}{\Delta} = \sum_{n=0}^{\infty} \alpha^n \, (\frac{r_i}{a_i})^n \, (\frac{a_j}{r_j})^{(n+1)} \, P_n(\cos(S))$$

où α est le rapport des demi-grands axes $\frac{a_i}{a_j}$ (supposé inférieur à 1). En fait, ce type de développement n'est utilisable que si α est suffisamment petit pour n'avoir à calculer que les premiers termes de cette expression. C'est par exemple le cas de deux planètes fort éloignées, ou celui des perturbations d'un satellite par le Soleil.

Comme pour la partie indirecte, deux approches sont possibles, suivant le degré maximum souhaité pour ces développements :

3.4.2. Calcul explicite complet (degré faible)

Commencer, si ce n'est fait, par la phase d' initialisations

Si α est assez petit, on n'aura besoin que des premiers polynômes de Legendre et donc aussi que des premières puissances de $\cos(S)$. Un calcul explicite complet est alors possible :

On pose *da* = degré maxi en α et *degmx* = degré maxi en variables d'excentricités et inclinaisons:

```
> da:=8; degmx:= 4;
```

$$da := 8$$
$$degmx := 4$$

On peut lire le développement *UDxyij* correspondant en variables x, xb, y, yb s'il a déjà été calculé :

```
> read cat(chemin,`UDpxy`.da.`_`.degmx.`.m`):
> nops(UDxyij); op(%/2,UDxyij);
```

$$9798$$

$$\frac{1575}{1024}\, eL_j{}^2\, \alpha^6\, xb_i{}^3\, x_i$$

sinon, voici les instructions réalisant ce développement:

```
> with(orthopoly,P):
> CO:=TRONC(COSSXij10,degmx,varXYij):  nops(%);
> RSAiT:=TRONC(RSAi,degmx,varXYij):  nops(%);
> ASRjT:=TRONC(ASRj,degmx,varXYij):  nops(%);
```

$$228$$
$$14$$
$$13$$

```
> UD:=0:
> for k from 0 to da do:  deb:=time():  pk:=P(k,x):  pk:=subs(x=CO,pk):
> UD.k:=mtaylor(RSAiT^k*ASRjT^(k+1)*pk,varXYij,degmx+1):
> print(k,nops(%),"  ",time()-deb); UD:=UD+alpha^k*UD.k:
> od :
```

$$0,\ 13,\ \text{“ ”},\ 0$$
$$1,\ 202,\ \text{“ ”},\ .270$$
$$2,\ 256,\ \text{“ ”},\ .341$$
$$3,\ 257,\ \text{“ ”},\ .689$$
$$4,\ 257,\ \text{“ ”},\ .680$$
$$5,\ 258,\ \text{“ ”},\ 1.161$$
$$6,\ 258,\ \text{“ ”},\ 1.170$$
$$7,\ 259,\ \text{“ ”},\ 2.050$$

$$8, 259, \text{“ ”}, 1.581$$

```
> UDXYij:=sort(collect(expand(UD),varXYij,distributed),varXYij):
> nops(%);
```

$$299$$

```
> save UDXYij, cat(chemin,‘UDXY‘.da.‘_‘.degmx.‘.m‘):
```

On substitue les variables x et y aux variables X et Y (on a $x = Xb\ \exp(I\ L)$ ...) , mais on y remplace $\exp(I\ L)$ par une variable eL , ceci pour permettre des manipulations de polynômes généralisés (polynômes dont certaines variables peuvent être à des degrés négatifs). On y remplace alors aussi $\sigma = e^{(I\,(L_i - L_j))}$ par $\frac{eL_i}{eL_j}$:

```
> UDxyij1:=subs( X[i]=xb[i]*eL[i],Xb[i]=x[i]/eL[i],
> Y[i]=yb[i]*eL[i],Yb[i]=y[i]/eL[i], X[j]=xb[j]*eL[j],Xb[j]=x[j]/eL[j],
> Y[j]=yb[j]*eL[j],Yb[j]=y[j]/eL[j],expand(UDXYij)):  nops(%);
```

$$9798$$

```
> UDxyij2:=expand(algsubs(1/sigma=eL[j]/eL[i],UDxyij1)):  nops(%);
```

$$9798$$

```
> UDxyij:=expand(algsubs(sigma=eL[i]/eL[j],UDxyij2)):nops(%);
```

$$9798$$

exemple de terme :

```
> op(nops(UDxyij)/2,UDxyij);
```

$$\frac{1575}{1024}\, {eL_j}^2\, \alpha^6\, {xb_i}^3\, x_i$$

On sauvegarde enfin le résultat :

```
> save UDxyij, cat(chemin,‘UDpxy‘.da.‘_‘.degmx.‘.m‘):
```

L'extraction d'une inégalité peut se faire en recherchant le coefficient des puissances adéquates (notées ici *pLi* et *pLj*) des variables

eL_i et eL_j (avec *pLi* et *pLj* nuls, on obtient l'inégalité séculaire). Exemple :

```
> pLi:=1; pLj:=-4;
> coeff(coeff(UDxyij,eL[i],pLi),eL[j],pLj):
> UDc:=map(sort,sort(collect(%,varxyij,distributed),varxyij),alpha);
> nops(%);
```

$$pLi := 1$$

$$pLj := -4$$

$$\begin{aligned}UDc := & (-\frac{32109}{8192}\,\alpha^8 - \frac{14217}{4096}\,\alpha^6 - \frac{6545}{1536}\,\alpha^4)\,{x_i}^3 + (\frac{175959}{16384}\,\alpha^7 + \frac{9555}{1024}\,\alpha^5 + \frac{1425}{128}\,\alpha^3)\,{x_i}^2\,x_j \\ & + (-\frac{97335}{8192}\,\alpha^8 - \frac{9975}{1024}\,\alpha^6 - \frac{265}{32}\,\alpha^4 - \frac{153}{16}\,\alpha^2)\,x_i\,{x_j}^2 + (-\frac{31185}{2048}\,\alpha^8 - \frac{4725}{512}\,\alpha^6 - \frac{35}{8}\,\alpha^4)\,x_i\,{y_i}^2 \\ & + (\frac{31185}{1024}\,C_i\,C_j\,\alpha^8 + \frac{4725}{256}\,C_i\,C_j\,\alpha^6 + \frac{35}{4}\,C_i\,C_j\,\alpha^4)\,x_i\,y_i\,y_j + (-\frac{31185}{2048}\,\alpha^8 - \frac{4725}{512}\,\alpha^6 - \frac{35}{8}\,\alpha^4)\,x_i\,{y_j}^2 \\ & + (\frac{22225}{6144}\,\alpha^7 + \frac{745}{256}\,\alpha^5 + \frac{77}{32}\,\alpha^3 + \frac{8}{3}\,\alpha)\,{x_j}^3 + (\frac{33075}{2048}\,\alpha^7 + \frac{315}{32}\,\alpha^5 + \frac{75}{16}\,\alpha^3)\,x_j\,{y_i}^2 \\ & + (-\frac{33075}{1024}\,C_i\,C_j\,\alpha^7 - \frac{315}{16}\,C_i\,C_j\,\alpha^5 - \frac{75}{8}\,C_i\,C_j\,\alpha^3)\,x_j\,y_i\,y_j + (\frac{33075}{2048}\,\alpha^7 + \frac{315}{32}\,\alpha^5 + \frac{75}{16}\,\alpha^3)\,x_j\,{y_j}^2\end{aligned}$$

$$10$$

Si on avait conservé les fonctions exponentielles plutot que d'introduire les variables *eL* , on aurait pu regrouper les inégalités par la dernière des instructions suivantes, mais les développements deviennent trop importants dès que la puissance de α dépasse 5 .

```
> UDxyeij1:=subs( X[i]=xb[i]*exp(I*L[i]),Xb[i]=x[i]/exp(I*L[i]),
> Y[i]=yb[i]*exp(I*L[i]),Yb[i]=y[i]/exp(I*L[i]),
> X[j]=xb[j]*exp(I*L[j]),Xb[j]=x[j]/exp(I*L[j]),
> Y[j]=yb[j]*exp(I*L[j]),Yb[j]=y[j]/exp(I*L[j]),expand(UDXYij)):
> UDxyeij2:=expand(algsubs(1/sigma=exp(I*L[j])/exp(I*L[i]),UDxyeij1)):
> UDxyij3:=expand(algsubs(sigma=exp(I*L[i])/exp(I*L[j]),UDxyeij2)):
> nops(%);
```

$$9798$$

```
> UDxyij:=collect(map(combine,UDxyij3,exp),exp):  nops(%);
```

$$410$$

```
> UDxyij_exp:=UDxyij:
> save UDxyij_exp, cat(chemin,`UDxy_exp`.da.`_`.degmx.`.m`):
```

L'extraction d'une inégalité pourrait alors se faire ainsi:

```
> inegalite:=exp(4*I*L[i]-I*L[j]):
> map(sort,map(collect,select(has,UDxyij,inegalite),varxyij,distributed
> ),varxyij);
```

$$((-\frac{7}{128}\alpha^3+\frac{1}{6}\alpha+\frac{5}{512}\alpha^5+\frac{1925}{49152}\alpha^7)\,xb_i{}^3+(\frac{1575}{32768}\alpha^8-\frac{5}{256}\alpha^4-\frac{3}{16}\alpha^2-\frac{315}{4096}\alpha^6)\,xb_i{}^2\,xb_j$$
$$+(\frac{15}{256}\alpha^3-\frac{2457}{32768}\alpha^7+\frac{105}{4096}\alpha^5)\,xb_i\,xb_j{}^2+(\frac{105}{128}\alpha^5-\frac{4725}{4096}\alpha^7+\frac{45}{32}\alpha^3)\,xb_i\,yb_i{}^2$$
$$+(-\frac{105}{64}C_i\,C_j\,\alpha^5-\frac{45}{16}C_i\,C_j\,\alpha^3+\frac{4725}{2048}C_i\,C_j\,\alpha^7)\,xb_i\,yb_i\,yb_j+(\frac{105}{128}\alpha^5-\frac{4725}{4096}\alpha^7+\frac{45}{32}\alpha^3)\,xb_i\,yb_j{}^2$$
$$+(\frac{16401}{131072}\alpha^8-\frac{35}{6144}\alpha^4+\frac{231}{8192}\alpha^6)\,xb_j{}^3+(\frac{2835}{1024}\alpha^6+\frac{35}{64}\alpha^4+\frac{51975}{8192}\alpha^8)\,xb_j\,yb_i{}^2$$
$$+(-\frac{2835}{512}C_i\,C_j\,\alpha^6-\frac{35}{32}C_i\,C_j\,\alpha^4-\frac{51975}{4096}C_i\,C_j\,\alpha^8)\,xb_j\,yb_i\,yb_j+(\frac{2835}{1024}\alpha^6+\frac{35}{64}\alpha^4+\frac{51975}{8192}\alpha^8)\,xb_j\,yb_j{}^2)$$
$$e^{(4\,I\,L_i-I\,L_j)}$$

ou celle du terme séculaire :

```
> sort(collect(remove(has,UDxyij,exp),varxyij,distributed),varxyij):
> PRINTRONC(%,2,varxyij);
```

$$(\frac{2625}{2048}\alpha^6+\frac{135}{512}\alpha^4+\frac{231525}{65536}\alpha^8)\,x_i{}^2\,xb_i{}^2,\ +\ldots+,(\frac{11025}{8192}\alpha^8+\frac{45}{64}\alpha^4+\frac{3}{8}\alpha^2+\frac{525}{512}\alpha^6)\,x_i\,xb_i$$
$$+(-\frac{4725}{4096}\alpha^7-\frac{105}{128}\alpha^5-\frac{15}{32}\alpha^3)\,x_i\,xb_j+(-\frac{4725}{4096}\alpha^7-\frac{105}{128}\alpha^5-\frac{15}{32}\alpha^3)\,xb_i\,x_j$$
$$+(\frac{11025}{8192}\alpha^8+\frac{45}{64}\alpha^4+\frac{3}{8}\alpha^2+\frac{525}{512}\alpha^6)\,x_j\,xb_j+(-\frac{3}{2}\alpha^2-\frac{11025}{2048}\alpha^8-\frac{525}{128}\alpha^6-\frac{45}{16}\alpha^4)\,y_i\,yb_i$$
$$+(\frac{3}{2}\alpha^2\,C_i\,C_j+\frac{525}{128}C_i\,C_j\,\alpha^6+\frac{45}{16}C_i\,C_j\,\alpha^4+\frac{11025}{2048}C_i\,C_j\,\alpha^8)\,y_i\,yb_j$$
$$+(\frac{3}{2}\alpha^2\,C_i\,C_j+\frac{525}{128}C_i\,C_j\,\alpha^6+\frac{45}{16}C_i\,C_j\,\alpha^4+\frac{11025}{2048}C_i\,C_j\,\alpha^8)\,yb_i\,y_j+(-\frac{3}{2}\alpha^2-\frac{11025}{2048}\alpha^8-\frac{525}{128}\alpha^6-\frac{45}{16}\alpha^4)\,y_j\,yb_j$$
$$+\frac{1225}{16384}\alpha^8+1+\frac{1}{4}\alpha^2+\frac{9}{64}\alpha^4+\frac{25}{256}\alpha^6$$

3.4.3. Calcul par inégalités

Pour aller plus loin en degré des excentricités et inclinaisons et en puissances de α, on peut utliser les procédures données dans ce paragraphe qui permettent de calculer directement le développement d'une inégalité quelconque donnée:

```
> with(orthopoly,P):
> read cat(chemin,`initmo.pr`):
> read cat(chemin,`Xnm10.m`);
> read cat(chemin,`COSSYm1a10.m`):
> # read cat(chemin,`COSSYm11a20.m`):
```

Le fichier suivant, qui contient le développement complet construit dans la section précédente, est lu simplement à des fins de comparaison avec les résultats de cette section.

```
> read cat(chemin,`UDpxy8_4.m`):
```

Si les fichiers COSSYm1a10.m et COSSYm11a20.m ne sont pas disponibles, les instructions suivantes permettent de les créer. Elles calculent les puissances croissantes de $\cos(S)$, à partir de l'expression générale de la puissance *m* de $\cos(S)$. On va ici jusqu'à la puissance 20 , permettant donc de considérer les polynômes de Legendre $P_n(x)$ jusqu'à l'indice 20 , et des développements jusqu'à α^{20} :

```
> read cat(chemin,`COSSYpuissm.m`) :      # -> COSSYm :  puissance  m   de
> cos(S)
> for k  from 1 to 10 do:    deb:=time():
> COSSYm.k:=expand(simplify(subs(m=k,COSSYm))):  print(k,nops(%),"
> ",time()-deb);od:
```

1, 22, “ ”, .729

2, 180, “ ”, .720

3, 800, “ ”, 1.331

4, 2000, “ ”, 1.970

5, 3456, “ ”, 4.629

6, 4562, “ ”, 8.691

7, 5668, “ ”, 5.260

8, 6774, “ ”, 5.819

9, 7880, “ ”, 5.530

10, 8986, “ ”, 5.780

```
> save
> COSSYm1,COSSYm2,COSSYm3,COSSYm4,COSSYm5,COSSYm6,COSSYm7,COSSYm8,COSSYm
> 9,COSSYm10, cat(chemin,‘COSSYm1a10.m‘):
> for k  from 11 to 20 do:    deb:=time():
> COSSYm.k:=expand(simplify(subs(m=k,COSSYm))):  print(k,nops(%),"
> ",time()-deb);od:
```

11, 10092, “ ”, 7.429

12, 11198, “ ”, 6.410

13, 12304, “ ”, 7.800

14, 13410, “ ”, 7.831

15, 14516, “ ”, 8.800

16, 15622, “ ”, 10.629

17, 16728, “ ”, 11.540

18, 17834, “ ”, 10.480

19, 18940, “ ”, 11.371

$$20,\ 20046,\ \text{“ ”},\ 13.280$$

```
> save
> COSSYm11,COSSYm12,COSSYm13,COSSYm14,COSSYm15,COSSYm16,COSSYm17,COSSYm1
> 8,COSSYm19,COSSYm20, cat(chemin,`COSSYm11a20.m`):
```

Exemple :

```
> COSSYm1;
```

$$-\frac{1}{2}\frac{\theta_i\,\theta_j\,{Y_i}^2\,Y_j\,Yb_j}{\sigma}+\frac{1}{2}\frac{\theta_i\,{Y_i}^2\,{Yb_j}^2}{\theta_j\,\sigma}-\frac{1}{2}\theta_i\,\theta_j\,\sigma\,Y_i\,{Y_j}^2\,Yb_i+\frac{1}{2}\frac{\theta_j\,Y_i\,Yb_i\,Y_j\,Yb_j}{\theta_i\,\sigma}+\frac{1}{2}\frac{\theta_i\,\sigma\,Y_i\,Yb_i\,Y_j\,Yb_j}{\theta_j}-\frac{1}{2}\frac{Y_i\,Yb_i\,{Yb_j}^2}{\theta_i\,\theta_j\,\sigma}$$
$$+\frac{1}{2}\frac{\theta_j\,\sigma\,{Y_j}^2\,{Yb_i}^2}{\theta_i}-\frac{1}{2}\frac{\sigma\,Y_j\,{Yb_i}^2\,Yb_j}{\theta_i\,\theta_j}+\frac{1}{2}\frac{\theta_i\,\theta_j\,{Y_i}^2}{\sigma}-\theta_i\,\theta_j\,C_i\,C_j\,Y_i\,Y_j-\frac{1}{2}\frac{\theta_j\,Y_i\,Yb_i}{\theta_i\,\sigma}-\frac{1}{2}\frac{\theta_i\,\sigma\,Y_i\,Yb_i}{\theta_j}+\frac{\theta_i\,C_i\,C_j\,Y_i\,Yb_j}{\theta_j}$$
$$+\frac{1}{2}\theta_i\,\theta_j\,\sigma\,{Y_j}^2+\frac{\theta_j\,C_i\,C_j\,Y_j\,Yb_i}{\theta_i}-\frac{1}{2}\frac{\theta_j\,Y_j\,Yb_j}{\theta_i\,\sigma}-\frac{1}{2}\frac{\theta_i\,\sigma\,Y_j\,Yb_j}{\theta_j}+\frac{1}{2}\frac{\sigma\,{Yb_i}^2}{\theta_i\,\theta_j}-\frac{C_i\,C_j\,Yb_i\,Yb_j}{\theta_i\,\theta_j}+\frac{1}{2}\frac{{Yb_j}^2}{\theta_i\,\theta_j\,\sigma}+\frac{1}{2}\frac{\theta_j}{\theta_i\,\sigma}$$
$$+\frac{1}{2}\frac{\theta_i\,\sigma}{\theta_j}$$

Voici les procédures utiles pour calculer directement le développement en variables $\{x,\ xb,\ y,\ yb\}$ d'une inégalité $(I\,(p\,L_i+q\,L_j))$ désignée par la combinaison (p, q) des longitudes moyennes. On retrouve d'abord celles utilisées pour le calcul de la partie indirecte par inégalités :

```
> COEFTTS:=proc(s,ni,nbi,nj,nbj) local c,r,k,km:
> r:=table(): km:=1:
> c:=coeff(coeff(coeff(coeff(s,Y[i],ni),Y[j],nj),Yb[i],nbi),Yb[j],nbj);
> if type(c,`+`) then
> km:=nops(c):
> for k from 1 to km do
```

```
> r[k]:=COEF(op(k,c)):
> # print(op(k,c),r[k])
> od
> else r[1]:=COEF(c):
> # print(c,r[1])
> fi:
> RETURN([seq(r[k],k=1..km)]):
> end:

> COEF:=proc(u) local ui,uj,us,cf:
> ui:=degree(u,theta[i]); uj:=degree(u,theta[j]);us:=degree(u,sigma);
> cf:=coeff(coeff(coeff(u,theta[i],ui),theta[j],uj),sigma,us);
> RETURN([ui,uj,us,cf]):
> end:

> COEFXXb:=proc(XXnm,nn,mm,n1,nb1) local cf:
> cf:=coeff(coeff(XXnm,X,n1),Xb,nb1);
> # print(cf):
> cf:=subs(n=nn,m=mm,cf):
> RETURN(cf):
> end:
```

Procédure *INVDLEG* pour le calcul de l'inégalité (*p*,*q*) aux degrés compris entre *degmin* et *degmax*, dans le développement de $\frac{a_j}{\Delta}$ en polynômes de Legendre jusqu'au degré *damax* en α. Les puissances utiles de $\cos(S)$ sont supposées calculées au préalable et accessibles dans les variables *SERCOS.k* pour $k = 1..damax$ (séries tronquées préalablement au degré *degmax* en inclinaisons).

Les développements de $(\frac{r}{a})^n\theta^m$ tronqués au degré *degmax* en excentricités sont supposés accessibles par la variable *XXnm* . Le développement formel en polynômes de Legendre est passé par la variable *SanPnx* .

```
> INVDLEG:=proc(p,q,degmin,degmax,damax,SERCOS,XXnm,SanPnx) local
> ic,ip,r,mon,nmon,m1,m2,mono,n1,n2,n3,n4,n5,n6,n7,n8,k1,k2,k,pk,cfPk,t,
> ncfPk, moncos,nmoncos,monoc,ns,c,nti,ntj,c1,c2,c3,cfa,lda,nni,nnj:
> ic:=p+q:  r:=0:    ip:=0:
> mon:=initmo(degmin,degmax,ic,ip,2):  nmon:=nops(mon):
> for m1 from 1 to nmon do:
> c:=0:
> mono:=op(m1,mon):
> n1:=op(3,mono):n2:=op(4,mono):n3:=op(5,mono):n4:=op(6,mono):
> n5:=op(7,mono):n6:=op(8,mono):n7:=op(9,mono):n8:=op(10,mono):
> for k2 from 0 to damax do:
> pk:=coeff(SanPnx,x,k2):
> cfPk:=[coeffs(pk,alpha,'t')]:  ncfPk:=nops(cfPk):
> moncos:=COEFTTS(SERCOS.k2,n7,n8,n5,n6); nmoncos:=nops(moncos):
> k:= p +n4-n3+n8-n7:
> for m2 from 1 to nmoncos do:
> monoc:=op(m2,moncos):  c3:=op(4,monoc):#
> print("monoc=",monoc):
> if c3<>0 then
> nti:=op(1,monoc):  ntj:=op(2,monoc):  ns:=op(3,monoc):
> for k1 from 1 to ncfPk do:
> cfa:=op(k1,[t]):
> lda:=degree(cfa,alpha):
> nni:=lda:  nnj:=-(lda+1):
```

```
> c1:=COEFXXb(XXnm,nni,nti,n3,n4):
> c2:=COEFXXb(XXnm,nnj,ntj,n1,n2);
> c:=c+c1*c2*c3*piecewise(k=ns,1,0)*cfa*op(k1,cfPk):
> od:
> fi :
> od:
> od:
> r:=r+c*op(11,mono):
> od:
> RETURN(r);
> end:
```

Initialisation des calculs de développement d'inégalités à un degré *degmax* donné. On tronque les séries à ce degré avant leur utilisation, pour diminuer les temps de calcul.

```
> IniINEGLeg:=proc(degmin,degmax,damax) local k,r:
> r:=table():
> r[1]:=TRONC(Xnm,degmax,[X,Xb]);
> r[2]:=1:
> for k from 1 to damax do:
> r[k+2]:=TRONC(COSSYm.k,degmax,[Y[i],Yb[i],Y[j],Yb[j]]):
> od:
> RETURN([seq(r[k],k=1..damax+2)]):
> end:
```

Exemple d'utilisation: On fixe les degrés utiles et la puissance maximum souhaitée pour α , puis on tronque les séries avec *IniINEGLeg* :

```
> Dmin:=0; Dmax:=4;
> da:=8;
```

$$Dmin := 0$$
$$Dmax := 4$$
$$da := 8$$

On tronque les développements en inclinaisons et en excentricités au degré *Dmax* et ceux en α au degré *da* . On calcule aussi le développement formel en polynômes de Legendre limité à ce degré (factorisé en une variable *x* qui remplace formellement $\cos(S)$):

```
> deb0:=time():
> inineg:=IniINEGLeg(Dmin,Dmax,da):  print(time()-deb0):
> XnmT:= op(1,inineg):
> k:='k':  for k from 0 to da do:  COSSYmT.k:=op(k+2,inineg) od:
> som2alfa:=collect(expand(sum(alpha^n*P(n,x),n=0..da)),x);
```

$$4.930$$

$$som2alfa := \frac{6435}{128}\,\alpha^8\,x^8 + \frac{429}{16}\,\alpha^7\,x^7 + (\frac{231}{16}\,\alpha^6 - \frac{3003}{32}\,\alpha^8)\,x^6 + (\frac{63}{8}\,\alpha^5 - \frac{693}{16}\,\alpha^7)\,x^5 + (\frac{3465}{64}\,\alpha^8 - \frac{315}{16}\,\alpha^6 + \frac{35}{8}\,\alpha^4)\,x^4$$
$$+ (\frac{5}{2}\,\alpha^3 + \frac{315}{16}\,\alpha^7 - \frac{35}{4}\,\alpha^5)\,x^3 + (-\frac{15}{4}\,\alpha^4 + \frac{3}{2}\,\alpha^2 - \frac{315}{32}\,\alpha^8 + \frac{105}{16}\,\alpha^6)\,x^2 + (\alpha + \frac{15}{8}\,\alpha^5 - \frac{3}{2}\,\alpha^3 - \frac{35}{16}\,\alpha^7)\,x + 1 + \frac{3}{8}\,\alpha^4$$
$$+ \frac{35}{128}\,\alpha^8 - \frac{1}{2}\,\alpha^2 - \frac{5}{16}\,\alpha^6$$

Il suffit ensuite de définir les entiers (p, q) représentant une inégalité, et d'exécuter *INVDLEG* :

```
> pp:=0; qq:=0;
```

$$pp := 0$$

$$qq := 0$$

```
> deb:=time():
> UD:=INVDLEG(pp,qq,Dmin,Dmax,da,COSSYmT,XnmT,som2alfa):
> UD:=map(sort,sort(collect(UD,varxyij,distributed),varxyij),alpha):
> nops(%);
> print("duree=",time()-deb):
```

$$65$$

$$\text{“duree=”},\ 8.349$$

```
> PRINTRONC(UD,2,varxyij);
```

$$\begin{aligned}
&(\frac{231525}{65536}\,\alpha^8 + \frac{2625}{2048}\,\alpha^6 + \frac{135}{512}\,\alpha^4)\,{x_i}^2\,{xb_i}^2,\ + \ldots + ,(\frac{11025}{8192}\,\alpha^8 + \frac{525}{512}\,\alpha^6 + \frac{45}{64}\,\alpha^4 + \frac{3}{8}\,\alpha^2)\,x_i\,xb_i \\
&+ (-\frac{4725}{4096}\,\alpha^7 - \frac{105}{128}\,\alpha^5 - \frac{15}{32}\,\alpha^3)\,x_i\,xb_j + (-\frac{4725}{4096}\,\alpha^7 - \frac{105}{128}\,\alpha^5 - \frac{15}{32}\,\alpha^3)\,xb_i\,x_j \\
&+ (\frac{11025}{8192}\,\alpha^8 + \frac{525}{512}\,\alpha^6 + \frac{45}{64}\,\alpha^4 + \frac{3}{8}\,\alpha^2)\,x_j\,xb_j + (-\frac{11025}{2048}\,\alpha^8 - \frac{525}{128}\,\alpha^6 - \frac{45}{16}\,\alpha^4 - \frac{3}{2}\,\alpha^2)\,y_i\,yb_i \\
&+ (\frac{11025}{2048}\,C_i\,C_j\,\alpha^8 + \frac{525}{128}\,C_i\,C_j\,\alpha^6 + \frac{45}{16}\,C_i\,C_j\,\alpha^4 + \frac{3}{2}\,C_i\,C_j\,\alpha^2)\,y_i\,yb_j \\
&+ (\frac{11025}{2048}\,C_i\,C_j\,\alpha^8 + \frac{525}{128}\,C_i\,C_j\,\alpha^6 + \frac{45}{16}\,C_i\,C_j\,\alpha^4 + \frac{3}{2}\,C_i\,C_j\,\alpha^2)\,yb_i\,y_j + (-\frac{11025}{2048}\,\alpha^8 - \frac{525}{128}\,\alpha^6 - \frac{45}{16}\,\alpha^4 - \frac{3}{2}\,\alpha^2)\,y_j\,yb_j \\
&+ \frac{1225}{16384}\,\alpha^8 + 1 + \frac{9}{64}\,\alpha^4 + \frac{25}{256}\,\alpha^6 + \frac{1}{4}\,\alpha^2
\end{aligned}$$

Vérifications en comparant avec les résultats tirés du développement complet au degré 4 pour *da* = 8 :

```
> UDxy:=TRONC(UDxyij,Dmax,varxyij):
> pLi:=pp:  pLj:=qq:
> coeff(coeff(UDxy,eL[i],pLi),eL[j],pLj):
> b:=map(sort,sort(collect(%,varxyij,distributed),varxyij),alpha):
> simplify(UD-b);
```

$$0$$

calcul pour une inégalité de caractéristique donnée , p restant indéterminé:

```
> ic:=0; pp:='pp'; qq:=ic-pp;
```

$$ic := 0$$

$$pp := pp$$

$$qq := -pp$$

```
> deb:=time():
> UDp:=INVDLEG(pp,qq,Dmin,Dmax,da,COSSYmT,XnmT,som2alfa):
> UDp:=map(sort,sort(collect(UDp,varxyij,distributed),varxyij),alpha):
> nops(%);
> print("duree=",time()-deb):
```

$$105$$

$$\text{“duree=”},\ 9.770$$

exemple de terme :

```
> coeff(coeff(TRONC(UDp,2,varxyij),y[i],1),yb[i],1);
```

$$
\begin{aligned}
&-\frac{5355}{1024}\,\%2\,\alpha^8-\frac{6435}{4096}\left(\begin{cases}1 & pp=-8\\ 0 & otherwise\end{cases}\right)\alpha^8-\frac{11025}{2048}\,\%1\,\alpha^8-\frac{3861}{1024}\left(\begin{cases}1 & pp=6\\ 0 & otherwise\end{cases}\right)\alpha^8\\
&-\frac{6435}{4096}\left(\begin{cases}1 & pp=8\\ 0 & otherwise\end{cases}\right)\alpha^8-\frac{4851}{1024}\left(\begin{cases}1 & pp=4\\ 0 & otherwise\end{cases}\right)\alpha^8-\frac{5355}{1024}\,\%3\,\alpha^8-\frac{8883}{2048}\left(\begin{cases}1 & pp=-3\\ 0 & otherwise\end{cases}\right)\alpha^7\\
&-\frac{8883}{2048}\left(\begin{cases}1 & pp=3\\ 0 & otherwise\end{cases}\right)\alpha^7-\frac{4851}{1024}\left(\begin{cases}1 & pp=-4\\ 0 & otherwise\end{cases}\right)\alpha^8-\frac{3861}{1024}\left(\begin{cases}1 & pp=-6\\ 0 & otherwise\end{cases}\right)\alpha^8-\frac{9625}{2048}\,\%4\,\alpha^7\\
&-\frac{7161}{2048}\left(\begin{cases}1 & pp=-5\\ 0 & otherwise\end{cases}\right)\alpha^7-\frac{3003}{2048}\left(\begin{cases}1 & pp=7\\ 0 & otherwise\end{cases}\right)\alpha^7-\frac{1995}{512}\,\%3\,\alpha^6-\frac{525}{128}\,\%1\,\alpha^6\\
&-\frac{693}{512}\left(\begin{cases}1 & pp=6\\ 0 & otherwise\end{cases}\right)\alpha^6-\frac{819}{256}\left(\begin{cases}1 & pp=-4\\ 0 & otherwise\end{cases}\right)\alpha^6-\frac{1995}{512}\,\%2\,\alpha^6-\frac{7161}{2048}\left(\begin{cases}1 & pp=5\\ 0 & otherwise\end{cases}\right)\alpha^7\\
&-\frac{3003}{2048}\left(\begin{cases}1 & pp=-7\\ 0 & otherwise\end{cases}\right)\alpha^7-\frac{9625}{2048}\,\%5\,\alpha^7-\frac{33}{16}\,\%5\,\alpha^3-\frac{15}{16}\left(\begin{cases}1 & pp=-3\\ 0 & otherwise\end{cases}\right)\alpha^3\\
&-\frac{15}{16}\left(\begin{cases}1 & pp=3\\ 0 & otherwise\end{cases}\right)\alpha^3-\frac{33}{16}\,\%4\,\alpha^3-\frac{5}{2}\,\%2\,\alpha^4-\frac{35}{32}\left(\begin{cases}1 & pp=-4\\ 0 & otherwise\end{cases}\right)\alpha^4-\frac{35}{32}\left(\begin{cases}1 & pp=4\\ 0 & otherwise\end{cases}\right)\alpha^4\\
&-\frac{45}{16}\,\%1\,\alpha^4-\frac{5}{2}\,\%3\,\alpha^4-\frac{315}{256}\left(\begin{cases}1 & pp=-5\\ 0 & otherwise\end{cases}\right)\alpha^5-\frac{735}{256}\left(\begin{cases}1 & pp=-3\\ 0 & otherwise\end{cases}\right)\alpha^5-\frac{435}{128}\,\%4\,\alpha^5\\
&-\frac{315}{256}\left(\begin{cases}1 & pp=5\\ 0 & otherwise\end{cases}\right)\alpha^5-\frac{735}{256}\left(\begin{cases}1 & pp=3\\ 0 & otherwise\end{cases}\right)\alpha^5-\frac{435}{128}\,\%5\,\alpha^5-\frac{693}{512}\left(\begin{cases}1 & pp=-6\\ 0 & otherwise\end{cases}\right)\alpha^6\\
&-\frac{819}{256}\left(\begin{cases}1 & pp=4\\ 0 & otherwise\end{cases}\right)\alpha^6-\frac{1}{2}\,\%5\,\alpha-\frac{1}{2}\,\%4\,\alpha-\frac{3}{4}\,\%3\,\alpha^2-\frac{3}{4}\,\%2\,\alpha^2-\frac{3}{2}\,\%1\,\alpha^2
\end{aligned}
$$

$$\%1 := \begin{cases}1 & pp=0\\ 0 & otherwise\end{cases}$$

$$\%2 := \begin{cases}1 & pp=2\\ 0 & otherwise\end{cases}$$

$$\%3 := \begin{cases}1 & pp=-2\\ 0 & otherwise\end{cases}$$

$$\%4 := \begin{cases}1 & pp=-1\\ 0 & otherwise\end{cases}$$

$$\%5 := \begin{cases}1 & pp=1\\ 0 & otherwise\end{cases}$$

```
> sort(eval(subs(pp=0,TRONC(UDp,2,varxyij))),varxyij);
```

$$\left(\frac{11025}{8192}\,\alpha^8+\frac{525}{512}\,\alpha^6+\frac{45}{64}\,\alpha^4+\frac{3}{8}\,\alpha^2\right)x_i\,xb_i+\left(-\frac{4725}{4096}\,\alpha^7-\frac{105}{128}\,\alpha^5-\frac{15}{32}\,\alpha^3\right)x_i\,xb_j+\left(-\frac{4725}{4096}\,\alpha^7-\frac{105}{128}\,\alpha^5-\frac{15}{32}\,\alpha^3\right)xb_i\,x_j$$
$$+\left(\frac{11025}{8192}\,\alpha^8+\frac{525}{512}\,\alpha^6+\frac{45}{64}\,\alpha^4+\frac{3}{8}\,\alpha^2\right)x_j\,xb_j+\left(-\frac{11025}{2048}\,\alpha^8-\frac{525}{128}\,\alpha^6-\frac{45}{16}\,\alpha^4-\frac{3}{2}\,\alpha^2\right)y_i\,yb_i$$
$$+\left(\frac{11025}{2048}\,C_i\,C_j\,\alpha^8+\frac{525}{128}\,C_i\,C_j\,\alpha^6+\frac{45}{16}\,C_i\,C_j\,\alpha^4+\frac{3}{2}\,C_i\,C_j\,\alpha^2\right)y_i\,yb_j$$
$$+\left(\frac{11025}{2048}\,C_i\,C_j\,\alpha^8+\frac{525}{128}\,C_i\,C_j\,\alpha^6+\frac{45}{16}\,C_i\,C_j\,\alpha^4+\frac{3}{2}\,C_i\,C_j\,\alpha^2\right)yb_i\,y_j+\left(-\frac{11025}{2048}\,\alpha^8-\frac{525}{128}\,\alpha^6-\frac{45}{16}\,\alpha^4-\frac{3}{2}\,\alpha^2\right)y_j\,yb_j$$
$$+1+\frac{1225}{16384}\,\alpha^8+\frac{1}{4}\,\alpha^2+\frac{9}{64}\,\alpha^4+\frac{25}{256}\,\alpha^6$$

3.4.4. Développement en coefficients de Laplace

On peut aussi développer l'inverse de la distance en série de Fourier de S ; ses coefficients sont appelés **coefficients de Laplace**: C25.1.4

$$\left(\frac{r_j}{\Delta}\right)^n = {r_j}^n\left(\sum_{s=-\infty}^{\infty} b_{\{|s|,\,n\}}(\rho)\,\exp\,(I\,s\,S)\right)\qquad\text{avec}\qquad b_{\{s,\,n\}}(\rho)=\frac{[n]_s}{[1]_s}\rho^s\,\mathrm{F}(n,\,n+s,\,s+1,\,\rho^2)$$

pour s positif ou nul, où les notations $[n]_s$ représentent les symboles de Pochhammer et où F est la fonction hypergéométrique de Gauss. En remplaçant ρ par $\alpha\,\frac{r_i}{a_i}\,\frac{a_j}{r_j}$, et ρ^2 par $\alpha^2\,(1+\delta)$ avec $\delta=(\frac{r_i}{a_i})^2\,(\frac{a_j}{r_j})^2-1$, on a encore le développement des fonctions hypergéométriques au voisinage de $\delta=0$:

$$\mathrm{F}(a,\,b,\,c,\,\alpha^2\,(1+\delta))=\sum_m \frac{[a]_m\,[b]_m\,\alpha^{(2\,m)}\,\delta^m\,\mathrm{F}(a+m,\,b+m,\,c+m,\,\alpha^2)}{[c]_m\,[1]_m}$$

car on note que δ est une quantité petite, de degré 1 au moins en excentricités.

En fait, on commence par développer de cette façon la partie indépendante des inclinaisons (associée au problème plan) :

$$\frac{1}{\mathrm{D}} = \frac{1}{\sqrt{1+\rho^2-2\,\rho\cos(w_i-w_j)}}$$

où $w_i - w_j$ représente la différence des longitudes vraies. On a alors

$$(\frac{1}{\mathrm{D}})^n = \sum_{s=-\infty}^{\infty} b_{\{|s|,\,n\}}(\rho)\, e^{(I\,s\,(w_i-w_j))}$$

Puis on développe en inclinaisons en écrivant :

$$\frac{r_j}{\Delta} = \frac{1}{\sqrt{1+\rho^2-2\,\rho\cos(w_i-w_j)-2\,\rho\,(\cos(S)-\cos(w_i-w_j))}}$$

soit $\frac{r_j}{\Delta} = \frac{1}{\mathrm{D}}\frac{1}{\sqrt{1-\frac{\rho\,\mathrm{W}}{\mathrm{D}^2}}}$ avec $\mathrm{W} = 2(\cos(S)-\cos(w_i-w_j))$. Rappelons qu'on a développé cette quantité et ses puissances sous le nom *DDELTACOSS* et que c'est une fonction de degré 2 au moins en inclinaisons . On a alors le développement :

$$\frac{r_j}{\Delta} = \frac{1}{\mathrm{D}}\sum_k \frac{[\frac{1}{2}]_k\,\rho^k\,\mathrm{W}^k}{[1]_k\,\mathrm{D}^{(2\,k)}}$$

Finalement le développement de l'inverse de la distance s'organise sous la forme :

$$\frac{a_j}{\Delta} = \frac{a_j}{r_j}\sum_s\left(\sum_m\left(\sum_k \mathrm{C}(s,\,k,\,m,\,\alpha)\,(\frac{r_i}{a_i})^{(|s|+k)}\,(\frac{a_j}{r_j})^{(|s|+k)}\,\mathrm{W}^k\right)\,\delta^m\right)\,(\frac{\theta_i}{\theta_j})^s\,\exp\left(I\,s\,(L_i-L_j)\right)$$

où le coefficient est la fonction de α donnée par la formule :

$$\mathrm{C}(s,\,k,\,m,\,\alpha) = \frac{[\frac{1}{2}]_k\,\alpha^k\,\phi(|s|\,,\,2\,k+1,\,m,\,\alpha)}{[1]_k}$$

avec :

$$\phi(j,\,n,\,m,\,\alpha) = \frac{[\frac{n}{2}]_j\,[\frac{n}{2}]_m\,[\frac{n}{2}+j]_m\,\alpha^{(j+2\,m)}\,\mathrm{F}(\frac{n}{2}+m,\,\frac{n}{2}+j+m,\,1+j+m,\,\alpha^2)}{[1]_j\,[1]_m\,[1+j]_m}$$

.

Pour développer à un degré *Dmax* en excentricités et en inclinaisons, la somme sur m doit être calculée de $m=0$ à $m=Dmax$ et celle sur k de $k=0$ à $k=Dmax/2$. La somme sur s concerne l'ensemble des entiers relatifs.

Notons que les coefficients de Laplace proprement dits n'apparaissent explicitement dans cette expression que pour $m=0$; les fonctions ϕ sont des fonctions des dérivées de ces coefficients au voisinage d'une valeur de α .

On ne propose ici que la détermination d'une inégalité queconque dans ce développement.

3.4.5. Calcul par inégalités

On relit les fichiers utiles pour cette section, plus ceux utilisés pour obtenir le développement en polynômes de Legendre (afin de pouvoir comparer les résultats)

```
> read cat(chemin,`initmo.pr`):
> read cat(chemin,`Xnm10.m`);
> read cat(chemin,`DELTACOSSYm1a5.m`):
> read cat(chemin,`IngUDLeg.pr`):  with(orthopoly,P):
```

```
> read cat(chemin,‘COSSYm1a10.m‘):
```

En notant β la quantité $\frac{r_i}{a_i}\frac{a_j}{r_j}$, le développement de l'inverse de la distance au degré *Dmax* est formellement donné par les sommes sur m et k données ci-dessus, avec en facteur pour chaque valeur de s, la quantité $(\frac{\theta_i}{\theta_j})^s\ \exp\left(I\, s\,(L_i - L_j)\right)$ égale aussi à $(\frac{\theta_i}{\theta_j})^s\,\sigma^s$ (avec la notation σ déjà introduite dans les développements en inclinaisons).

On obtient par exemple au degré 3, pour une valeur quelconque de s, en ne tenant pas compte du facteur commun $\beta^{|s|}$ dans le développement de UJ:

```
> m:='m':k:='k':s:='s':  Dmax:=3:
> UJ:=expand(sum(sum(C[s,k,m]*beta^(k)*(beta^2-1)^m*W^k,k=0..Dmax/2),m=
> 0..Dmax)):  nops(%);
> Uj:=beta^(abs(s))*collect(combine(UJ,power),W); nops(%);
```

$$20$$

$$Uj := \beta^{|s|}((-C_{s,1,1}\,\beta - 2\,C_{s,1,2}\,\beta^3 + C_{s,1,1}\,\beta^3 + C_{s,1,2}\,\beta^5 + C_{s,1,0}\,\beta + C_{s,1,2}\,\beta + C_{s,1,3}\,\beta^7 - 3\,C_{s,1,3}\,\beta^5 + 3\,C_{s,1,3}\,\beta^3 - C_{s,1,3}\,\beta)\mathrm{W} + C_{s,0,0} + 3\,C_{s,0,3}\,\beta^2 + C_{s,0,1}\,\beta^2 - C_{s,0,1} + C_{s,0,2} - 3\,C_{s,0,3}\,\beta^4 + C_{s,0,2}\,\beta^4 - 2\,C_{s,0,2}\,\beta^2 + C_{s,0,3}\,\beta^6 - C_{s,0,3})$$

$$2$$

En fixant une valeur pour s et en choisissant une puissance de W , on obtient une expression directement utilisable pour la détermination d'une inégalité dont la caractéristique impliquerait cette valeur de s :

```
> subs(s=-1,coeff(Uj,W,0));
```

$$\beta^{|-1|}\,(C_{-1,0,0} + 3\,C_{-1,0,3}\,\beta^2 + C_{-1,0,1}\,\beta^2 - C_{-1,0,1} + C_{-1,0,2} - 3\,C_{-1,0,3}\,\beta^4 + C_{-1,0,2}\,\beta^4 - 2\,C_{-1,0,2}\,\beta^2 + C_{-1,0,3}\,\beta^6 - C_{-1,0,3})$$

```
> collect(eval(%),beta);
```

$$C_{-1,0,3}\,\beta^7 + (C_{-1,0,2} - 3\,C_{-1,0,3})\,\beta^5 + (C_{-1,0,1} + 3\,C_{-1,0,3} - 2\,C_{-1,0,2})\,\beta^3 + (C_{-1,0,0} - C_{-1,0,3} - C_{-1,0,1} + C_{-1,0,2})\,\beta$$

On peut alors proposer la procédure suivante, *INVDLAP* , pour déterminer l'inégalité (p, q) en degrés compris entre *Dmin* et *Dmax* dans le développement de l'inverse de la distance en coefficients de Laplace. On passe aussi en paramètre le développement formel en variables β et W, limité au degré *Dmax* en excentricités et inclinaisons, ainsi que le nom générique de la variable qui représente les diverses puissances de W.

```
> INVDLAP:=proc(p,q,Dmin,Dmax,DbetaW,PuissW,XXnm) local
> ic,ip,r,mon,nmon,m1,m2,mono,n1,n2,n3,n4,n5,n6,n7,n8,k1,k2,k,pkb,pk,cf
> Pk,t,ncfPk,
> moncos,nmoncos,monoc,ns,c,nti,ntj,c1,c2,c3,cfb,ldb,nni,nnj,j,js,di:
> ic:=p+q:  r:=0:    ip:=0:
> mon:=initmo(Dmin,Dmax,ic,ip,2):  nmon:=nops(mon):
> for m1 from 1 to nmon do:
> c:=0:
> mono:=op(m1,mon):
> n1:=op(3,mono):n2:=op(4,mono):n3:=op(5,mono):n4:=op(6,mono):
> n5:=op(7,mono):n6:=op(8,mono):n7:=op(9,mono):n8:=op(10,mono):
> j:= p +n4-n3+n8-n7:
> di:=(n5+n6+n7+n8)/2:
> for k2 from 0 to di do:
> moncos:=COEFTTS(PuissW.k2,n7,n8,n5,n6); nmoncos:=nops(moncos):
> pkb:=coeff(DbetaW,W,k2):
> for m2 from 1 to nmoncos do:
> monoc:=op(m2,moncos):    c3:=op(4,monoc):
> if c3 <> 0 then
```

```
> ns:=op(3,monoc):
> js:=j-ns:
> nti:=js+op(1,monoc):  ntj:=op(2,monoc)-js:
> pk:=collect(eval(subs(s=js,pkb)),beta):
> cfPk:=[coeffs(pk,beta,'t')]:  ncfPk:=nops(cfPk):
> for k1 from 1 to ncfPk do:
> cfb:=op(k1,[t]):
> ldb:=degree(cfb,beta)+abs(js):
> nni:=ldb:  nnj:=-(ldb+1):
> c1:=COEFXXb(XXnm,nni,nti,n3,n4):
> c2:=COEFXXb(XXnm,nnj,ntj,n1,n2);
> c:=c+c1*c2*c3*op(k1,cfPk):
> od:
> fi :
> od:
> od:
> r:=r+c*op(11,mono):
> od:
> RETURN(r);
> end:
```

Initialisation des calculs de développement d'inégalités à un degré *degmax* donné. On tronque les séries à ce degré avant leur utilisation, pour diminuer les temps de calcul.

```
> IniINEGLap:=proc(degmin,degmax) local k,r:
> r:=table():
```

```
> r[1]:=TRONC(Xnm,degmax,[X,Xb]);
> r[2]:=1:
> for k from 1 to degmax/2 do:
> r[k+2]:=TRONC(DDELTACOSSY.k,degmax,[Y[i],Yb[i],Y[j],Yb[j]]):
> od:
> RETURN([seq(r[k],k=1..degmax/2+2)]):
> end:
```

Les deux procédures suivantes effectuent le calcul des fonctions de α, tronquées ici au degré *da* pour comparaison avec le résultat complet obtenu par les polynômes de Legendre. L'option *remember* permet conserver le résultat de chaque évaluation, pour éviter de le refaire plusieurs fois :

```
> PHIalphaT:=proc(j,n,m,alpha) option remember:  local c:
> c:=pochhammer(n/2,j)*pochhammer(n/2,m)*pochhammer(n/2+j,m)/(j!*m!*poc
> hhammer(1+j,m));
> eval(c*alpha^(2*m+j)*mtaylor(hypergeom([n/2+m,n/2+j+m],[1+j+m],alpha^
> 2),alpha,da+1));
> end:
```

```
> CalphaT:=proc(s,k,m,alpha) option remember:
> TRONC(expand(pochhammer(1/2,k)*alpha^k/k!*PHIalphaT(abs(s),2*k+1,m,al
> pha)),da,alpha):
> end:
```

Pour un calcul non tronqué en α , ou un calcul numérique pour α donné, utiliser les procédures suivantes:

```
> PHIalpha:=proc(j,n,m,alpha) option remember:  local c:
> c:=pochhammer(n/2,j)*pochhammer(n/2,m)*pochhammer(n/2+j,m)/(j!*m!*poc
> hhammer(1+j,m));
> eval(c*alpha^(2*m+j)*hypergeom([n/2+m,n/2+j+m],[1+j+m],alpha^2));
```

```
> end:

> Calpha:=proc(s,k,m,alpha) option remember:
> expand(pochhammer(1/2,k)*alpha^k/k!*PHIalpha(abs(s),2*k+1,m,alpha)):
> end:

> alias(F=hypergeom):
```

Lorsque le paramètre formel s dans Calpha est non numérique, l'expression qu'on obtient est très compliquée.

Pour un calcul permettant de différer l'évaluation du coefficient venant en facteur de la fonction de α , on peut encore écrire ce calcul sous la forme :

```
> Clap:=proc(s,k,m) local h:  h:=(2*k+1)/2:
> pochhammer(1/2,k)/k!* pochhammer(h,s)/s!* pochhammer(h,m)/m!*
> pochhammer(h+s,m)/pochhammer(1+s,m):
> end;
```

$Clap :=$ **proc**(s, k, m)
local h;
$h := k + 1/2$;
$\mathrm{pochhammer}(1/2, k) \times \mathrm{pochhammer}(h, s) \times \mathrm{pochhammer}(h, m) \times \mathrm{pochhammer}(h + s, m)/(k! \times s! \times m! \times \mathrm{pochhammer}(s + 1, m))$
end

```
> Calfa:=proc(s,k,m,alpha) local h,j:  h:=(2*k+1)/2:
> j:=abs(s):Clap(j,k,m)*alpha^(k+2*m+j)*hypergeom([h+m,h+j+m],[1+j+m],al
> pha^2):
> end;
```

$$\begin{aligned}&\mathit{Calfa} := \mathbf{proc}(s,\, k,\, m,\, \alpha)\\ &\mathbf{local}\, h,\, j;\\ &\quad h := k + 1/2\,;\, j := \mathrm{abs}(s)\,;\, \mathrm{Clap}(j,\, k,\, m) \times \alpha^{(k+2\times m+j)} \times \mathrm{F}([h+m,\, h+j+m],\, [1+j+m],\, \alpha^2)\\ &\mathbf{end}\end{aligned}$$

```
> Calf:=proc(s,k,m,alpha) local h,j:  h:=(2*k+1)/2:  j:=abs(s):
> Cl(j,k,m)*alpha^(k+2*m+j)*hypergeom([h+m,h+j+m],[1+j+m],alpha^2):
> end;
```

$$\begin{aligned}&\mathit{Calf} := \mathbf{proc}(s,\, k,\, m,\, \alpha)\\ &\mathbf{local}\, h,\, j;\\ &\quad h := k + 1/2\,;\, j := \mathrm{abs}(s)\,;\, \mathrm{Cl}(j,\, k,\, m) \times \alpha^{(k+2\times m+j)} \times \mathrm{F}([h+m,\, h+j+m],\, [1+j+m],\, \alpha^2)\\ &\mathbf{end}\end{aligned}$$

Cette dernière procédure laisse en outre le coefficient sous la forme $\mathrm{Cl}(j,k,m)$ qu'il suffit d'évaluer, lorsqu'on en a besoin, en faisant eval(subs(Cl=Clap,...)).

Par exemple:

```
> p:='p':  collect(Calpha(p,1,3,alpha),F); eval(subs(p=4,%));
```

$$\begin{aligned}&(\frac{3675}{256}\,\frac{\alpha^7\,\alpha^{|p|}\,\%1}{|p|\,!\,(|p|+1)\,(2+|p|)\,(3+|p|)\,\sqrt{\pi}} + \frac{385}{8}\,\frac{\alpha^7\,\alpha^{|p|}\,\%1\,|p|}{|p|\,!\,(|p|+1)\,(2+|p|)\,(3+|p|)\,\sqrt{\pi}}\\ &+ \frac{1505}{32}\,\frac{\alpha^7\,\alpha^{|p|}\,\%1\,|p|^2}{|p|\,!\,(|p|+1)\,(2+|p|)\,(3+|p|)\,\sqrt{\pi}} + \frac{35}{2}\,\frac{\alpha^7\,\alpha^{|p|}\,\%1\,|p|^3}{|p|\,!\,(|p|+1)\,(2+|p|)\,(3+|p|)\,\sqrt{\pi}}\\ &+ \frac{35}{16}\,\frac{\alpha^7\,\alpha^{|p|}\,\%1\,|p|^4}{|p|\,!\,(|p|+1)\,(2+|p|)\,(3+|p|)\,\sqrt{\pi}})\mathrm{F}([\frac{9}{2},\, \frac{9}{2}+|p|],\, [4+|p|],\, \alpha^2)\\ &\%1 := \Gamma(\frac{1}{2}+|p|)\end{aligned}$$

$$\frac{225225}{65536}\,\alpha^{11}\,\mathrm{F}([\frac{9}{2},\ \frac{17}{2}],\ [8],\ \alpha^2)$$

```
> Calfa(p,1,3,alpha);eval(subs(p=4,%));
```

$$\frac{35}{32}\,\frac{\mathrm{pochhammer}(\frac{3}{2},\ |p|)\,\mathrm{pochhammer}(\frac{3}{2}+|p|\,,\ 3)\,\alpha^{(7+|p|)}\,\mathrm{F}([\frac{9}{2},\ \frac{9}{2}+|p|],\ [4+|p|],\ \alpha^2)}{|p|\,!\,\mathrm{pochhammer}(|p|+1,\ 3)}$$

$$\frac{225225}{65536}\,\alpha^{11}\,\mathrm{F}([\frac{9}{2},\ \frac{17}{2}],\ [8],\ \alpha^2)$$

```
> Calf(p,1,3,alpha); eval(subs(p=4,%)); eval(subs(Cl=Clap,%%));
> eval(subs(p=4,%));
```

$$\mathrm{Cl}(|p|\,,\ 1,\ 3)\,\alpha^{(7+|p|)}\,\mathrm{F}([\frac{9}{2},\ \frac{9}{2}+|p|],\ [4+|p|],\ \alpha^2)$$

$$\mathrm{Cl}(4,\ 1,\ 3)\,\alpha^{11}\,\mathrm{F}([\frac{9}{2},\ \frac{17}{2}],\ [8],\ \alpha^2)$$

$$\frac{35}{32}\,\frac{\mathrm{pochhammer}(\frac{3}{2},\ |p|)\,\mathrm{pochhammer}(\frac{3}{2}+|p|\,,\ 3)\,\alpha^{(7+|p|)}\,\mathrm{F}([\frac{9}{2},\ \frac{9}{2}+|p|],\ [4+|p|],\ \alpha^2)}{|p|\,!\,\mathrm{pochhammer}(|p|+1,\ 3)}$$

$$\frac{225225}{65536}\,\alpha^{11}\,\mathrm{F}([\frac{9}{2},\ \frac{17}{2}],\ [8],\ \alpha^2)$$

Exemple d'utilisation: On fixe le degré utile, puis on tronque à ce degré les séries en excentricités et inclinaisons avec *IniINEGLap*. Le degré *da* en α n'est utile ici que pour la compaison avec les développements en polynômes de Legendre; il est fixé à 8 comme dans la section précédente:

```
> Dmin:=0; Dmax:=5; da:=8;
```

$$Dmin := 0$$
$$Dmax := 5$$
$$da := 8$$

On tronque les développements en inclinaisons et en excentricités au degré *Dmax* . On calcule aussi le développement formel *DbetaW* en fonction de ce degré, et où les coefficients fonctions de α sont désignés par la notation *Ca* . Le calcul est fait formellement en fonction de cette notation. C'est seulement ensuite que l'on choisit pour un résultat tronqué ou non en α , simplement en remplaçant *Ca* par *CalphaT* ou *Calpha* :

```
> deb0:=time():
> inineg:=IniINEGLap(Dmin,Dmax):
> XnmT:= op(1,inineg):
> for k from 0 to Dmax/2 do:  DELTACOSSYmT.k:=op(k+2,inineg) od:
> s:='s':  k:='k':  m:='m':
> Us:=expand(sum(sum(Ca(s,k,m,alpha)*beta^(k)*(beta^2-1)^m*W^k,k=0..Dmax
> /2),m=0..Dmax)):
> DbetaW:=combine(Us,power);
> print(time()-deb0):
```

$$
\begin{aligned}
DbetaW :=\ & \mathrm{Ca}(s,2,0,\alpha)\,\beta^2\,\mathrm{W}^2+\mathrm{Ca}(s,1,0,\alpha)\,\beta\,\mathrm{W}-\mathrm{Ca}(s,2,5,\alpha)\,\beta^2\,\mathrm{W}^2-\mathrm{Ca}(s,1,5,\alpha)\,\beta\,\mathrm{W}+\mathrm{Ca}(s,2,4,\alpha)\,\beta^2\,\mathrm{W}^2\\
&+\mathrm{Ca}(s,1,4,\alpha)\,\beta\,\mathrm{W}-\mathrm{Ca}(s,2,3,\alpha)\,\beta^2\,\mathrm{W}^2-\mathrm{Ca}(s,1,3,\alpha)\,\beta\,\mathrm{W}+\mathrm{Ca}(s,2,2,\alpha)\,\beta^2\,\mathrm{W}^2+\mathrm{Ca}(s,1,2,\alpha)\,\beta\,\mathrm{W}\\
&-\mathrm{Ca}(s,2,1,\alpha)\,\beta^2\,\mathrm{W}^2-\mathrm{Ca}(s,1,1,\alpha)\,\beta\,\mathrm{W}+\mathrm{Ca}(s,2,3,\alpha)\,\beta^8\,\mathrm{W}^2+3\,\mathrm{Ca}(s,1,3,\alpha)\,\beta^3\,\mathrm{W}-3\,\mathrm{Ca}(s,1,3,\alpha)\,\beta^5\,\mathrm{W}\\
&+\mathrm{Ca}(s,1,3,\alpha)\,\beta^7\,\mathrm{W}-2\,\mathrm{Ca}(s,2,2,\alpha)\,\beta^4\,\mathrm{W}^2+\mathrm{Ca}(s,2,2,\alpha)\,\beta^6\,\mathrm{W}^2-2\,\mathrm{Ca}(s,1,2,\alpha)\,\beta^3\,\mathrm{W}+\mathrm{Ca}(s,1,2,\alpha)\,\beta^5\,\mathrm{W}\\
&+\mathrm{Ca}(s,2,1,\alpha)\,\beta^4\,\mathrm{W}^2+\mathrm{Ca}(s,1,1,\alpha)\,\beta^3\,\mathrm{W}-4\,\mathrm{Ca}(s,2,4,\alpha)\,\beta^4\,\mathrm{W}^2+6\,\mathrm{Ca}(s,2,4,\alpha)\,\beta^6\,\mathrm{W}^2\\
&-4\,\mathrm{Ca}(s,2,4,\alpha)\,\beta^8\,\mathrm{W}^2+\mathrm{Ca}(s,2,4,\alpha)\,\beta^{10}\,\mathrm{W}^2-4\,\mathrm{Ca}(s,1,4,\alpha)\,\beta^3\,\mathrm{W}+6\,\mathrm{Ca}(s,1,4,\alpha)\,\beta^5\,\mathrm{W}\\
&-4\,\mathrm{Ca}(s,1,4,\alpha)\,\beta^7\,\mathrm{W}+\mathrm{Ca}(s,1,4,\alpha)\,\beta^9\,\mathrm{W}+3\,\mathrm{Ca}(s,2,3,\alpha)\,\beta^4\,\mathrm{W}^2-3\,\mathrm{Ca}(s,2,3,\alpha)\,\beta^6\,\mathrm{W}^2\\
&-5\,\mathrm{Ca}(s,1,5,\alpha)\,\beta^9\,\mathrm{W}+\mathrm{Ca}(s,1,5,\alpha)\,\beta^{11}\,\mathrm{W}+10\,\mathrm{Ca}(s,1,5,\alpha)\,\beta^7\,\mathrm{W}-10\,\mathrm{Ca}(s,1,5,\alpha)\,\beta^5\,\mathrm{W}\\
&+\mathrm{Ca}(s,2,5,\alpha)\,\beta^{12}\,\mathrm{W}^2+5\,\mathrm{Ca}(s,1,5,\alpha)\,\beta^3\,\mathrm{W}+10\,\mathrm{Ca}(s,2,5,\alpha)\,\beta^8\,\mathrm{W}^2-5\,\mathrm{Ca}(s,2,5,\alpha)\,\beta^{10}\,\mathrm{W}^2\\
&+5\,\mathrm{Ca}(s,2,5,\alpha)\,\beta^4\,\mathrm{W}^2-10\,\mathrm{Ca}(s,2,5,\alpha)\,\beta^6\,\mathrm{W}^2+\mathrm{Ca}(s,0,0,\alpha)-\mathrm{Ca}(s,0,1,\alpha)-\mathrm{Ca}(s,0,3,\alpha)+\mathrm{Ca}(s,0,2,\alpha)\\
&-\mathrm{Ca}(s,0,5,\alpha)+\mathrm{Ca}(s,0,4,\alpha)+\mathrm{Ca}(s,0,5,\alpha)\,\beta^{10}-5\,\mathrm{Ca}(s,0,5,\alpha)\,\beta^8+10\,\mathrm{Ca}(s,0,5,\alpha)\,\beta^6-10\,\mathrm{Ca}(s,0,5,\alpha)\,\beta^4\\
&+5\,\mathrm{Ca}(s,0,5,\alpha)\,\beta^2+\mathrm{Ca}(s,0,2,\alpha)\,\beta^4-2\,\mathrm{Ca}(s,0,2,\alpha)\,\beta^2+\mathrm{Ca}(s,0,3,\alpha)\,\beta^6-3\,\mathrm{Ca}(s,0,3,\alpha)\,\beta^4\\
&+3\,\mathrm{Ca}(s,0,3,\alpha)\,\beta^2+\mathrm{Ca}(s,0,4,\alpha)\,\beta^8-4\,\mathrm{Ca}(s,0,4,\alpha)\,\beta^6+6\,\mathrm{Ca}(s,0,4,\alpha)\,\beta^4-4\,\mathrm{Ca}(s,0,4,\alpha)\,\beta^2\\
&+\mathrm{Ca}(s,0,1,\alpha)\,\beta^2
\end{aligned}
$$

$$.069$$

Initialisations analogues pour la comparaison avec les résultats utilisant les polynômes de Legendre :

```
> deb0:=time():
> inineg:=IniINEGLeg(Dmin,Dmax,da):
> XnmT:= op(1,inineg):
> k:='k':  for k from 0 to da do:  COSSYmT.k:=op(k+2,inineg) od:
> som2alfa:=collect(expand(sum(alpha^n*P(n,x),n=0..da)),x):
> print(time()-deb0):
```

$$5.260$$

Il suffit ensuite de définir les entiers (p, q) représentant une inégalité, et d'exécuter *INVDLEG* :

```
> pp:=1; qq:=-6;
```

$$pp := 1$$
$$qq := -6$$

Calcul en coefficients de Laplace :

```
> deb:=time():
> UD1:=INVDLAP(pp,qq,Dmin,Dmax,DbetaW,DELTACOSSYmT,XnmT):
> UD1:=sort(collect(UD1,varxyij,distributed),varxyij):  op(1,%);
> print(nops(%),"duree=",time()-deb):
> deb:=time():
```

On remplace ici *Ca* par *CalphaT* pour obtenir un résultat tronqué en α ($\Rightarrow$ UDTr).

```
> UDTr:=sort(collect(map(sort,eval(subs(Ca=CalphaT,UD1)),alpha),varxyij
> ,distributed),varxyij):  op(1,%);
> print(nops(%%),"duree=",time()-deb):
```

On remplace ici *Ca* par *Calpha* pour obtenir un résultat complet en α , analytique ($\Rightarrow$ UDa) si on ne donne pas de valeur à α , ou numérique ($\Rightarrow$ UDn) sinon.

```
> deb:=time():
> UDa:=sort(collect(eval(subs(Ca=Calpha,UD1)),varxyij,distributed),varx
> yij):  op(1,%);
> print(nops(%%),"duree=",time()-deb):
```

$$(-\frac{43171}{640}\,\mathrm{Ca}(6,\,0,\,0,\,\alpha)-\frac{29461}{384}\,\mathrm{Ca}(6,\,0,\,1,\,\alpha)-\frac{453}{16}\,\mathrm{Ca}(6,\,0,\,3,\,\alpha)-\frac{1397}{24}\,\mathrm{Ca}(6,\,0,\,2,\,\alpha)-\mathrm{Ca}(6,\,0,\,5,\,\alpha)-8\,\mathrm{Ca}(6,\,0,\,4,\,\alpha))\,{x_i}^5$$

$$2,\ \text{“duree=”},\ .621$$

$$(-\frac{19792487}{1310720}\,\alpha^8-\frac{9972501}{655360}\,\alpha^6)\,{x_i}^5$$

$$28,\ \text{“duree=”},\ 1.189$$

$$(-\frac{9972501}{655360}\,\alpha^6\,\mathrm{F}([\frac{1}{2},\,\frac{13}{2}],\,[7],\,\alpha^2)-\frac{4212923}{524288}\,\alpha^8\,\mathrm{F}([\frac{3}{2},\,\frac{15}{2}],\,[8],\,\alpha^2)-\frac{27531075}{16777216}\,\alpha^{12}\,\mathrm{F}([\frac{7}{2},\,\frac{19}{2}],\,[10],\,\alpha^2)$$
$$-\frac{8989695}{2097152}\,\alpha^{10}\,\mathrm{F}([\frac{5}{2},\,\frac{17}{2}],\,[9],\,\alpha^2)-\frac{5555277}{134217728}\,\alpha^{16}\,\mathrm{F}([\frac{23}{2},\,\frac{11}{2}],\,[12],\,\alpha^2)-\frac{1616615}{4194304}\,\alpha^{14}\,\mathrm{F}([\frac{21}{2},\,\frac{9}{2}],\,[11],\,\alpha^2)){x_i}^5$$

$$28,\ \text{“duree=”},\ .551$$

On donne ici une valeur numérique à α et on évalue les fonctions de α.

```
> deb:=time():
> alpha0:= 0.628729981643458 ;   Precision:=18:
> UDn:=subs(Ca =Calpha,alpha=alpha0,UD1):
> UDn:=sort(collect(evalf(UDn,Precision),varxyij,distributed),varxyij);
> print("duree=",time()-deb):
```

$$\alpha 0 := .628729981643458$$

$$UDn := -1.74985557495267939\, x_i^5 + 12.0994781728790396\, x_i^4\, x_j - 33.3406902989086577\, x_i^3\, x_j^2 - 10.5747300977608281\, x_i^3\, y_i^2 + 21.1494601955216566\, C_i\, C_j\, x_i^3\, y_i\, y_j - 10.5747300977608281\, x_i^3\, y_j^2 + 45.7464335452827679\, x_i^2\, x_j^3 + 50.3651710907754762\, x_i^2\, x_j\, y_i^2 - 100.730342181550951\, C_i\, C_j\, x_i^2\, x_j\, y_i\, y_j + 50.3651710907754762\, x_i^2\, x_j\, y_j^2 - 31.2169216600494532\, x_i\, x_j^4 - 81.1018580051226890\, x_i\, x_j^2\, y_i^2 + 162.203716010245376\, C_i\, C_j\, x_i\, x_j^2\, y_i\, y_j - 81.1018580051226890\, x_i\, x_j^2\, y_j^2 - 8.18181286909976464\, x_i\, y_i^4 + 32.727251476399065\, C_i\, C_j\, x_i\, y_i^3\, y_j + (-16.3636257381995293 - 32.727251476399065\, C_i^2\, C_j^2)\, x_i\, y_i^2\, y_j^2 + 32.727251476399065\, C_i\, C_j\, x_i\, y_i\, y_j^3 - 8.18181286909976464\, x_i\, y_j^4 + 8.43560972854292515\, x_j^5 + 44.4201543324733570\, x_j^3\, y_i^2 - 88.8403086649467l0\, C_i\, C_j\, x_j^3\, y_i\, y_j + 44.4201543324733570\, x_j^3\, y_j^2 + 14.7867767320250431\, x_j\, y_i^4 - 59.14710692810017\, C_i\, C_j\, x_j\, y_i^3\, y_j + (29.5735534640500862 + 59.14710692810017\, C_i^2\, C_j^2)\, x_j\, y_i^2\, y_j^2 - 59.14710692810017\, C_i\, C_j\, x_j\, y_i\, y_j^3 + 14.7867767320250431\, x_j\, y_j^4$$

$$\text{“duree=”},\ .090$$

Calcul de la même inégalité à partir de la méthode des polynômes de Legendre (à condition que l'inégalité y soit présente)

```
> UD:=INVDLEG(pp,qq,Dmin,Dmax,da,COSSYmT,XnmT,som2alfa):
> UDLeg:=map(sort,sort(collect(UD,varxyij,distributed),varxyij),alpha):
> nops(%);
> print("duree=",time()-deb):
```

$$28$$

$$\text{“duree=”},\ 5.629$$

```
> UDTr-UDLeg:  simplify(%);
```

$$0$$

Instructions pour comparer terme à terme ces 2 développements par visualisation directe (changer comme on veut le 1er paramètre de la fonction op(xxx,UDTr) suivante, mais xxx doit être inférieur à nops(UDTr))

```
> term:=op(10,UDTr);
> nxi:=degree(term,x[i]):  nxj:=degree(term,x[j]):
> nxbi:=degree(term,xb[i]):  nxbj:=degree(term,xb[j]):
> nyi:=degree(term,y[i]):  nyj:=degree(term,y[j]):
> nybi:=degree(term,yb[i]):  nybj:=degree(term,yb[j]):
> coeff(coeff(coeff(coeff(coeff(coeff(coeff(coeff(UDLeg,x[i],nxi),xb[i]
> ,nxbi),x[j],nxj),xb[j],nxbj),y[i],nyi),yb[i],nybi),y[j],nyj),yb[j],nyb
> j);
```

$$term := (\frac{2488563}{8192}\,\alpha^7 + \frac{28665}{256}\,\alpha^5)\,{x_i}^2\,x_j\,{y_j}^2$$

$$\frac{2488563}{8192}\,\alpha^7 + \frac{28665}{256}\,\alpha^5$$

3.5. Fonction perturbatrice d'une planète

On reconstitue la fonction perturbatrice :

$$K\,m_j\,(\frac{1}{\Delta} - \frac{r_i\cos(S)}{{r_j}^2})$$

à partir des développements construits en section 3.3 et 3.4. Si on peut utiliser le développement en polynômes de Legendre, on obtient un développement complet à un degré donné mais modéré, sinon, on construit la fonction perturbatrice inégalité par inégalité.

Ce paragraphe permet aussi de calculer les équations de Lagrange avec la fonction perturbatrice calculée ici; les équations de Lagrange seront vues plus loin dans la section 5.

3.5.1. calcul explicite complet (degré faible)

```
> read cat(chemin,`EqLag.pr`):
> da:=8; degmx:= 4;
```

$$da := 8$$
$$degmx := 4$$

Lecture de $\frac{a_j}{\Delta}$ (où l'on a implicitement supposé que la planète j était extérieure à la planète i , donc $a_i < a_j$):

```
> read  cat(chemin,`UDpxy`.da.`_`.degmx.`.m`):  nops(UDxyij);
> op(%/2,UDxyij);
```

$$9798$$

$$\frac{158025}{4096}\, eL_i\, {eL_j}^3\, \alpha^7\, {xb_j}^2\, {yb_i}^2$$

Lecture de la partie indirecte :

```
> read  cat(chemin,`Uind`.degmx.`.m`):  nops(Uindxyij);
> op(%/2,Uindxyij);
```

$$262$$

$$\frac{3}{2}\, \frac{x_i\, yb_i\, y_i\, x_j}{{eL_j}^2}$$

Si les indices des planètes vont en croissant avec leur distance au Soleil, pour $i < j$ on a $\alpha = \frac{a_i}{a_j} < 1$ et on a alors :

$$U_i = \frac{K\, m_j}{a_j}\left(\frac{a_j}{\Delta} - \alpha\, \frac{r_i}{a_i}\, (\frac{a_j}{r_j})^2 \cos(S)\right)$$

```
> Uij:=expand(UDxyij - alpha*Uindxyij):  nops(Uij);
```

$$9536$$

```
> UP[i]:=K*m[j]/a[j]*Uij:
```

Dans le même temps, la planète j est perturbée par la planète i , et on a pour elle :

$$U_j = \frac{K\,m_i}{a_j}\left(\frac{a_j}{\Delta} - \alpha^{-2}\,\frac{r_j}{a_j}\,(\frac{a_i}{r_i})^2\cos(S)\right)$$

Il faut donc inverser les indices i et j dans la partie indirecte telle qu'elle a été construite:

```
> Uji:=expand(UDxyij- 1/alpha^2*subs(i=ii,j=i,ii=j,Uindxyij)):
> nops(Uji);
```

$$10060$$

```
> UP[j]:=K*m[i]/a[j]*Uji:
```

```
> UPxyij:=[UP[i],UP[j]]:
> save UPxyij, cat(chemin,`UPxyij`.da.`_`.degmx.`.m`):
```

Pour relire ce fichier, il faut faire :

```
> read cat(chemin,`UPxyij8_4.m`):
> UP[i]:=expand(op(1,UPxyij)):  UP[j]:=expand(op(2,UPxyij)):
> nops(UP[i]);nops(UP[j]);
```

$$9536$$
$$10060$$

La forme suivante restitue les fonctions exponentielles, mais est assez longue à obtenir s'il y a beaucoup de termes

```
> # UPexp[i]:=combine(subs(eL[i]=exp(I*L[i]),eL[j]=exp(I*L[j]),UP[i]),exp):
> # UPexp[j]:=combine(subs(eL[i]=exp(I*L[i]),eL[j]=exp(I*L[j]),UP[j]),exp):
> # collect(TRONC(UPexp[i],2,varxyij),exp);
```

On peut d'ici calculer les équations de Lagrange relatives à cette fonction perturbatrice :
vers EqLag(UP)

3.5.2. Calcul par inégalités (par coefficients de Laplace)

```
> read cat(chemin,`initmo.pr`):
> read cat(chemin,`IngUind.pr`):
> read cat(chemin,`IngUDLap.pr`):
> read cat(chemin,`EqLag.pr`):
```

On fixe les degrés utiles et la puissance maximum souhaitée éventuellement pour α (si on désire un développement tronqué en α par l'utilisation de *CalphaT*) , puis on initialise les calculs avec *IniINEGLap* Anx2.4

```
> Dmin:=0; Dmax:=4; da:=8;
> deb0:=time():
> inineg:=IniINEGLap(Dmin,Dmax):
> XnmT:= op(1,inineg):
> COSSYm1T:=TRONC(COSSYm1,Dmax,[Y[i],Yb[i],Y[j],Yb[j]]):
> for k from 0 to Dmax/2 do:  DELTACOSSYmT.k:=op(k+2,inineg) od:
```

```
> s:='s': k:='k': m:='m':
> Us:=expand(sum(sum(Ca(s,k,m,alpha)*beta^(k)*(beta^2-1)^m*W^k,k=0..Dmax
> /2),m=0..Dmax)):
> DbetaW:=combine(Us,power): nops(%);
> print(time()-deb0):
```

$$Dmin := 0$$
$$Dmax := 4$$
$$da := 8$$
$$45$$
$$.041$$

On choisit une inégalité $pp\,L_i + qq\,L_j$ (pp associé à la planète intérieure et qq à celle extérieure) :

```
> pp:=2; qq:=-1;
```

$$pp := 2$$
$$qq := -1$$

Calcul de la partie indirecte concernant cette inégalité (**attention** aux deux cas planète intérieure et planète extérieure: ici on ne doit pas permuter les indices pour passer d'un cas à l'autre, sauf à permuter aussi pp et qq):

```
> deb:=time():
> ingINDi:=INEG(pp,qq,Dmin,Dmax,1,-2,COSSYm1T,XnmT):
> PRINTRONC(ingINDi,2,varxyij):
> ingINDj:=INEG(pp,qq,Dmin,Dmax,-2,1,COSSYm1T,XnmT):
> PRINTRONC(ingINDj,2,varxyij):
> UiPerturBParj:=alpha*ingINDi:
> UjPerturBPari:=1/alpha^2*ingINDj:
> print("duree=",time()-deb):
```

$$-\frac{3}{16}\,x_i\,{xb_i}^2,\ +\ldots+,\ \frac{1}{4}\,xb_i$$

$$-\frac{3}{4}\,x_i\,{xb_i}^2,\ +\ldots+,\ xb_i$$

$$\text{“duree=”},\ .389$$

Calcul de la partie directe en coefficients de Laplace :

```
> deb:=time():
> UD1:=INVDLAP(pp,qq,Dmin,Dmax,DbetaW,DELTACOSSYmT,XnmT):
> UD1:=sort(collect(UD1,varxyij,distributed),varxyij):  op(1,%);
> print(nops(%%),"duree=",time()-deb):
```

$$\left(-3\,\mathrm{Ca}(1,\,0,\,2,\,\alpha)+\frac{11}{8}\,\mathrm{Ca}(1,\,0,\,1,\,\alpha)-\frac{3}{8}\,\mathrm{Ca}(1,\,0,\,0,\,\alpha)-3\,\mathrm{Ca}(1,\,0,\,3,\,\alpha)\right)x_i\,{xb_i}^2$$

$$8,\ \text{“duree=”},\ .040$$

On montre dans ce qui suit comment utiliser ce développement, soit en évaluant les fonctions de α de manière tronquée, ou complètement:

Calcul du développement tronqué en α:

```
> deb:=time():
> UD2:=eval(subs(Ca=CalphaT,UD1)):
> UD2i:=sort(collect(map(sort,UD2-UiPerturBParj,alpha),varxyij,distribu
> ted),varxyij):  op(1,%);
> UPTr[i]:=K*m[j]/a[j]*UD2i*eL[i]^pp*eL[j]^qq:
```

```
> UD2j:=sort(collect(map(sort,UD2-UjPerturBPari,alpha),varxyij,distribu
> ted),varxyij):  op(1,%);
```

$$(-\frac{5775}{8192}\alpha^7 - \frac{75}{1024}\alpha^5 + \frac{3}{16}\alpha^3)\, x_i\, xb_i^{\,2}$$

$$(-\frac{5775}{8192}\alpha^7 - \frac{75}{1024}\alpha^5 + \frac{3}{16}\alpha^3 - \frac{3}{16}\alpha + \frac{3}{4}\frac{1}{\alpha^2})\, x_i\, xb_i^{\,2}$$

```
> UPTr[j]:=K*m[i]/a[j]*UD2j*eL[i]^pp*eL[j]^qq:
> print(nops(%%%),"duree=",time()-deb):
```

$$11,\ \text{"duree="},\ .041$$

On remplace ici *Ca* par *Calpha* pour obtenir un résultat complet en α , analytique si on ne donne pas de valeur à α , ou numérique sinon.

```
> deb:=time():
> UDa:=eval(subs(Ca=Calpha,UD1)):
> UDai:=sort(collect(UDa-UiPerturBParj,varxyij,distributed),varxyij):
> op(1,%);
> UPa[i]:=K*m[j]/a[j]*UDai*eL[i]^pp*eL[j]^qq:
> UDaj:=sort(collect(UDa-UjPerturBPari,varxyij,distributed),varxyij):
> op(1,%);
> UPa[j]:=K*m[i]/a[j]*UDaj*eL[i]^pp*eL[j]^qq:
> print(nops(%%%),"duree=",time()-deb):
```

$$(-\frac{45}{128}\alpha^5\,\mathrm{F}([\frac{5}{2},\frac{7}{2}],[4],\alpha^2) + \frac{33}{128}\alpha^3\,\mathrm{F}([\frac{3}{2},\frac{5}{2}],[3],\alpha^2) - \frac{3}{16}\alpha\,\mathrm{F}([\frac{1}{2},\frac{3}{2}],[2],\alpha^2) - \frac{525}{2048}\alpha^7\,\mathrm{F}([\frac{9}{2},\frac{7}{2}],[5],\alpha^2) + \frac{3}{16}\alpha)\, x_i\, xb_i^{\,2}$$

$$(-\frac{45}{128}\alpha^5\,\mathrm{F}([\frac{5}{2},\frac{7}{2}],[4],\alpha^2) + \frac{33}{128}\alpha^3\,\mathrm{F}([\frac{3}{2},\frac{5}{2}],[3],\alpha^2) - \frac{3}{16}\alpha\,\mathrm{F}([\frac{1}{2},\frac{3}{2}],[2],\alpha^2) - \frac{525}{2048}\alpha^7\,\mathrm{F}([\frac{9}{2},\frac{7}{2}],[5],\alpha^2) + \frac{3}{4}\frac{1}{\alpha^2})\, x_i\, xb_i^{\,2}$$

$$11,\ \text{"duree="},\ 0$$

Pour la visualisation du caractère analytique, on ne considère que la partie de degré 2 de cette inégalité :

```
> TRONC(expand(UPa[i]),2,varxyij); TRONC(expand(UPa[j]),2,varxyij);
```

$$\left(-\frac{3}{16}\,\frac{K\,m_j\,eL_i{}^2\,\alpha^3\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)}{a_j\,eL_j}-\frac{1}{4}\,\frac{K\,m_j\,eL_i{}^2\,\alpha}{a_j\,eL_j}+\frac{1}{4}\,\frac{K\,m_j\,eL_i{}^2\,\alpha\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2)}{a_j\,eL_j}\right)xb_i$$
$$+\left(\frac{5}{32}\,\frac{K\,m_j\,eL_i{}^2\,\alpha^4\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2)}{a_j\,eL_j}-\frac{3}{16}\,\frac{K\,m_j\,eL_i{}^2\,\alpha^2\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)}{a_j\,eL_j}\right)xb_j$$

$$\left(-\frac{3}{16}\,\frac{K\,m_i\,eL_i{}^2\,\alpha^3\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)}{a_j\,eL_j}+\frac{1}{4}\,\frac{K\,m_i\,eL_i{}^2\,\alpha\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2)}{a_j\,eL_j}-\frac{K\,m_i\,eL_i{}^2}{a_j\,\alpha^2\,eL_j}\right)xb_i$$
$$+\left(\frac{5}{32}\,\frac{K\,m_i\,eL_i{}^2\,\alpha^4\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2)}{a_j\,eL_j}-\frac{3}{16}\,\frac{K\,m_i\,eL_i{}^2\,\alpha^2\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)}{a_j\,eL_j}\right)xb_j$$

Calcul numérique pour une valeur donnée de α :

```
> deb:=time():
> alpha0:= 0.628729981643458 ;   Precision:=18:
> UDni:=subs(Ca=Calpha,alpha=alpha0,UD1-UiPerturBParj):
> UDnj:=subs(Ca=Calpha,alpha=alpha0,UD1-UjPerturBPari):
> UPn[i]:=K*m[j]/a[j]*
> sort(collect(evalf(UDni,Precision),varxyij,distributed),varxyij)*eL[i]
> ^pp*eL[j]^qq:  factor(TRONC(expand(UPn[i]),2,varxyij));
```

```
> UPn[j]:=K*m[i]/a[j]*
> sort(collect(evalf(UDnj,Precision),varxyij,distributed),varxyij)*eL[i]
> ^pp*eL[j]^qq:  factor(TRONC(expand(UPn[j]),2,varxyij));
> print("duree=",time()-deb):
```

$$\alpha 0 := .628729981643458$$

$$-\frac{K\, m_j\, {eL_i}^2\, (.05504701330\, xb_i + .04412579150\, xb_j)}{a_j\, eL_j}$$

$$-\frac{K\, m_i\, {eL_i}^2\, (2.427579884\, xb_i + .04412579149\, xb_j)}{a_j\, eL_j}$$

"duree=", .029

Pour garder une certaine analyticité par rapport aux demi-grands axes, on peut aussi développer les fonctions de α au voisinage d'une valeur α_0 : La fonction *taylor* ayant des difficultés à calculer au voisinage d'une valeur non nulle, on reprogramme ce calcul à l'ordre *n* avec une précision de *p* digits:

```
> TAYLORf:=proc(f,x,x0,n,p) local s,k,f1:
> s:=evalf(subs(x=x0,f),p):  f1:=f:
> for k from 1 to n do :
> f1:=diff(f1,x):  s:=s+(x-x0)^k/k!*evalf(subs(x=x0,f1),p):
> od:
> RETURN(s):
> end;
```

$$
\begin{array}{l}
\mathit{TAYLORf} := \textbf{proc}(f,\, x,\, x0,\, n,\, p)\\
\textbf{local}\; s,\, k,\, f1;\\
\quad s := \mathrm{evalf}(\mathrm{subs}(x = x0,\, f),\, p)\,;\\
\quad f1 := f\,;\\
\quad \textbf{for}\; k\; \textbf{to}\; n\; \textbf{do}\; f1 := \mathrm{diff}(f1,\, x)\,;\; s := s + (x - x0)^k \times \mathrm{evalf}(\mathrm{subs}(x = x0,\, f1),\, p)/k!\; \textbf{od}\,;\\
\quad \mathrm{RETURN}(s)\\
\textbf{end}
\end{array}
$$

Pour l'usage dans les équations de Lagrange à l'ordre 1 des masses, il suffit de développer à l'ordre $n = 1$:

```
> UPna[i]:=TAYLORf(UPa[i],alpha,alpha0,1,Precision);
```

$$
\begin{aligned}
UPna_i := {} & K\, m_j(-.0554806993101901 40\, x_i\, xb_i^2 + .442184385376296847\, x_i\, xb_i\, xb_j - .188420221114353984\, x_i\, xb_j^2\\
& + .143872057580334582\, xb_i^2\, x_j - .450851214079232182\, xb_i\, x_j\, xb_j + .157182495410864500\, xb_i\, y_i\, yb_i\\
& - .314364990821729000\, C_i\, C_j\, xb_i\, yb_i\, y_j + .157182495410864500\, xb_i\, y_j\, yb_j + .249306636621938952\, x_j\, xb_j^2\\
& - .055047013307675651\, xb_i - .044125791490179338 6\, xb_j) eL_i^2 \Big/ (a_j\, eL_j) + ((\alpha - .628729981643458)\, K\, m_j(\\
& -1.39632118296485533\, x_i\, xb_i^2 + 4.41419832373478358\, x_i\, xb_i\, xb_j - 2.30977257976616104\, x_i\, xb_j^2\\
& + 2.30772959542576571\, xb_i^2\, x_j - 5.66355169656027571\, xb_i\, x_j\, xb_j + .250000000000000000\, xb_i\, y_i\, yb_i\\
& - .500000000000000000\, C_i\, C_j\, xb_i\, yb_i\, y_j + .250000000000000000\, xb_i\, y_j\, yb_j + 3.07218944374175051\, x_j\, xb_j^2\\
& - .443102338225822874\, xb_i + .057503104890499379\, xb_j) eL_i^2) \Big/ (a_j\, eL_j)
\end{aligned}
$$

```
> UPna[i]:=collect(collect(expand(
> %),varxyij,distributed),[K,m[i],m[j],a[i],a[j],eL[i],eL[j]],distribut
> ed);
```

$$UPna_i := (((3.07218944374175051\,\alpha - 1.682270976)\,x_j\,{xb_j}^2 + (-2.30977257976616104\,\alpha + 1.263803051)\,x_i\,{xb_j}^2 - .500000000000000000\,C_i\,C_j\,\alpha\,xb_i\,yb_i\,y_j + (-2.333154446 + 4.41419832373478358\,\alpha)\,x_i\,xb_i\,xb_j + (2.30772959542576571\,\alpha - 1.307066728)\,{xb_i}^2\,x_j + (.2235447116 - .443102338225822874\,\alpha)\,xb_i + (-.08027971757 + .0575031048904993379\,\alpha)\,xb_j + (-1.39632118296485533\,\alpha + .8224282924)\,x_i\,{xb_i}^2 + .250000000000000000\,\alpha\,xb_i\,y_i\,yb_i + .250000000000000000\,\alpha\,xb_i\,y_j\,yb_j + (-5.66355169656027571\,\alpha + 3.109993540)\,xb_i\,x_j\,xb_j)K\,m_j\,{eL_i}^2)\Big/(a_j\,eL_j)$$

```
> UPna[j]:=TAYLORf(UPa[j],alpha,alpha0,1,Precision);
```

$$UPna_j := K\,m_i(1.72391895358956110\,x_i\,{xb_i}^2 + .442184385376296847\,x_i\,xb_i\,xb_j - .188420221114353984\,x_i\,{xb_j}^2 + .143872057580334582\,{xb_i}^2\,x_j + .735415221187268638\,xb_i\,x_j\,xb_j + 2.52971536594386614\,xb_i\,y_i\,yb_i - 5.05943073188773228\,C_i\,C_j\,xb_i\,yb_i\,y_j + 2.52971536594386614\,xb_i\,y_j\,yb_j + .249306636621938952\,x_j\,{xb_j}^2 - 2.42757988384067729\,xb_i - .0441257914901793386\,xb_j)eL_i{}^2\Big/(a_j\,eL_j) + ((\alpha - .628729981643458)\,K\,m_i(-7.61911957767006690\,x_i\,{xb_i}^2 + 4.41419832373478358\,x_i\,xb_i\,xb_j - 2.30977257976616104\,x_i\,{xb_j}^2 + 2.30772959542576571\,{xb_i}^2\,x_j - 9.81208395969708342\,xb_i\,x_j\,xb_j - 8.04706452627361542\,xb_i\,y_i\,yb_i + 16.0941290525472308\,C_i\,C_j\,xb_i\,yb_i\,y_j - 8.04706452627361542\,xb_i\,y_j\,yb_j + 3.07218944374175051\,x_j\,{xb_j}^2 + 7.85396218804779255\,xb_i + .0575031048904993379\,xb_j)\,eL_i{}^2)\Big/(a_j\,eL_j)$$

```
> UPna[j]:=collect(collect(expand(
> %),varxyij,distributed),[K,m[i],m[j],a[i],a[j],eL[i],eL[j]],distribut
> ed);
```

$$UPna_j := (((3.07218944374175051\,\alpha - 1.682270976)\,x_j\,{xb_j}^2 + (-2.30977257976616104\,\alpha + 1.263803051)\,x_i\,{xb_j}^2$$
$$+ (16.0941290525472308\,\alpha\,C_i\,C_j - 15.17829219\,C_i\,C_j)\,y_j\,yb_i\,xb_i$$
$$+ (-2.333154446 + 4.41419832373478358\,\alpha)\,x_i\,xb_i\,xb_j + (2.30772959542576571\,\alpha - 1.307066728)\,{xb_i}^2\,x_j$$
$$+ (-7.365601386 + 7.85396218804779255\,\alpha)\,xb_i + (-.08027971757 + .0575031048904993790\,\alpha)\,xb_j$$
$$+ (6.514287866 - 7.61911957767006690\,\alpha)\,x_i\,{xb_i}^2 + (7.589146097 - 8.04706452627361542\,\alpha)\,xb_i\,y_i\,yb_i$$
$$+ (7.589146097 - 8.04706452627361542\,\alpha)\,xb_i\,y_j\,yb_j + (-9.81208395969708342\,\alpha + 6.904566589)\,xb_i\,x_j\,xb_j)$$
$$K\,m_i\,{eL_i}^2) \Big/ (a_j\,eL_j)$$

vers EqLag(UPa) (Utilisation des équations de Lagrange avec la forme analytique en α (UPa) de la fonction perturbatrice)

vers EqLag(UPTr) (Utilisation des équations de Lagrange avec la forme analytique tronquée en α (UPTr) de la fonction perturbatrice)

vers EqLag(UPn) (Utilisation des équations de Lagrange avec la forme UPna (semi-numérique en α) de la fonction perturbatrice)

4. Développements de fonctions perturbatrices de satellites

4.1. Introduction

On se propose de construire le développement analytique de la fonction perturbatrice d'un satellite orbitant une planète non sphérique. On suppose cependant que l'on recherche ces développements en fonction des éléments osculateurs de l'orbite du satellite et des angles donnant l'orientation du repère lié à la planète. On utilise pour cela les développements du mouvement keplérien et ceux en inclinaisons introduits dans la section 2. Dans le cas du mouvement planétocentrique d'un satellite S_i de masse μ_i autour d'une planète C(4.24)

P non sphérique de masse M, la fonction perturbatrice directe du satellite est développée en fonctions harmoniques sphériques sous la forme :

$$\frac{K\,(M+\mu_i)}{r_i}\sum_{n=2}^{\infty}\left(\sum_{p=0}^{n}\left(\frac{a_e}{r_i}\right)^n \cos^p\delta_i\, Q_{n,\,p}(\sin(\delta_i))\,(a_{n,\,p}\,\exp\,(I\,p\,\beta_i)+b_{n,\,p}\,\exp\,(-I\,p\,\beta_i))\right)$$

où a_e est le rayon équatorial de la planète, r_i, δ_i et β_i les coordonnées sphériques équatoriales du satellite dans un repère équatorial lié à la planète, et où $a_{n,\,p}$ et $b_{n,\,p}$ sont des coefficients d'aplatissement de la planète, reliés aux coefficients classiques J_n, $c_{n,\,p}$ et $s_{n,\,p}$ par des relations explicitées en 4.3 . Pour obtenir ce développement, on a utilisé notamment la propriété des fonctions associées de Legendre de pouvoir s'écrire:

$$P_{n,\,p}(\sin(\delta)) = \cos^p\delta_i\, Q_{n,\,p}(\sin(\delta_i))$$

ce qui permet de mettre en évidence les puissances p de $[\cos(\delta_i)\,\exp\,(I\,\beta_i)]$ ou de $[\cos(\delta_i)\,\exp\,(-I\,\beta_i)]$, fonctions qui ont été développées en 2.4.

La masse μ_i du satellite intervient dans cette expression par le facteur $M+\mu_i$ si l'on considère le mouvement planétocentrique du satellite. En effet, le potentiel de gravitation entre la planète et son satellite s'écrit initialement:

$$U = M\,\mu_i\,\mathrm{V}(r_i,\,\delta_i,\,\beta_i) = \frac{K\,M\,\mu_i}{r_i}+\frac{K\,M\,\mu_i}{r_i}\sum_{n=2}^{\infty}\left(\sum_{p=0}^{n}\left(\frac{a_e}{r_i}\right)^n \cos^p\delta_i\, Q_{n,\,p}(\sin(\delta_i))\,(a_{n,\,p}\,\exp\,(I\,p\,\beta_i)+b_{n,\,p}\,\exp\,(-I\,p\,\beta_i))\right)$$

où $r_i = PS_i = ||OS_i - OP||$ et où $\cos(\delta_i)\cos(\beta_i)$, $\cos(\delta_i)\sin(\beta_i)$, $\sin(\delta_i)$ sont les projections du vecteur unitaire de PS_i sur les axes principaux d'inertie de la planète. Ceci donne les équations suivantes pour le mouvement de P et de S_i en repère galiléen :

$$\frac{d^2\,OP}{dt^2} = \mu_i\, grad_P(V) \qquad \text{et} \qquad \frac{d^2\,OS_i}{dt^2} = M\, grad_{S_i}(V)$$

d'où l'on déduit l'équation du mouvement de S_i en repère planétocentrique:

$$\frac{d^2\,PS_i}{dt^2} = M\, grad_{S_i}(V) - \mu_i\, grad_P(V)$$

Vues les variables dont dépend V, on a $grad_P(V) = -grad_{S_i}(V)$ d'où l'expression annoncée de la fonction perturbatrice, ayant en facteur $M + \mu_i$. C(6.43)

4.2. Initialisations

On relit les développements en X, Xb, Y et Yb construits dans la première partie, et on les prépare à pouvoir être utilisés pour deux corps d'indices i et j:

Si ce n'est pas déjà fait, commencer par exécuter la phase d' initialisations générales, puis :

```
> read cat(chemin,`cosS.m`):
> read cat(chemin,`sincosd.m`):
> read cat(chemin,`kepler20.m`):
> read cat(chemin,`SINDCOSDXYie10.m`);
> read cat(chemin,`SINDCOSDEXPBsig.m`);
> read cat(chemin,`COSSXij10.m`);
> read cat(chemin,`Xnm10.m`);
```

```
> Dmax:=10;
```

$$Dmax := 10$$

```
> ASR:=TRONC(ASR,Dmax,[X,Xb]):  PRINTRONC(ASR,3,[X,Xb]);
> RSA:=TRONC(RSA,Dmax,[X,Xb]):  PRINTRONC(RSA,3,[X,Xb]);
> THETA:=TRONC(THETA,Dmax,[X,Xb]):  PRINTRONC(THETA,3,[X,Xb]);
> THETAB:=TRONC(THETAB,Dmax,[X,Xb]):  PRINTRONC(THETAB,3,[X,Xb]);
> PRODPOL(THETA,THETAB,Dmax,[X,Xb]);
```

$$\frac{390625}{145152}\,X^{10},\ + \dots + ,\ \frac{9}{16}\,X^3 - \frac{1}{16}\,X^2\,Xb - \frac{1}{16}\,X\,Xb^2 + \frac{9}{16}\,Xb^3 + \frac{1}{2}\,X^2 + \frac{1}{2}\,Xb^2 + \frac{1}{2}\,X + \frac{1}{2}\,Xb + 1$$

$$-\frac{78125}{290304}\,X^{10},\ + \dots + ,\ -\frac{3}{16}\,X^3 + \frac{3}{16}\,X^2\,Xb + \frac{3}{16}\,X\,Xb^2 - \frac{3}{16}\,Xb^3 - \frac{1}{4}\,X^2 + \frac{1}{2}\,X\,Xb - \frac{1}{4}\,Xb^2 - \frac{1}{2}\,X - \frac{1}{2}\,Xb + 1$$

$$\frac{25937424601}{3715891200}\,X^{10},\ + \dots + ,\ \frac{4}{3}\,X^3 - \frac{5}{4}\,X^2\,Xb - \frac{1}{12}\,Xb^3 + \frac{9}{8}\,X^2 - X\,Xb - \frac{1}{8}\,Xb^2 + X - Xb + 1$$

$$-\frac{4782969}{45875200}\,X^{10},\ + \dots + ,\ -\frac{1}{12}\,X^3 - \frac{5}{4}\,X\,Xb^2 + \frac{4}{3}\,Xb^3 - \frac{1}{8}\,X^2 - X\,Xb + \frac{9}{8}\,Xb^2 - X + Xb + 1$$

$$1$$

```
> ASRj:=subs(X=X[j],Xb=Xb[j],ASR): ASRi:=subs(j=i,ASRj):
> RSAi:=subs(X=X[i],Xb=Xb[i],RSA): RSAj:=subs(i=j,RSAi):
> THETAi:=subs(X=X[i],Xb=Xb[i],THETA):
> THETAj:=subs(i=j,THETAi):
> THETABi:=subs(X=X[i],Xb=Xb[i],THETAB):
> THETABj:=subs(i=j,THETABi):
> varXYij:=[X[i],Xb[i],Y[i],Yb[i],X[j],Xb[j],Y[j],Yb[j]]:
> varXij:=[X[i],Xb[i],X[j],Xb[j]]:
> varXYije:=[X[i],Xb[i],Y[i],Yb[i],X[j],Xb[j],Y[j],Yb[j],Y[e],Yb[e]]:
> varXYie:=[X[i],Xb[i],Y[i],Yb[i],Y[e],Yb[e]]:
```

```
> varxyij:=[x[i],xb[i],x[j],xb[j],y[i],yb[i],y[j],yb[j]]:
> varxyije:=[x[i],xb[i],x[j],xb[j],y[i],yb[i],y[j],yb[j],y[e],yb[e]]:
> varxyie:=[x[i],xb[i],y[i],yb[i],y[e],yb[e]]:
```

4.3. Développement en harmoniques sphériques, du potentiel de gravitation d'une planète en un point S

Le développement du potentiel de gravitation d'une planète non sphérique en un point S utilise les fonctions harmoniques sphériques, qui s'expriment en polynômes de Legendre et en fonctions associées de Legendre ayant comme argument le sinus de la latitude δ de S, latitude dans un repère équatorial principal d'inertie lié à cette planète et ayant son origine en son centre de masse. Si r, β, δ sont les coordonnées équatoriales de S, le développement du potentiel en ce point peut alors se mettre sous la forme (voir le cours): C15.4.3

$$\mathrm{U}(S) = \sum_{n=0}^{\infty} K M \frac{{a_e}^n}{r^{(n+1)}} (\sum_{p=0}^{n} P_{n,\,p}(\sin(\delta))\,(c_{n,\,p}\cos(p\,\beta) + s_{n,\,p}\sin(p\,\beta)))$$

où a_e est le rayon équatorial de la planète, M sa masse et où les $P_{n,\,p}(x)$ sont les fonctions associées de Legendre. On peut aussi l'écrire sous forme complexe :

$$\mathrm{U}(S) = \sum_{n=0}^{\infty} K M \frac{{a_e}^n}{r^{(n+1)}} \left(\sum_{p=0}^{n} P_{n,\,p}(\sin(\delta))\,(a_{n,\,p}\,\exp\left(I\,p\,\beta\right) + b_{n,\,p}\,\exp\left(-I\,p\,\beta\right))\right)$$

Les coefficients d'aplatissement $a_{n,\,p}$ et $b_{n,\,p}$ sont alors complexes, reliés aux coefficients classiques J_n, $c_{n,\,p}$ et $s_{n,\,p}$ par les relations:

$c_{n,\,0} = -J_n$ et $s_{n,\,0} = 0$ $\forall n$ et en particulier: $c_{0,\,0} = 1$ et $c_{1,\,0} = 0$

puis

$$a_{n,\,p} = \frac{c_{n,\,p} - I\,s_{n,\,p}}{2} \qquad \text{et} \qquad b_{n,\,p} = \frac{c_{n,\,p} + I\,s_{n,\,p}}{2}$$

soit encore:

$$c_{n,\,p} = a_{n,\,p} + b_{n,\,p} \qquad \text{et} \qquad s_{n,\,p} = I\,(a_{n,\,p} - b_{n,\,p})$$

On a enfin une autre expression de ce potentiel, faisant intervenir $\cos(\delta)\ \exp(I\,\beta)$, c'est-à-dire la coordonnée complexe de S dans le plan équatorial et ses puissances:

$$\mathrm{U}(S) = \sum_{n=0}^{\infty} K\,M\,\frac{{a_e}^n}{r^{(n+1)}}\left(\sum_{p=0}^{n} Q_{n,\,p}(\sin(\delta))\cos(\delta)^p\,(a_{n,\,p}\ \exp(I\,p\,\beta) + b_{n,\,p}\ \exp(-I\,p\,\beta))\right)$$

Les fonctions $Q_{n,\,p}(x)$ se calculent par récurrence, comme les fonctions associées de Legendre, à partir des polynômes de Legendre $P_n(x)$, selon la formule classique :

$$(n-p)\,Q_{n,\,p}(x) - (2\,n-1)\,x\,Q_{n-1,\,p}(x) + (n-1+p)\,Q_{n-2,\,p}(x) = 0$$

avec $Q_{n,\,p}(x)$ nul pour $n < p$ et $Q_{n,\,0} = P_n$ .Voici la procédure correspondante :

```
> with(orthopoly,P):
> Q:=proc(n,p,x) option remember;
> if p=0 then P(n,x) elif p>n then 0 elif p=n then (2*n)!/2^n/n!  else
> ((2*n-1)*x*Q(n-1,p,x)-(n-1+p)*Q(n-2,p,x))/(n-p) fi;
> end:
```

4.3.1. Développements analytiques complets (degré modéré)

On développe ici séparément les termes de $U(P)$, correspondant chacun à un harmonique $a_{n,\,p}$ ou $b_{n,\,p}$:

```
> degmx:=6;
```

$$degmx := 6$$

```
> ASRiT:=TRONC(ASRi,degmx,varXij):  nops(%);
> SI:=TRONC(SINDXYie10,degmx,varXYie):  nops(%);
> CO:=TRONC(COSDEXPBXYie10,degmx,varXYie):  nops(%);
> COB:=TRONC(COSDEXPBBXYie10,degmx,varXYie):  nops(%);
```

$$25$$
$$130$$
$$134$$
$$134$$

Pour calculer le développement au degré *degmax* du terme (n, p) de $U(S)$ en fonction des variables $(X,\, Xb,\, Y,\, Yb)$ ou des $(x,\, xb,\, y,\, yb)$ ou des $(e,\, \varpi,\, \gamma,\, \Omega)$ on propose d'utiliser la procédure suivante où le paramètre *cas* prend les valeurs 1, 2 ou 3 suivant ces 3 cas respectifs. Le résultat correspond au cas choisi. Si p est nul, on calcule directement la fonction dépendant du polynôme de Legendre $P_n(\sin(\delta))$. Si p est positif, on fait les calculs avec les fonctions associées de Legendre redéfinies ci-dessus et qui viennent en facteur de $a_{n,\,p}$, tandis que pour p négatif, ce sont celles en facteur de $b_{n,\,p}$.

```
> ABnp:=proc(n,p,degmax,cas) local Uk,UkXY,Ukxy,Ukei,var;
> var:= [X[i],Xb[i],Y[i],Yb[i],Y[e],Yb[e]]:
> if p=0 then
> Uk:=mtaylor(ASRiT^(n+1)*P(n,SI),var,degmax+1)
> elif p>0 then
> Uk:=mtaylor(ASRiT^(n+1)*Q(n,p,SI)*CO^p,var,degmax+1)
> else
> Uk:=mtaylor( ASRiT^(n+1)*Q(n,-p,SI)*COB^abs(p),var,degmax+1):
> fi:  if cas=0 then RETURN(Uk) fi:
> subs(sigma=exp(I*L[i])/exp(I*L[e]),Uk):
> UkXY:=collect(combine(%,exp),exp,factor);
> if cas=1 then RETURN(UkXY) fi:
```

```
> Ukxy:=collect(combine(
> subs(X[i]=xb[i]*exp(I*L[i]),Xb[i]=x[i]/exp(I*L[i]),
> Y[i]=yb[i]*exp(I*L[i]),Yb[i]=y[i]/exp(I*L[i]),
> Y[e]=yb[e]*exp(I*L[e]),Yb[e]=y[e]/exp(I*L[e]),
> expand(UkXY)),exp),exp,factor);
> if cas=2 then RETURN(Ukxy) fi:
> collect(combine(
> subs(x[i]=e[i]*exp(I*omega[i]),xb[i]=e[i]/exp(I*omega[i]),
> y[i]=gamma[i]*exp(I*Omega[i]),yb[i]=gamma[i]/exp(I*Omega[i]),
> y[e]=gamma[e]*exp(I*Omega[e]),yb[e]=gamma[e]/exp(I*Omega[e]),
> expand(Ukxy)),exp),exp,factor);
> Ukei:=collect(simplify(convert(%,trig)),cos,factor);
> if cas=3 then RETURN(Ukei) fi:
> RETURN(Uk):
> end:
```

Quelques exemples (qui permettraient le calcul des termes en J_2, c_{22} et s_{22}); on peut vérifier que $UJ(2,2,-,-)$ et $UJ(2,-2,-,-)$ sont bien conjugués l'un de l'autre. Le premier exemple permet de comparer le développement du terme en J_2 avec les résultats classiques donnés par exemple en dans le cours en section 21.5.

```
> PRINTRONC(ABnp(2,0,degmx,3),2,[e[i],gamma[i],gamma[e]]);
```

$$(-\frac{35}{32}-\frac{129}{160}\cos(-4\,\omega_i+4\,L_i)-\frac{141}{64}\cos(-2\,\omega_i+2\,L_i)-\frac{3167}{320}\cos(-6\,\omega_i+6\,L_i))\,{e_i}^6,\ +\ldots+,(-\frac{3}{4}-\frac{9}{4}\cos(-2\,\omega_i+2\,L_i))\,{e_i}^2$$
$$+(-3\,{C_i}^2\cos(-2\,\Omega_i+2\,L_i)+3\,{C_i}^2)\,{\gamma_i}^2+(6\,C_e\,C_i\cos(2\,L_i-\Omega_e-\Omega_i)-6\,C_e\,C_i\cos(\Omega_e-\Omega_i))\,\gamma_i\,\gamma_e$$
$$+(-3\,{C_e}^2\cos(2\,L_i-2\,\Omega_e)+3\,{C_e}^2)\,{\gamma_e}^2-\frac{3}{2}\cos(-\omega_i+L_i)\,e_i-\frac{1}{2}$$

```
> PRINTRONC(ABnp(2,0,degmx,2),2,[x[i],xb[i],y[i],yb[i],y[e],yb[e]]);
```

$$-\frac{3167}{640}\, e^{(-6\, I\, L_i)}\, {x_i}^6,\; + \ldots +\, ,\, -\frac{9}{8}\, \%2\, {x_i}^2 - \frac{3}{4}\, x_i\, xb_i - \frac{9}{8}\, \%1\, {xb_i}^2 - \frac{3}{2}\, {C_i}^2\, \%2\, {y_i}^2 + 3\, {C_i}^2\, y_i\, yb_i + 3\, C_e\, C_i\, \%2\, y_i\, y_e$$
$$- 3\, C_e\, C_i\, y_i\, yb_e - \frac{3}{2}\, {C_i}^2\, \%1\, {yb_i}^2 - 3\, C_e\, C_i\, yb_i\, y_e + 3\, C_e\, C_i\, \%1\, yb_i\, yb_e - \frac{3}{2}\, {C_e}^2\, \%2\, {y_e}^2 + 3\, {C_e}^2\, y_e\, yb_e$$
$$- \frac{3}{2}\, {C_e}^2\, \%1\, {yb_e}^2 - \frac{3}{4}\, e^{(-I\, L_i)}\, x_i - \frac{3}{4}\, e^{(I\, L_i)}\, xb_i - \frac{1}{2}$$
$$\%1 := e^{(2\, I\, L_i)}$$
$$\%2 := e^{(-2\, I\, L_i)}$$

```
> PRINTRONC(ABnp(2,2,degmx,2),2,[x[i],xb[i],y[i],yb[i],y[e],yb[e]]);
```

$$\frac{4}{15}\, e^{(-2\, I\, L_e - 4\, I\, L_i)}\, {x_i}^6,\; + \ldots +\, ,\, -\frac{15}{2}\, \%1\, x_i\, xb_i + \frac{51}{2}\, e^{(4\, I\, L_i - 2\, I\, L_e)}\, {xb_i}^2 + 6\, e^{(-2\, I\, L_e)}\, {y_i}^2 - 6\, \%1\, y_i\, yb_i - 12\, C_i\, C_e\, e^{(-2\, I\, L_e)}\, y_i\, y_e$$
$$+ 12\, C_e\, C_i\, \%1\, yb_i\, y_e + 6\, e^{(-2\, I\, L_e)}\, {y_e}^2 - 6\, \%1\, y_e\, yb_e - \frac{3}{2}\, e^{(I\, L_i - 2\, I\, L_e)}\, x_i + \frac{21}{2}\, e^{(3\, I\, L_i - 2\, I\, L_e)}\, xb_i + 3\, \%1$$
$$\%1 := e^{(2\, I\, L_i - 2\, I\, L_e)}$$

```
> PRINTRONC(ABnp(2,-2,degmx,2),2,[x[i],xb[i],y[i],yb[i],y[e],yb[e]]);
```

$$\frac{73369}{240}\, e^{(-8\, I\, L_i + 2\, I\, L_e)}\, {x_i}^6,\; + \ldots +\, ,\, \frac{51}{2}\, e^{(-4\, I\, L_i + 2\, I\, L_e)}\, {x_i}^2 - \frac{15}{2}\, \%1\, x_i\, xb_i - 6\, \%1\, y_i\, yb_i + 12\, C_e\, C_i\, \%1\, y_i\, yb_e + 6\, e^{(2\, I\, L_e)}\, {yb_i}^2$$
$$- 12\, C_e\, C_i\, e^{(2\, I\, L_e)}\, yb_i\, yb_e - 6\, \%1\, y_e\, yb_e + 6\, e^{(2\, I\, L_e)}\, {yb_e}^2 + \frac{21}{2}\, e^{(-3\, I\, L_i + 2\, I\, L_e)}\, x_i - \frac{3}{2}\, e^{(-I\, L_i + 2\, I\, L_e)}\, xb_i + 3\, \%1$$
$$\%1 := e^{(-2\, I\, L_i + 2\, I\, L_e)}$$

Boucle permettant de calculer un ensemble de ces fonctions (ici toutes les harmoniques jusqu'à l'indice 6):

```
> deb:=time():  ABxy:=array(2..6,-6..6):
> for nj from 2 to 6 do:
> for np from -nj to nj do:
```

```
> ABxy[nj,np]:=ABnp(nj,np,degmx,2):
> od:
> od:
> print(time()-deb):
```

$$1734.871$$

```
> # print(ABxy):
```

Exemple pour changer l'indice du satellite dans l'un de ces développements:

```
> subs(i=j,ABxy[2,2]):
```

sauvegarde de cet ensemble de développements :

```
> save  ABxy , cat(chemin,`ABxy`.degmx.`.m`):
```

4.3.2. Développements analytiques par inégalité

On calcule ici les termes par inégalité

```
> with(orthopoly,P):
> Q:=proc(n,p,x) option remember;
```

```
> if p=0 then P(n,x) elif p>n then 0 elif p=n then (2*n)!/2^n/n!  else
> ((2*n-1)*x*Q(n-1,p,x)-(n-1+p)*Q(n-2,p,x))/(n-p) fi;
> end:
```

Le fichier "initmo.pr" contient aussi la procédure *initmoApla*, qui établit la liste des monômes présents dans le développement d'une inégalité donnée intervenant dans les perturbations par l'aplatissement (avec les indices i et e pour les variables concernées) :

```
> read cat(chemin,`initmo.pr`):
```

La procédure suivante est une version de *COEFTTS* adaptée aux variables d'aplatissement, utilisée dans *INEGApla*.

```
> COEFTTSie:=proc(s,ni,nbi,ne,nbe) local c,r,k,km:
> r:=table():  km:=1:
> c:=coeff(coeff(coeff(coeff(s,Y[i],ni),Y[e],ne),Yb[i],nbi),Yb[e],nbe);
> if type(c,`+`) then
> km:=nops(c):
> for k from 1 to km do
> r[k]:=COEF(op(k,c)):
> od
> else r[1]:=COEF(c):
> fi:
> RETURN([seq(r[k],k=1..km)]):
> end:
```

Les deux procédures suivantes ont déjà été utilisées pour le développement d'inégalités pour les perturbations mutuelles de planètes:

```
> COEF:=proc(u) local ui,uj,us,cf:
```

```
> ui:=degree(u,theta[i]); uj:=degree(u,theta[j]);us:=degree(u,sigma);
> cf:=coeff(coeff(coeff(u,theta[i],ui),theta[j],uj),sigma,us);
> RETURN([ui,uj,us,cf]):
> end:

> COEFXXb:=proc(XXnm,nn,mm,n1,nb1) local cf:
> cf:=coeff(coeff(XXnm,X,n1),Xb,nb1);
> # print(cf):
> cf:=subs(n=nn,m=mm,cf):
> RETURN(cf):
> end:
```

Cette procédure initialise les séries en inclinaisons et en excentricité pour les harmoniques souhaités:

```
> IniINEGApla:=proc(degmax,dJ) local k,p,x,y,res:
> res:=array(-dJ..dJ):
> res[0]:= SUBS([x=SINDsig],P(dJ,x),degmax,[Y[i],Yb[i],Y[e],Yb[e]]):
> for p from 1 to dJ do:
> res[p]:= SUBS([x=SINDsig,y=COSDEXPBsig],Q(dJ,p,x)*y^p,
> degmax,[Y[i],Yb[i],Y[e],Yb[e]]):
> res[-p]:= SUBS([x=SINDsig,y=COSDEXPBBsig],Q(dJ,p,x)*y^p,
> degmax,[Y[i],Yb[i],Y[e],Yb[e]]):
> od:
> RETURN(res):
> end:
```

La procédure *INEGApla* construit le développement d'une inégalité donnée ($pp\,L_i + qq\,L_e$) entre les degrés *degmin* et *degmax*, pour les perturbations dépendant de l'harmonique (n, p) dans la représentation en harmoniques sphériques. *SERINCL* est le nom formel du tableau qui contient les développements en inclinaisons (variables $\{Y_i,\ Yb_i,\ Y_e,\ Yb_e\}$ et ayant en facteur des puissances de θ_i ou de $\sigma = \exp(I(L_i - L_e))$. Ce nom est utilisé avec l'indice p passé en paramètre. Les développements sont pairs ou impairs en inclinaisons suivant les valeurs de n et p, à savoir même parité que $n - p$.

```
> INEGApla:=proc(pp,qq,degmin,degmax,n,p,SERINCL,XXnm) local
> ic,ip,r,mon,nmon,m1,m2,mono,n1,n2,n3,n4,n5,n6,n7,n8,k1,k2,k,
> moncos,nmoncos,monoc,ns,c,nti,ntj,c1,c2,c3,cfa,lda,nni,nnj:
> ic:=pp+qq:  r:=0:  if abs(p)>n then RETURN(r) fi:
> if type(eval(n-p),even) then ip:=0 else ip:=1 fi:  print(n-p,ip):
> mon:=initmoApla(degmin,degmax,ic,ip,2):  nmon:=nops(mon):
> for m1 from 1 to nmon do:
> c:=0:
> mono:=op(m1,mon):
> n3:=op(5,mono):n4:=op(6,mono):
> n5:=op(7,mono):n6:=op(8,mono):n7:=op(9,mono):n8:=op(10,mono):
> moncos:=COEFTTSie(SERINCL[p],n7,n8,n5,n6); nmoncos:=nops(moncos):
> k:= pp +n4-n3+n8-n7:
> for m2 from 1 to nmoncos do:
> monoc:=op(m2,moncos):  c3:=op(4,monoc):
> if c3<>0 then
> nti:=op(1,monoc):  ns:=op(3,monoc):
> nni:=-(n+1):
> c1:=COEFXXb(XXnm,nni,nti,n3,n4):
> c:=c+c1*c3*piecewise(k=ns,1,0):
> fi :
> od:
```

```
> r:=r+c*op(11,mono):
> od:
> if p<0 then r:=r*b[n,-p] else r:=r*a[n,p] fi:
> RETURN(r);
> end:
```

Exemple d'utilisation: On fixes les degrés utiles en excentricités et inclinaisons, et l'indice *dJn* des harmoniques sphériques dont on veut le développement; on calcule tous les ici tous les harmoniques sphériques correspondant à cet indice, soit pour p entre -*dJn* et *dJn* :

```
> Dmin:=0; Dmax:=6;
> dJn:=4;
```

$$Dmin := 0$$
$$Dmax := 6$$
$$dJn := 4$$

```
> PQnp:=array(-dJn..dJn):
> XnmT:=TRONC(Xnm,Dmax,[X,Xb]):
```

La procédure *IniINEGApla* tronque les développements utiles en excentricités au degré *Dmax* et prépare les développements en inclinaisons à leur utilisation dans la procédure *INEGApla* . Ceux-ci sont alors disponibles dans le tableau *PQnp*.

```
> deb0:=time():
> PQnp:=IniINEGApla(Dmax,dJn):  print(time()-deb0):
```

$$.931$$

Il suffit ensuite de définir les entiers (pp, qq) représentant une inégalité, et d'exécuter *INEGApla* pour obtenir son développement. Notons cependant que pour une fonction harmonique d'indices (n, p), on a forcément $\exp\left(I\left(-p\right)L_e\right)$ en facteur:

```
> pp:=2;  p:=0; qq:=-p;
```

$$pp := 2$$

$$p := 0$$

$$qq := 0$$

```
> PQnp[p]:
```

```
> deb:=time():
> INEGApla(pp,qq,Dmin,Dmax,dJn,p,PQnp,XnmT):
> sort(collect(expand(%),varxyie,distributed),varxyie):  nops(%);
> USineg:=%%*eL[i]^pp*eL[e]^qq:
> print("duree=",time()-deb):
> # USineg;
```

$$4,\ 0$$

$$66$$

$$\text{“duree=”},\ .230$$

L'utilisation de ce développement dans les équations de Lagrange (définies en 5.1) est alors immédiate: Il suffit de multiplier le développement *USineg* par les facteurs dynamiques qui interviennent dans la fonction perturbatrice d'un satellite par l'aplatissement de sa planète:

on commence par lire la procedure de calcul des équations de Lagrange

```
> read cat(chemin,‘EqLag.pr‘):
```

puis on calcule la fonction perturbatrice, puis les équations de Lagrange correspondantes (pour un satellite, on utilise cette procédure pour le cas "interieur":

```
> Up:=expand(‘K*(M+mu[i])‘/a[i]*(a[e]/a[i])^dJn*USineg):
> DVS:=EqLag(Up,"interieur",i,e):
> print(seq(nops(expand(DVS[k])),k=1..4));
```

$$97,\ 253,\ 99,\ 92$$

On peut comparer ce résultat avec celui *DVSi* issu du développement complet au degré 6 (il faut au préalable avoir effectué le calcul de *USiab* (réalisé en 4.4.1) et de *DVSi*, ou bien exécuter les 3 groupes de commande suivants)

```
> read cat(chemin,‘USiab_6.m‘):
> if p<0 then Upa:=expand(coeff(USiab,b[dJn,-p])*b[dJn,-p]) else
> Upa:=expand(coeff(USiab,a[dJn,p])*a[dJn,p]) fi:  nops(%);
```

$$606$$

```
> DVSi:=EqLag(Upa,"interieur",i,e):
> print(seq(nops(expand(DVSi[k])),k=1..4));
```

$$490,\ 1629,\ 481,\ 428$$

Comparaison :

```
> for k to 4 do:
> r1:=  coeff(coeff(DVS[k],eL[i],pp),eL[e],qq):
> r2:= coeff(coeff(DVSi[k],eL[i],pp),eL[e],qq):
> print(k," :  difference=",simplify(r1-r2));  #
> print(‘DVS[‘.k.‘]=‘,sort(collect(
> %,varxyie,distributed,factor),varxyie)*exp(I*(pp*L[i]+qq*L[e])));
```

```
> od:
```

$$1,\ \text{“ : difference=”},\ 0$$
$$2,\ \text{“ : difference=”},\ 0$$
$$3,\ \text{“ : difference=”},\ 0$$
$$4,\ \text{“ : difference=”},\ 0$$

4.4. Fonction perturbatrice d'un satellite par l'aplatissement de sa planète

4.4.1. Développement complet

Calcul du développement complet, à partir des harmoniques sphériques obtenues en 4.3.1 de manière complète au degré 6 et pour tous les harmoniques possibles jusqu'à l'indice 6.

```
> degmx:=6;
> read cat(chemin,‘ABxy‘.degmx.‘.m‘):
> read cat(chemin,‘EqLag.pr‘):
```

$$degmx := 6$$

Développement en coefficients complexes $a_{n,\,p}$ et $b_{n,\,p}$:

```
> n:='n':  p:='p':
> USiab:=expand(‘K*(M+mu[i])‘/a[i]*sum((a[e]/a[i])^n*(a[n,0]*ABxy[n,0]+s
> um(a[n,p]*ABxy[n,p]+b[n,p]*ABxy[n,-p],p=1..n)),n=2..6)):  nops(%);
```

$$26368$$

```
> USiab:=subs(exp(I*L[i])=eL[i], exp(I*L[e])=eL[e], USiab):
> nops(%);
```

$$26368$$

exemple de terme:

```
> op(190,USiab);
```

$$315\,\frac{K*(M+mu[i])\,{a_e}^4\,a_{4,\,0}\,{y_e}^2\,{yb_i}^2\,y_i\,yb_e\,{C_i}^3\,C_e}{{a_i}^5}$$

```
> save USiab, cat(chemin,`USiab_6.m`):
```

Développement en coefficients J_n , $c_{n,\,p}$ et $s_{n,\,p}$:

```
> USiJcs:=expand(subs(seq(a[n,0]=-J[n],n=2..6),
> seq(seq(a[n,p]=(c[n,p]-I*s[n,p])/2,p=1..n),n=2..6),
> seq(seq(b[n,p]=(c[n,p]+I*s[n,p])/2,p=1..n),n=2..6),USiab)):
> nops(%);
```

$$50082$$

exemple de terme

```
> op(190,USiJcs);
```

$$15\,\frac{I\,K*(M+mu[i])\,{a_e}^3\,J_3\,{eL_i}^3\,{yb_i}^3\,yb_e\,y_e\,{C_i}^3}{{a_i}^4}$$

```
> save USiJcs, cat(chemin,`USiJcs_6.m`):
```

Ces développements sont alors prets à pouvoir être utilisés dans les équations de Lagrange

vers les équations de Lagrange

4.4.2. Développement par inégalités

Ce type de développement est calculable, ainsi que les équations de Lagrange correspondantes, dans la section 4.3.2

4.5. Cas de deux satellites massifs : termes de perturbation indirecte par l'aplatissement

Dans le cas de plusieurs satellites massifs, interviennent leur perturbations mutuelles, analogues aux perturbations entre planètes vues dans la section 3. Il faut aussi faire intervenir l'aplatissement de la planète, qui agit sur chaque satellite par une fonction perturbatrice de la forme décrite en 4.3 et 4.4, mais la masse non négligeable des satellites engendre sur leur mouvement planétocentrique, des perturbations indirectes : En effet, en notant $M\,\mu_i\,V_i = M\,\mu_i\,\mathrm{V}(r_i,\,\delta_i,\,\beta_i)$ le développement en harmoniques sphériques du potentiel de gravitation de la planète aplatie sur un satellite S_i, le potentiel de gravitation de la planète et de deux satellites S_i et S_j s'écrit (avec une constante de gravitation prise égale à 1):

$$U = M\,\mu_i\,V_i + M\,\mu_j\,V_j + \frac{\mu_i\,\mu_j}{r_{i,\,j}}$$

où $r_{i,\,j}$ est la distance $S_i\,S_j$ (et où $\{r_i,\,\delta_i,\,\beta_i\}$ sont les coordonnées sphériques du vecteur PS_i dans un repère planétocentrique). Des équations du mouvement dans un repère galiléen :

$$\frac{d^2\,OP}{dt^2} = \mu_i\,grad_P(V_i) + \mu_j\,grad_P(V_j) \qquad \text{et} \qquad \frac{d^2\,OS_i}{dt^2} = M\,grad_{S_i}(V_i) + \mu_j\,grad_{S_i}(\frac{1}{r_{i,\,j}})$$

on déduit, sachant que $grad_P(V_i) = -grad_{S_i}(V_i)$:

$$\frac{d^2\,PS_i}{dt^2} = (M+\mu_i)\,grad_{S_i}(V_i) + \mu_j\,grad_{S_i}(\frac{1}{r_{i,\,j}}) + \mu_j\,grad_{S_j}(V_j)$$

On aurait une équation analogue pour PS_j. Le dernier terme de cette équation représente une perturbation indirecte de S_i par S_j, dont la partie principale est $\mu_j\,grad_{S_j}(\frac{1}{r_j})$, laquelle peut se mettre sous la forme classique de la perturbation indirecte d'un problème planétaire : $\mu_j\,grad_{S_i}(-\frac{r_i}{{r_j}^2}\cos S_{i,\,j})$ où $S_{i,\,j}$ est l'angle entre PS_i et PS_j. L'autre partie de ce terme est au moins de l'ordre de $\mu_j\,J_2$ puisqu'elle s'écrit (cf. 4.3):

$$\mu_j\,grad_{S_j}(\sum_{n=2}^{\infty}(\sum_{p=0}^{n} U_{j,\,n,\,p})) \qquad \text{où} \qquad U_{j,\,n,\,p} = \frac{{a_e}^n}{{r_j}^{(n+1)}}\,Q_{n,\,p}(\sin\delta_j)\,A_{n,\,p}(\delta_j,\,\beta_j)$$

avec $A_{n,\,p}(\delta_j,\,\beta_j) = a_{n,\,p}\,(\cos\delta_j\ \exp\,(I\,\beta_j))^p + b_{n,\,p}\,(\cos\delta_j\ \exp\,(-I\,\beta_j))^p$

On peut montrer que cette perturbation indirecte se transforme aussi sous la forme d'un gradient en S_i:

$$\mu_j\ grad_{S_i}(\sum_{n=2}^{\infty}(\sum_{p=0}^{n}\mathrm{W}_{j,\,n,\,p}))$$

En effet, on a (voir Annexe A):

$$grad_{S_j}(\frac{1}{{r_j}^{(n+1)}}) = grad_{S_i}(-\frac{(n+1)\,r_i\cos S_{i,\,j}}{{r_j}^{(n+2)}})$$
$$grad_{S_j}(Q_{n,\,p}(\sin\delta_j)) = grad_{S_i}\left((\frac{r_i\sin\delta_i}{r_j} - \frac{\sin\delta_j\,r_i\cos S_{i,\,j}}{r_j})\,(\frac{\partial}{\partial x}\,Q_{n,\,p}(x))_{\{x=\sin\delta_j\}}\right)$$
$$grad_{S_j}((\cos\delta_j\,e^{(I\,\beta_j)})^p) = grad_{S_i}\left(\frac{r_i\cos\delta_i\,e^{(I\,\beta_i)}}{r_j} - \frac{\cos\delta_j\,e^{(I\,\beta_j)}\,r_i\cos S_{i,\,j}}{r_j}\right)\,p\,(\cos\delta_j\ \exp\,(I\,\beta_j))^{(p-1)}$$

d'où l'on tire :

$$\begin{aligned}
\mathrm{W}_{j,\,n,\,p} &= \frac{r_i\sin\delta_i}{{r_j}^{(n+2)}}\,(\frac{\partial}{\partial x}\,Q_{n,\,p}(x))_{\{x=\sin\delta_j\}}A_{n,\,p}(\delta_j,\,\beta_j)\\
&+ p\,a_{n,\,p}Q_{n,\,p}(\sin\delta_j)\frac{r_i\cos\delta_i\,e^{(I\,\beta_i)}}{{r_j}^{(n+2)}}(\cos\delta_j\,e^{(I\,\beta_j)})^{(p-1)} + p\,b_{n,\,p}Q_{n,\,p}(\sin\delta_j)\frac{r_i\cos\delta_i\,e^{(-I\,\beta_i)}}{{r_j}^{(n+2)}}(\cos\delta_j\,e^{(-I\,\beta_j)})^{(p-1)}\\
&+ \frac{r_i\cos S_{i,\,j}}{{r_j}^{(n+2)}}\left(-(n+1+p)\,Q_{n,\,p}(\sin\delta_j) - \sin\delta_j\,(\frac{\partial}{\partial x}\,Q_{n,\,p}(x))_{\{x=\sin\delta_j\}}\right)\,A_{n,\,p}(\delta_j,\,\beta_j)
\end{aligned}$$

C'est la partie de cette fonction qui vient en facteur d'un coefficient $a_{n,\,p}$ ou $b_{n,\,p}$ qui est calculée dans la procédure *Windj(n,p)* donnée ci-dessous. On commence par définir les fonctions de $\mathrm{W}_{j,\,n,\,p}$ qui ne dépendent que de $\sin\delta_j$:

```
> with(orthopoly,P):
```

```
> Q:=proc(n,p,x) option remember;
> if p=0 then P(n,x) elif p>n then 0 elif p=n then (2*n)!/2^n/n!  else
> ((2*n-1)*x*Q(n-1,p,x)-(n-1+p)*Q(n-2,p,x))/(n-p) fi;
> end:
> p1:=('n','x')->diff(P('n','x'),'x');
> p2:=('n','x')->('n'+1)*P('n','x')+'x'*p1('n','x');
> q1:=('n','p','x')->diff(Q('n','p','x'),'x');
> q2:=('n','p','x')->('n'+1+'p')*Q('n','p','x')+'x'*q1('n','p','x');
```

$$p1 := (n, x) \to \mathrm{diff}(\mathrm{P}(n, x), x)$$

$$p2 := (n, x) \to (n+1)\,\mathrm{P}(n, x) + x\,\mathrm{p1}(n, x)$$

$$q1 := (n, p, x) \to \mathrm{diff}(\mathrm{Q}(n, p, x), x)$$

$$q2 := (n, p, x) \to (n+1+p)\,\mathrm{Q}(n, p, x) + x\,\mathrm{q1}(n, p, x)$$

Exemples :

```
> # Q(4,2,x); q1(4,2,x); q2(4,2,x);
```

On initialise ensuite les développements nécessaires en excentricités et inclinaisons, limités au degré utile. On remplace aussi σ par $\sigma_i = \exp(I\,L_i - I\,L_e)$ dans les développements relatifs au satellite S_i, et on remplace i par j pour obtenir ceux relatifs au satellite S_j. De ce fait, dans le développement de $\cos S_{i,\,j}$ obtenu lorsqu'on a considéré l'angle de deux planètes (en 2.3), la variable $\sigma = \exp(I\,L_i - I\,L_j)$ est ici remplacée par $\frac{\sigma_i}{\sigma_j}$.

```
> degmx:=3;
```

$$degmx := 3$$

```
> ASRiT:=TRONC(ASRi,degmx,varXij):  nops(%);
> RSAiT:=TRONC(RSAi,degmx,varXij):  nops(%);
> SINDi:=subs(sigma=sigma[i],TRONC(SINDXYie10,degmx,varXYie)):
> nops(%);
```

```
> COSDEXPBi:=subs(sigma=sigma[i],TRONC(COSDEXPBXYie10,degmx,varXYie)):
> nops(%);
> COSDEXPBBi:=subs(sigma=sigma[i],TRONC(COSDEXPBBXYie10,degmx,varXYie))
> :  nops(%);
> ASRjT:=subs(i=j,ASRiT):
> RSAiT:=subs(i=j,RSAiT):
> SINDj:=subs(i=j,SINDi):
> COSDEXPBj:=subs(i=j,COSDEXPBi):  COSDEXPBBj:=subs(i=j,COSDEXPBBi):
> COSSij:=subs(sigma=sigma[i]/sigma[j],TRONC(COSSXij10,degmx,varXYij)):
> nops(%);
```

$$9$$
$$10$$
$$30$$
$$27$$
$$27$$
$$86$$

```
> SINDj:COSSij:
```

La procédure *Windj* suivante utilise la fonction *mtaylor* pour établir les développements, dans la mesure où ces perturbations indirectes (qui sont de l'ordre de $J_2\,\mu_j$ au moins) sont généralement calculées à un degré faible. En sortie, le cas 1 correspond à un résultat en variables $(X, Xb, Y, Yb$ et $\sigma)$, le cas 2 en variables $(x, xb, y, yb$ et $eL)$, le cas 3 en variables $(e,\ \gamma,\ \omega,\ \Omega$ et $L)$.

```
> Windj:=proc(n,p,degmax,cas) local q,Uk,UkXY,Ukxy,Ukei;
> if p=0 then:
> Uk:=mtaylor(
> RSAi*ASRj^(n+2)*SINDi*subs(x=SINDj,p1(n,x)),varXYije,degmax+1)
> -mtaylor(
> RSAi*ASRj^(n+2)*COSSij*subs(x=SINDj,p2(n,x)),varXYije,degmax+1)
> elif p>0 then:
```

```
> Uk:=mtaylor( RSAi*ASRj^(n+2)*
> (p*COSDEXPBi*subs(x=SINDj,Q(n,p,x))+SINDi*COSDEXPBj*subs(x=SINDj,q1(n,
> p,x)))*COSDEXPBj^(p-1),varXYije,degmax+1)
> -mtaylor(RSAi*ASRj^(n+2)*COSSij*COSDEXPBj^p*subs(x=SINDj,q2(n,p,x)),
> varXYije,degmax+1)
> elif p<0 then:  q:=-p:
> Uk:=mtaylor( RSAi*ASRj^(n+2)*
> (q*COSDEXPBBi*subs(x=SINDj,Q(n,q,x))+SINDi*COSDEXPBBj*subs(x=SINDj,q1(
> n,q,x)))*COSDEXPBBj^(q-1),varXYije,degmax+1)
> -mtaylor(RSAi*ASRj^(n+2)*COSSij*COSDEXPBBj^q*subs(x=SINDj,q2(n,q,x)),
> varXYije,degmax+1)
> fi:
> UkXY:=sort(collect(Uk,varXYije,distributed),varXYije);
> if cas=1 then RETURN(UkXY) fi;
> Ukxy:=sort(collect(subs(X[i]=xb[i]*eL[i],Xb[i]=x[i]/eL[i],
> X[j]=xb[j]*eL[j],Xb[j]=x[j]/eL[j], Y[i]=yb[i]*eL[i],Yb[i]=y[i]/eL[i],
> Y[j]=yb[j]*eL[j],Yb[j]=y[j]/eL[j],
> Y[e]=yb[e]*eL[e],Yb[e]=y[e]/eL[e],
> sigma[i]=eL[i]/eL[e],sigma[j]=eL[j]/eL[e],
> expand(UkXY)),varxyije,distributed),varxyije);
> if cas=2 then RETURN(Ukxy) fi;
> collect(combine(
> subs(x[i]=e[i]*exp(I*omega[i]),xb[i]=e[i]/exp(I*omega[i]),
> x[j]=e[j]*exp(I*omega[j]),xb[j]=e[j]/exp(I*omega[j]),
> y[i]=gamma[i]*exp(I*Omega[i]),yb[i]=gamma[i]/exp(I*Omega[i]),
> y[j]=gamma[j]*exp(I*Omega[j]),yb[j]=gamma[j]/exp(I*Omega[j]),
> y[e]=gamma[e]*exp(I*Omega[e]),yb[e]=gamma[e]/exp(I*Omega[e]),
> eL[i]=exp(I*L[i]),eL[j]=exp(I*L[j]),eL[e]=exp(I*L[e]),
> expand(Ukxy)),exp),exp,factor);
> Ukei:=collect(simplify(convert(%,trig)),cos,factor);
> RETURN(Ukei):
```

```
> end:
```

Quelques exemples de Windj(n,p,degmax,cas) :

cas = 1

```
> Windj(2,0,2,1):  PRINTRONC(%,1,varXYije);
```

$$(\frac{3}{32}\frac{\sigma_j}{\sigma_i}+\frac{9}{32}\frac{\sigma_i}{\sigma_j})\,{X_i}^2,\ +\ldots+\ ,\ (\frac{3}{8}\frac{\sigma_i}{\sigma_j}-\frac{9}{8}\frac{\sigma_j}{\sigma_i})\,X_i+(-\frac{9}{8}\frac{\sigma_i}{\sigma_j}+\frac{3}{8}\frac{\sigma_j}{\sigma_i})\,Xb_i+(\frac{3}{4}\frac{\sigma_i}{\sigma_j}+\frac{9}{4}\frac{\sigma_j}{\sigma_i})\,X_j+(\frac{9}{4}\frac{\sigma_i}{\sigma_j}+\frac{3}{4}\frac{\sigma_j}{\sigma_i})\,Xb_j+\frac{3}{4}\frac{\sigma_j}{\sigma_i}+\frac{3}{4}\frac{\sigma_i}{\sigma_j}$$

cas =2 :

```
> Windj(2,0,2,2):  PRINTRONC(%,1,varxyije);
```

$$(\frac{3}{32}\frac{1}{eL_i\,eL_j}+\frac{9}{32}\frac{eL_j}{{eL_i}^3})\,{x_i}^2,\ +\ldots+\ ,$$

$$(\frac{3}{8}\frac{eL_j}{{eL_i}^2}-\frac{9}{8}\frac{1}{eL_j})\,x_i+(-\frac{9}{8}\,eL_j+\frac{3}{8}\frac{{eL_i}^2}{eL_j})\,xb_i+(\frac{9}{4}\frac{eL_i}{{eL_j}^2}+\frac{3}{4}\frac{1}{eL_i})\,x_j+(\frac{3}{4}\,eL_i+\frac{9}{4}\frac{{eL_j}^2}{eL_i})\,xb_j+\frac{3}{4}\frac{eL_j}{eL_i}+\frac{3}{4}\frac{eL_i}{eL_j}$$

cas = 3 :

```
> Windj(2,0,2,3);
```

$$
\begin{aligned}
&(-15\,\gamma_j{}^2\,C_j{}^2 - 9\,\gamma_e{}^2\,C_e{}^2 - \frac{3}{4}\,e_i{}^2 + 3\,e_j{}^2 - \frac{3}{2}\,\gamma_i{}^2 + \frac{3}{2} - \frac{3}{2}\,\gamma_j{}^2)\cos(L_i - L_j) + \frac{3}{2}\,\gamma_j{}^2\,(5\,C_j{}^2 + 1)\cos(L_j + L_i - 2\,\Omega_j)\\
&+ \frac{3}{16}\,e_i{}^2\cos(-2\,\omega_i + L_i + L_j) + \frac{9}{16}\,e_i{}^2\cos(-2\,\omega_i + 3\,L_i - L_j) - 9\,\gamma_j\,\gamma_e\,C_e\,C_j\cos(L_i - \Omega_e - \Omega_j + L_j)\\
&- \frac{27}{4}\,e_i\,e_j\cos(\omega_i + \omega_j - 2\,L_j) - 15\,\gamma_j\,\gamma_e\,C_e\,C_j\cos(\Omega_j - 3\,L_j + \Omega_e + L_i) - 9\,C_i\,C_j\,\gamma_i\,\gamma_j\cos(L_j + L_i - \Omega_i - \Omega_j)\\
&+ 9\,C_i\,C_j\,\gamma_i\,\gamma_j\cos(-L_j + L_i - \Omega_i + \Omega_j) - 6\,C_i\,C_e\,\gamma_i\,\gamma_e\cos(-L_j + L_i - \Omega_i + \Omega_e) + 6\,C_i\,C_e\,\gamma_i\,\gamma_e\cos(L_j + L_i - \Omega_i - \Omega_e)\\
&+ 15\,\gamma_j\,\gamma_e\,C_e\,C_j\cos(-\Omega_j - L_j + \Omega_e + L_i) + 9\,\gamma_j\,\gamma_e\,C_e\,C_j\cos(L_i + \Omega_j - L_j - \Omega_e) + \frac{3}{4}\,e_i\,e_j\cos(-\omega_i + 2\,L_i - \omega_j)\\
&+ \frac{9}{4}\,e_i\,e_j\cos(-\omega_i + 2\,L_i + \omega_j - 2\,L_j) + \frac{15}{2}\,\gamma_e{}^2\,C_e{}^2\cos(2\,\Omega_e - 3\,L_j + L_i) - \frac{9}{4}\,e_i\,e_j\cos(\omega_i - \omega_j)\\
&+ \frac{15}{2}\,\gamma_j{}^2\,C_j{}^2\cos(2\,\Omega_j - 3\,L_j + L_i) + \frac{3}{2}\,e_j\cos(-\omega_j + L_i) + \frac{3}{2}\,\gamma_e{}^2\,C_e{}^2\cos(L_i - 2\,\Omega_e + L_j) + \frac{3}{4}\,e_i\cos(-\omega_i + 2\,L_i - L_j)\\
&+ \frac{9}{2}\,e_j\cos(\omega_j - 2\,L_j + L_i) - \frac{9}{4}\,e_i\cos(\omega_i - L_j) + \frac{33}{16}\,e_j{}^2\cos(-2\,\omega_j + L_j + L_i) + \frac{3}{2}\,\gamma_i{}^2\cos(L_j + L_i - 2\,\Omega_i)\\
&+ \frac{159}{16}\,e_j{}^2\cos(2\,\omega_j - 3\,L_j + L_i)
\end{aligned}
$$

autres exemples :

```
> Windj(2,2,1,1);
```

$$(-15\,(\frac{1}{4}\,\frac{\sigma_i}{\sigma_j}-\frac{3}{4}\,\frac{\sigma_j}{\sigma_i})\,{\sigma_j}^2+3\,\%2)\,X_i+(-9\,\%2-15\,(-\frac{3}{4}\,\frac{\sigma_i}{\sigma_j}+\frac{1}{4}\,\frac{\sigma_j}{\sigma_i})\,{\sigma_j}^2)\,Xb_i+(18\,\%2-30\,\%1-15\,(\frac{1}{2}\,\frac{\sigma_i}{\sigma_j}+\frac{3}{2}\,\frac{\sigma_j}{\sigma_i})\,{\sigma_j}^2)\,X_j$$
$$+\,(30\,\%1-15\,(\frac{3}{2}\,\frac{\sigma_i}{\sigma_j}+\frac{1}{2}\,\frac{\sigma_j}{\sigma_i})\,{\sigma_j}^2+6\,\%2)\,Xb_j+6\,\%2-15\,\%1$$
$$\%1:=(\frac{1}{2}\,\frac{\sigma_j}{\sigma_i}+\frac{1}{2}\,\frac{\sigma_i}{\sigma_j})\,{\sigma_j}^2$$
$$\%2:=\sigma_i\,\sigma_j$$

```
> Windj(2,2,1,2);
```

$$\left(\frac{9}{4}\,\frac{eL_j}{{eL_e}^2}-\frac{15}{4}\,\frac{{eL_j}^3}{{eL_i}^2\,{eL_e}^2}\right)\,x_i+\left(-\frac{3}{4}\,\frac{{eL_i}^2\,eL_j}{{eL_e}^2}+\frac{45}{4}\,\frac{{eL_j}^3}{{eL_e}^2}\right)\,xb_i+\left(\frac{15}{2}\,\frac{{eL_j}^2}{{eL_e}^2\,eL_i}-\frac{3}{2}\,\frac{eL_i}{{eL_e}^2}\right)\,x_j+\left(-\frac{9}{2}\,\frac{eL_i\,{eL_j}^2}{{eL_e}^2}-\frac{75}{2}\,\frac{{eL_j}^4}{{eL_e}^2\,eL_i}\right)\,xb_j-\frac{3}{2}\,\frac{eL_i\,eL_j}{{eL_e}^2}$$
$$-\,\frac{15}{2}\,\frac{{eL_j}^3}{{eL_e}^2\,eL_i}$$

```
> Windj(2,2,1,3);
```

$$\frac{45}{4}\,e_i\cos(\omega_i-3\,L_j+2\,L_e)-\frac{3}{4}\,e_i\cos(2\,L_i-2\,L_e+L_j-\omega_i)+\frac{9}{4}\,e_i\cos(-2\,L_e+L_j+\omega_i)-\frac{9}{2}\,e_j\cos(L_i-2\,L_e+2\,L_j-\omega_j)$$
$$-\,\frac{15}{4}\,e_i\cos(-\omega_i+2\,L_i-3\,L_j+2\,L_e)-\frac{75}{2}\,e_j\cos(\omega_j-4\,L_j+2\,L_e+L_i)+\frac{15}{2}\,e_j\cos(-\omega_j-2\,L_j+2\,L_e+L_i)$$
$$-\,\frac{3}{2}\cos(L_i-2\,L_e+L_j)-\frac{3}{2}\,e_j\cos(L_i-2\,L_e+\omega_j)-\frac{15}{2}\cos(-3\,L_j+2\,L_e+L_i)+\frac{3}{4}I($$
$$-10\sin(-\omega_j-2\,L_j+2\,L_e+L_i)\,e_j-e_i\sin(2\,L_i-2\,L_e+L_j-\omega_i)-2\sin(L_i-2\,L_e+\omega_j)\,e_j$$
$$-\,15\,e_i\sin(\omega_i-3\,L_j+2\,L_e)-2\sin(L_i-2\,L_e+L_j)+3\,e_i\sin(-2\,L_e+L_j+\omega_i)+10\sin(-3\,L_j+2\,L_e+L_i)$$
$$+\,5\,e_i\sin(-\omega_i+2\,L_i-3\,L_j+2\,L_e)-6\sin(L_i-2\,L_e+2\,L_j-\omega_j)\,e_j+50\sin(\omega_j-4\,L_j+2\,L_e+L_i)\,e_j)$$

```
> Windj(2,-2,1,3);
```

$$\frac{45}{4}\,e_i\cos(\omega_i-3\,L_j+2\,L_e)-\frac{3}{4}\,e_i\cos(2\,L_i-2\,L_e+L_j-\omega_i)+\frac{9}{4}\,e_i\cos(-2\,L_e+L_j+\omega_i)-\frac{9}{2}\,e_j\cos(L_i-2\,L_e+2\,L_j-\omega_j)$$
$$-\frac{15}{4}\,e_i\cos(-\omega_i+2\,L_i-3\,L_j+2\,L_e)-\frac{75}{2}\,e_j\cos(\omega_j-4\,L_j+2\,L_e+L_i)+\frac{15}{2}\,e_j\cos(-\omega_j-2\,L_j+2\,L_e+L_i)$$
$$-\frac{3}{2}\cos(L_i-2\,L_e+L_j)-\frac{3}{2}\,e_j\cos(L_i-2\,L_e+\omega_j)-\frac{15}{2}\cos(-3\,L_j+2\,L_e+L_i)-\frac{3}{4}I($$
$$-10\sin(-\omega_j-2\,L_j+2\,L_e+L_i)\,e_j-e_i\sin(2\,L_i-2\,L_e+L_j-\omega_i)-2\sin(L_i-2\,L_e+\omega_j)\,e_j$$
$$-15\,e_i\sin(\omega_i-3\,L_j+2\,L_e)-2\sin(L_i-2\,L_e+L_j)+3\,e_i\sin(-2\,L_e+L_j+\omega_i)+10\sin(-3\,L_j+2\,L_e+L_i)$$
$$+5\,e_i\sin(-\omega_i+2\,L_i-3\,L_j+2\,L_e)-6\sin(L_i-2\,L_e+2\,L_j-\omega_j)\,e_j+50\sin(\omega_j-4\,L_j+2\,L_e+L_i)\,e_j)$$

```
> (Windj(2,2,1,3)+Windj(2,-2,1,3))/2;
```

$$\frac{45}{4}\,e_i\cos(\omega_i-3\,L_j+2\,L_e)-\frac{3}{4}\,e_i\cos(2\,L_i-2\,L_e+L_j-\omega_i)+\frac{9}{4}\,e_i\cos(-2\,L_e+L_j+\omega_i)-\frac{9}{2}\,e_j\cos(L_i-2\,L_e+2\,L_j-\omega_j)$$
$$-\frac{15}{4}\,e_i\cos(-\omega_i+2\,L_i-3\,L_j+2\,L_e)-\frac{75}{2}\,e_j\cos(\omega_j-4\,L_j+2\,L_e+L_i)+\frac{15}{2}\,e_j\cos(-\omega_j-2\,L_j+2\,L_e+L_i)$$
$$-\frac{3}{2}\cos(L_i-2\,L_e+L_j)-\frac{3}{2}\,e_j\cos(L_i-2\,L_e+\omega_j)-\frac{15}{2}\cos(-3\,L_j+2\,L_e+L_i)$$

Boucle permettant de calculer un ensemble de ces fonctions (ici toutes les harmoniques jusqu'à l'indice 6, dans le cas 2 permettant ensuite un calcul immédiat des équations de Lagrange):

```
> deb:=time():  ABindxy:=array(2..6,-6..6):
> for nj from 2 to 6 do:
> for np from -nj to nj do:
> ABindxy[nj,np]:=Windj(nj,np,degmx,2):
> od:
```

```
> od:
> print(time()-deb):
```

$$40.199$$

```
> save ABindxy , cat(chemin,‘ABindxy‘.degmx.‘.m‘):
```

Pour obtenir la partie indirecte de la perturbation du satellite S_i , il reste simplement à multiplier chaque élément (n, p) de ce tableau par $K\,\mu_j\,\frac{a_i\,{a_e}^n}{{a_j}^{(n+2)}}$ et par le coefficient $a_{n,\,p}$ ou $b_{n,\,p}$ suivant le signe de p. C'est ce qu'on réalise dans le paragraphe suivant.

4.6. Fonction perturbatrice indirecte de deux satellites massifs

On se contente ici de développer la partie indirecte des perturbations dues à l'aplatissement, la partie directe ayant été développée en 4.4 et les perturbations mutuelles (de type planétaire) en 3.5. Les perturbations solaires pourraient aussi être obtenues par les développements en polynômes de Legendre vus dans la section 3.

On commence par relire le développement construit dans la section précédente :

```
> degmx:=3;
> read cat(chemin,‘ABindxy3.m‘):
> read cat(chemin,‘EqLag.pr‘):
```

$$degmx := 3$$

Développement en coefficients complexes $a_{n,\,p}$ et $b_{n,\,p}$:

```
> n:='n':  p:='p':
> USiIndab:=expand(K*mu[j]*a[i]*sum(a[e]^n/a[j]^(n+2)*(a[n,0]*ABindxy[n,
> 0]+sum(a[n,p]*ABindxy[n,p]+b[n,p]*ABindxy[n,-p],p=1..n)),n=2..6)):
> nops(%);
```

$$10718$$

exemple de terme:

```
> op(190,USiIndab);
```

$$-\frac{945}{2}\,\frac{K\,\mu_j\,a_i\,{a_e}^3\,a_{3,\,3}\,xb_i\,y_j\,y_e\,{eL_j}^2\,C_e\,C_j}{{a_j}^5\,{eL_e}^3}$$

```
> save USiIndab, cat(chemin,‘USiIndab_‘.degmx.‘.m‘):
```

Développement en coefficients J_n, $c_{n,\,p}$ et $s_{n,\,p}$:

```
> USiIndJcs:=expand(subs(seq(a[n,0]=-J[n],n=2..6),
> seq(seq(a[n,p]=(c[n,p]-I*s[n,p])/2,p=1..n),n=2..6),
> seq(seq(b[n,p]=(c[n,p]+I*s[n,p])/2,p=1..n),n=2..6),USiIndab)):
> nops(%);
```

$$20202$$

exemple de terme

```
> op(190,USiIndJcs);
```

$$\frac{1575}{4}\,\frac{I\,K\,\mu_j\,a_i\,{a_e}^6\,C_e\,C_i\,{eL_e}^2\,x_i\,yb_i\,yb_e\,s_{6,\,2}}{{a_j}^8\,eL_j}$$

```
> save USiIndJcs, cat(chemin,‘USiIndJcs_‘.degmx.‘.m‘):
```

Ces développements sont alors prets à pouvoir être utilisés dans les équations de Lagrange relatives au satellite S_i. Pour l'autre satellite, S_j, il suffit d'échanger les indices *i* et *j* .

vers les équations de Lagrange

5. Equations du mouvement perturbé des planètes et des satellites

5.1. Equations de Lagrange

On considère ici les équations de Lagrange exprimées pour les variables osculatrices (a, x, xb, y, yb, L) décrivant les mouvements de planètes ou de satellites. Ces équations sont décrites par exemple dans le cours en (5.52) où les variables notées ici x et y s'appellent là z et ζ respectivement.

5.1.1. Procédure de calcul des équations de Lagrange

On donne ici la procédure de calcul des équations de Lagrange donnant les variations des éléments d'orbite (a_i, L_i, x_i, xb_i) d'une planète ou d'un satellite d'indice i dont la fonction perturbatrice *Upert* est elle-même exprimée en fonction de ces éléments et de ceux des autres planètes ou satellites (à cause de $\alpha = \frac{a_i}{a_j}$ qui doit être inférieur à 1 , il faut distinguer le cas intérieur du cas extérieur: Dans tous les cas, i est supposé la planète intérieure et j l'extérieure;

Notons que les variables en longitudes moyennes sont eL_i et eL_j, qui remplacent avantageusement les fonctions exponentielles du point de vue temps de calcul.

On peut utiliser cette procédure dans les cas intérieur et extérieur; on s'arrange pour permuter comme il faut les indices des deux corps: Le paramètre ii désigne implicitement le corps intérieur, et jj celui extérieur. On évalue dans la procédure le degré maxi de Upert en variables d'excentricités et d'inclinaisons, de façon à tronquer le résultat à ce degré ou à ce degré diminué de 1. En effet, à l'issue des dérivées par rapport aux C[i],on peut trouver dans dUsdyyb des termes de degré supérieur au degré maxi de Upert; ces termes sont alors incomplets, d'où leur élimination par troncature. De même, dUsdL est multiplié par x[i] (ou y[i]) dans dxSdt (ou dySdt), d'où la possibilité d'obtenir dans ces équations des termes de degré supérieur au degré de Upert, qu'il faut aussi éliminer. Enfin, dans les équations dxSdt et dySdt, le degré du résultat est forcément égal au degré de Upert diminué de 1 ; on tronque donc à ce degré.

```
> EqLag:=proc(Upert,cas,ii,jj) local
> i,j,U1,dmax,daSdt,dLSdt,dxSdt,dySdt,
> dUsda,dUsdL,dUsdx,dUsdxb,dUsdy,dUsdyb,dUsdxxb,dUsdyyb,var,ensvar;
```

```
> var :=[x[ii],xb[ii],x[jj],xb[jj],y[ii],yb[ii],y[jj],yb[jj]]:
> ensvar :={x[ii],xb[ii],x[jj],xb[jj],y[ii],yb[ii],y[jj],yb[jj]}:
> U1:=collect(Upert,var,distributed):
> dmax:=degree(U1,ensvar):
> U1:=expand(Upert):
> if cas="interieur" then i:=ii:  j:=jj:
> dUsda:=diff(U1,a[i])+diff(U1,alpha)/a[j]:
> elif cas="exterieur" then i:=jj:  j:=jj:
> dUsda:=diff(U1,a[i])-alpha*diff(U1,alpha)/a[j]:
> else print("cas doit valoir ""interieur"" ou ""exterieur"""):
> RETURN():
> fi:
> # phi[i]:=sqrt(1-x[i]*xb[i]):
> dUsdL:=I*eL[i]*diff(U1,eL[i]):
> dUsdx:=diff(U1,x[i]):
> dUsdxb:=diff(U1,xb[i]):
> dUsdyb:=diff(U1,yb[i])-diff(U1,C[i])*y[i]/2/C[i]:
> dUsdy:=diff(U1,y[i])-diff(U1,C[i])*yb[i]/2/C[i]:
> dUsdxxb:=x[i]*dUsdx+xb[i]*dUsdxb:
> dUsdyyb:=y[i]*dUsdy+yb[i]*dUsdyb:
> daSdt:=2/n[i]/a[i]*dUsdL:
> dLSdt:=TRONC(-2/n[i]/a[i]*dUsda +
> phi[i]/n[i]/a[i]^2/(1+phi[i])*dUsdxxb +
> 1/2/n[i]/a[i]^2/phi[i]*dUsdyyb,dmax,var):
> dxSdt:=TRONC(2*I*phi[i]/n[i]/a[i]^2*dUsdxb -
> phi[i]/n[i]/a[i]^2/(1+phi[i])*x[i]*dUsdL +
> I/2/n[i]/a[i]^2/phi[i]*x[i]*dUsdyyb, dmax-1,var):
> dySdt:=TRONC(I/2/n[i]/a[i]^2/phi[i]*dUsdyb -
> 1/2/n[i]/a[i]^2/phi[i]*y[i]*dUsdL -
> I/2/n[i]/a[i]^2/phi[i]*y[i]*(x[i]*dUsdx-xb[i]*dUsdxb), dmax-1,var):
```

```
> RETURN([daSdt,dLSdt,dxSdt,dySdt]):
> end:
```

petite vérification pour voir si la dérivation des fonctions hypergéométriques s'effectue convenablement :

```
> # U1:=15/64*alpha^5*hypergeom([7/2, 5/2],[4],alpha^2);
> # diff(U1,a[i])+diff(U1,alpha)/a[j];
```

5.1.2. Equations de Lagrange pour deux planètes, calcul complet (degré faible)

Fonction perturbatrice planétaire issue d'un calcul complet mais de degré faible en α et en excentricités et inclinaisons (signet : EqLag(UP))

Equations de Lagrange pour une planète **intérieure** (indice *i*) perturbée par une planète **extérieure** (indice *j*)

```
> Up:=TRONC(UP[i],2,varxyij):
```

Up contient mélangées toutes les inégalités concernées par le degré choisi; voici par exemple l'un des monômes :

```
> nops(Up);factor(op(17,Up));
```

$$67$$

$$-\frac{1}{2048} K\, m_j\, \alpha^2\, C_i\, C_j\, yb_i\, yb_j (5040\, \alpha^3\, {eL_i}^{11}\, {eL_j}^3 + 5544\, \alpha^4\, {eL_i}^{12}\, {eL_j}^2 + 4480\, \alpha^2\, {eL_i}^{10}\, {eL_j}^4 + 4608\, \alpha\, {eL_j}^7\, {eL_i}^7$$
$$+ 3840\, \alpha\, {eL_i}^9\, {eL_j}^5 + 4480\, \alpha^2\, {eL_j}^{10}\, {eL_i}^4 + 3840\, \alpha\, {eL_j}^9\, {eL_i}^5 + 5040\, \alpha^3\, {eL_j}^{11}\, {eL_i}^3 + 5544\, \alpha^4\, {eL_j}^{12}\, {eL_i}^2$$
$$+ 6006\, {eL_i}^{13}\, \alpha^5\, eL_j + 6006\, {eL_j}^{13}\, \alpha^5\, eL_i + 6435\, {eL_j}^{14}\, \alpha^6 + 6435\, {eL_i}^{14}\, \alpha^6 + 8316\, \alpha^5\, {eL_i}^{11}\, {eL_j}^3 + 9009\, \alpha^6\, {eL_i}^{12}\, {eL_j}^2$$
$$+ 7560\, \alpha^4\, {eL_i}^{10}\, {eL_j}^4 + 10395\, \alpha^6\, {eL_i}^{10}\, {eL_j}^4 + 11025\, {eL_i}^8\, {eL_j}^6\, \alpha^6 + 3072\, {eL_i}^8\, {eL_j}^6 + 6720\, \alpha^3\, {eL_i}^9\, {eL_j}^5$$
$$+ 9450\, \alpha^5\, {eL_i}^9\, {eL_j}^5 + 8400\, {eL_i}^8\, {eL_j}^6\, \alpha^4 + 5760\, {eL_i}^8\, {eL_j}^6\, \alpha^2 + 3072\, {eL_j}^8\, {eL_i}^6 + 9800\, \alpha^5\, {eL_j}^7\, {eL_i}^7$$
$$+ 7200\, \alpha^3\, {eL_j}^7\, {eL_i}^7 + 5760\, {eL_j}^8\, {eL_i}^6\, \alpha^2 + 11025\, {eL_j}^8\, {eL_i}^6\, \alpha^6 + 6720\, \alpha^3\, {eL_j}^9\, {eL_i}^5 + 9450\, \alpha^5\, {eL_j}^9\, {eL_i}^5$$
$$+ 8400\, {eL_j}^8\, {eL_i}^6\, \alpha^4 + 9009\, \alpha^6\, {eL_j}^{12}\, {eL_i}^2 + 8316\, \alpha^5\, {eL_j}^{11}\, {eL_i}^3 + 7560\, \alpha^4\, {eL_j}^{10}\, {eL_i}^4 + 10395\, \alpha^6\, {eL_j}^{10}\, {eL_i}^4) \Big/ ($$
$$a_j\, {eL_j}^6\, {eL_i}^6)$$

On pourrait cependant en extraire une inégalité (pp,qq) de cette façon:

```
> pp:=1:  qq:=-2:
> TRONC(coeff(coeff(expand(UP[i]),eL[i],pp),eL[j],qq),2,varxyij)*eL[i]^p
> p*eL[j]^qq;
```

$$\frac{((-\frac{945}{2048}\, \frac{K\, m_j\, \alpha^8}{a_j} - \frac{9}{8}\, \frac{K\, m_j\, \alpha^2}{a_j} - \frac{525}{1024}\, \frac{K\, m_j\, \alpha^6}{a_j} - \frac{5}{8}\, \frac{K\, m_j\, \alpha^4}{a_j})\, x_i + (\frac{9}{16}\, \frac{K\, m_j\, \alpha^3}{a_j} + \frac{875}{2048}\, \frac{K\, m_j\, \alpha^7}{a_j} + \frac{15}{32}\, \frac{K\, m_j\, \alpha^5}{a_j})\, x_j)\, eL_i}{{eL_j}^2}$$

Calcul des équations de Lagrange :

Equations de Lagrange pour une planète **intérieure** (indice *i*)perturbée par une planète **extérieure** (indice *j*)

```
> DVin:=EqLag(Up,"interieur",i,j):  print(seq(nops(DVin[k]),k=1..4));
```

$$6,\ 25,\ 5,\ 4$$

Equations de Lagrange pour une planète **extérieure** (indice *j*)perturbée par une planète **intérieure** (indice *i*)

```
> Up:=TRONC(UP[j],2,varxyij):  nops(Up);
> DVix:=EqLag(Up,"exterieur",i,j):  print(seq(nops(DVix[k]),k=1..4));
```

$$71$$

$$6,\ 25,\ 5,\ 4$$

Pour limiter l'affichage du résultat, extraction d'une inégalité :

```
> ppi:=-1:  ppj:=2:
> for k to 4 do:
> coeff(coeff(DVin[k],eL[i],ppi),eL[j],ppj):
> print('DVin['.k.']=',collect(
> %,varxyij,distributed,factor)*exp(I*(ppi*L[i]+ppj*L[j])));
> od:
```

$$DVin[1] =, \left(\frac{1}{1024}\,\frac{I\,m_j\,(945\,\alpha^6 + 1280\,\alpha^2 + 2304 + 1050\,\alpha^4)\,K\,\alpha^2\,xb_i}{n_i\,a_i\,a_j} - \frac{1}{1024}\,\frac{I\,K\,m_j\,\alpha^3\,(875\,\alpha^4 + 1152 + 960\,\alpha^2)\,xb_j}{n_i\,a_i\,a_j}\right) e^{(I\,(-L_i+2\,L_j))}$$

$$DVin[2] =, \left(\frac{1}{2048} K\,m_j\,\alpha(12600\,a_i\,\alpha^4 + 12600\,a_i\,\alpha^4\,\phi_i + 15120\,a_i\,\alpha^6 + 15120\,a_i\,\alpha^6\,\phi_i + 10240\,a_i\,\alpha^2 + 10240\,a_i\,\alpha^2\,\phi_i + 9216\,a_i + 9216\,a_i\,\phi_i - 945\,\phi_i\,\alpha^7\,a_j - 1280\,\phi_i\,\alpha^3\,a_j - 2304\,\phi_i\,\alpha\,a_j - 1050\,\phi_i\,\alpha^5\,a_j)xb_i \Big/ (n_i\,{a_i}^2\,{a_j}^2\,(1+\phi_i)) - \frac{1}{1024}\,\frac{K\,m_j\,\alpha^2\,(3456 + 4800\,\alpha^2 + 6125\,\alpha^4)\,xb_j}{n_i\,a_i\,{a_j}^2}\right) e^{(I\,(-L_i+2\,L_j))}$$

$$DVin[3] =, -\frac{1}{1024}\,\frac{I\,\phi_i\,K\,m_j\,\alpha^2\,(945\,\alpha^6 + 1280\,\alpha^2 + 2304 + 1050\,\alpha^4)\,e^{(I\,(-L_i+2\,L_j))}}{n_i\,{a_i}^2\,a_j}$$

$$DVin[4] =, 0$$

```
> for k to 4 do:
> coeff(coeff(DVix[k],eL[i],ppi),eL[j],ppj):
> print(‘DVix[‘.k.‘]=‘,collect(
> %,varxyij,distributed,factor)*exp(I*(ppi*L[i]+ppj*L[j])));
> od:
```

$$DVix[1] =, \left(-\frac{1}{512}\frac{I\,K\,m_i\,\alpha^2\,(2304+945\,\alpha^6+1280\,\alpha^2+1050\,\alpha^4)\,xb_i}{n_j\,{a_j}^2} + \frac{1}{512}\frac{I\,K\,m_i\,(2048\,\alpha^3+875\,\alpha^9-512+960\,\alpha^7+1152\,\alpha^5)\,xb_j}{n_j\,{a_j}^2\,\alpha^2}\right)e^{(I\,(-L_i+2\,L_j))}$$

$$DVix[2] =, \left(-\frac{1}{1024}\frac{K\,m_i\,\alpha^2\,(8505\,\alpha^6+6912+6400\,\alpha^2+7350\,\alpha^4)\,xb_i}{n_j\,{a_j}^3} + \frac{1}{2048}K\,m_i(14000\,\alpha^9+14875\,\alpha^9\,\phi_j+8192\,\alpha^3 + 10240\,\alpha^3\,\phi_j+1024+512\,\phi_j+11520\,\alpha^7+12480\,\alpha^7\,\phi_j+9216\,\alpha^5+10368\,\alpha^5\,\phi_j)xb_j\Big/(\alpha^2\,n_j\,{a_j}^3\,(1+\phi_j))\right) e^{(I\,(-L_i+2\,L_j))}$$

$$DVix[3] =, \frac{1}{1024}\frac{I\,\phi_j\,(2048\,\alpha^3+875\,\alpha^9-512+960\,\alpha^7+1152\,\alpha^5)\,K\,m_i\,e^{(I\,(-L_i+2\,L_j))}}{n_j\,{a_j}^3\,\alpha^2}$$

$$DVix[4] =, 0$$

autre inégalité

5.1.3. Equations de Lagrange pour deux planètes, calcul par inégalités

Fonction perturbatrice planétaire issue d'un calcul par inégalités (signet=EqLag(UPa))

Equations de Lagrange pour une planète **intérieure** (indice i) perturbée par une planète **extérieure** (indice j)

```
> Up:=TRONC(expand(UPa[i]),2,varxyij);nops(%);op(2,Up):
```

$$Up := \left(-\frac{3}{16}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^3\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)}{a_j\,eL_j}-\frac{1}{4}\,\frac{K\,m_j\,{eL_i}^2\,\alpha}{a_j\,eL_j}+\frac{1}{4}\,\frac{K\,m_j\,{eL_i}^2\,\alpha\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2)}{a_j\,eL_j}\right)xb_i$$
$$+\left(\frac{5}{32}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^4\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2)}{a_j\,eL_j}-\frac{3}{16}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^2\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)}{a_j\,eL_j}\right)xb_j$$

$$2$$

```
> DVai:=EqLag(Up,"interieur",i,j):  seq(nops(DVai[k]),k=1..4);
```

$$6,\ 2,\ 6,\ 1$$

Affichage des 4 équations:

```
> for k to 4 do:
> coeff(coeff(DVai[k],eL[i],pp),eL[j],qq):
> print(`DVai[`.k.`]=`,collect(
> %,varxyij,distributed,factor)*exp(I*(pp*L[i]+qq*L[j])));
> od:
```

$$DVai[1] =, \left(\frac{1}{4}\,\frac{I\,K\,m_j\,\alpha\,(-3\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2) - 4 + 4\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2))\,xb_i}{n_i\,a_i\,a_j} - \frac{1}{8}\,\frac{I\,K\,m_j\,\alpha^2\,(-5\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2) + 6\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2))\,xb_j}{n_i\,a_i\,a_j}\right)\,e^{(I\,(2\,L_i - L_j))}$$

$$\begin{aligned}DVai[2] =, \Big(&-\frac{1}{16}K\,m_j(-12\,a_i\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2) - 12\,a_i\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)\,\phi_i - 15\,a_i\,\alpha^4\,\mathrm{F}([\frac{5}{2},\,\frac{7}{2}],\,[4],\,\alpha^2)\\ &- 15\,a_i\,\alpha^4\,\mathrm{F}([\frac{5}{2},\,\frac{7}{2}],\,[4],\,\alpha^2)\,\phi_i - 8\,a_i - 8\,a_i\,\phi_i + 8\,a_i\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2) + 8\,a_i\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2)\,\phi_i\\ &+ 3\,\phi_i\,\alpha^3\,a_j\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2) + 4\,\phi_i\,\alpha\,a_j - 4\,\phi_i\,\alpha\,a_j\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2))xb_i \Big/ (n_i\,{a_i}^2\,{a_j}^2\,(1+\phi_i))\\ &- \frac{3}{128}\,\frac{K\,m_j\,\alpha\,(40\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2) + 35\,\alpha^4\,\mathrm{F}([\frac{5}{2},\,\frac{9}{2}],\,[5],\,\alpha^2) - 32\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2))\,xb_j}{n_i\,a_i\,{a_j}^2}\Big)\,e^{(I\,(2\,L_i - L_j))}\end{aligned}$$

$$DVai[3] =, \frac{1}{8}\,\frac{I\,\phi_i\,K\,m_j\,\alpha\,(-3\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2) - 4 + 4\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2))\,e^{(I\,(2\,L_i - L_j))}}{n_i\,{a_i}^2\,a_j}$$

$$DVai[4] =, 0$$

Equations de Lagrange pour une planète **extérieure** (indice *j*)perturbée par une planète **intérieure** (indice *i*)

```
> Up:=TRONC(expand(UPa[j]),2,varxyij);nops(%);op(2,Up):
```

$$Up := \left(-\frac{3}{16}\,\frac{K\,m_i\,{eL_i}^2\,\alpha^3\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)}{a_j\,eL_j} + \frac{1}{4}\,\frac{K\,m_i\,{eL_i}^2\,\alpha\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2)}{a_j\,eL_j} - \frac{K\,m_i\,{eL_i}^2}{a_j\,\alpha^2\,eL_j}\right) xb_i$$
$$+ \left(\frac{5}{32}\,\frac{K\,m_i\,{eL_i}^2\,\alpha^4\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2)}{a_j\,eL_j} - \frac{3}{16}\,\frac{K\,m_i\,{eL_i}^2\,\alpha^2\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)}{a_j\,eL_j}\right) xb_j$$

$$2$$

```
> DVaj:=EqLag(Up,"exterieur",i,j):    seq(nops(DVaj[k]),k=1..4);
```

$$6,\ 2,\ 6,\ 1$$

Affichage des 4 équations:

```
> for k to 4 do:
> coeff(coeff(DVaj[k],eL[i],pp),eL[j],qq):
> print(`DVaj[`.k.`]=`,collect(
> %,varxyij,distributed,factor)*exp(I*(pp*L[i]+qq*L[j])));
> od:
```

$$DVaj[1] =, \left(-\frac{1}{8}\,\frac{I\,K\,m_i\,(-3\,\alpha^5\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2) + 4\,\alpha^3\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2) - 16)\,xb_i}{n_j\,{a_j}^2\,\alpha^2}\right.$$
$$\left. + \frac{1}{16}\,\frac{I\,K\,m_i\,\alpha^2\,(-5\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2) + 6\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2))\,xb_j}{n_j\,{a_j}^2}\right) e^{(I\,(2\,L_i - L_j))}$$

$$DVaj[2] =, \left(\frac{1}{16}\,\frac{K\,m_i\,(32-18\,\alpha^5\,\mathrm{F}([\frac{3}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)+16\,\alpha^3\,\mathrm{F}([\frac{1}{2},\,\frac{3}{2}],\,[2],\,\alpha^2)-15\,\alpha^7\,\mathrm{F}([\frac{5}{2},\,\frac{7}{2}],\,[4],\,\alpha^2))\,xb_i}{\alpha^2\,n_j\,{a_j}^3}-\frac{1}{128}K\,m_i\,\alpha^2(\right.$$

$$144\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)+168\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2)\,\phi_j-160\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2)-180\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2)\,\phi_j$$

$$\left.-105\,\alpha^4\,\mathrm{F}([\frac{5}{2},\,\frac{9}{2}],\,[5],\,\alpha^2)-105\,\alpha^4\,\mathrm{F}([\frac{5}{2},\,\frac{9}{2}],\,[5],\,\alpha^2)\,\phi_j)xb_j\Big/(n_j\,{a_j}^3\,(1+\phi_j))\right)e^{(I\,(2\,L_i-L_j))}$$

$$DVaj[3] =, -\frac{1}{16}\,\frac{I\,\phi_j\,K\,m_i\,\alpha^2\,(-5\,\alpha^2\,\mathrm{F}([\frac{3}{2},\,\frac{7}{2}],\,[4],\,\alpha^2)+6\,\mathrm{F}([\frac{1}{2},\,\frac{5}{2}],\,[3],\,\alpha^2))\,e^{(I\,(2\,L_i-L_j))}}{n_j\,{a_j}^3}$$

$$DVaj[4] =, 0$$

calculer une nouvelle inégalité

Fonction perturbatrice planétaire issue d'un calcul d'inégalité de degré limité en α (signet : EqLag(UPTr))

Equations de Lagrange pour une planète **intérieure** (indice *i*) perturbée par une planète **extérieure** (indice *j*)

```
>  Up:=TRONC(expand(UPTr[i]),2,varxyij);nops(%);op(2,Up):
```

$$Up := (-\frac{875}{4096}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^7}{a_j\,eL_j}-\frac{3}{32}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^3}{a_j\,eL_j}-\frac{45}{256}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^5}{a_j\,eL_j})\,xb_i$$
$$+(\frac{315}{2048}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^6}{a_j\,eL_j}+\frac{5}{64}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^4}{a_j\,eL_j}-\frac{3}{16}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^2}{a_j\,eL_j}+\frac{1575}{8192}\,\frac{K\,m_j\,{eL_i}^2\,\alpha^8}{a_j\,eL_j})\,xb_j$$

$$2$$

```
> DVTri:=EqLag(Up,"interieur",i,j):  seq(nops(DVTri[k]),k=1..4);
```

$$6,\ 2,\ 6,\ 1$$

Affichage des 4 équations:

```
> for k to 4 do:
> coeff(coeff(DVTri[k],eL[i],pp),eL[j],qq):
> print(‘DVTri[‘.k.‘]=‘,collect(
> %,varxyij,distributed,factor)*exp(I*(pp*L[i]+qq*L[j])));
> od:
```

$$DVTri[1]=,\left(-\frac{1}{1024}\,\frac{I\,K\,m_j\,\alpha^3\,(875\,\alpha^4+384+720\,\alpha^2)\,xb_i}{n_i\,a_i\,a_j}+\frac{1}{2048}\,\frac{I\,K\,m_j\,\alpha^2\,(1260\,\alpha^4+640\,\alpha^2-1536+1575\,\alpha^6)\,xb_j}{n_i\,a_i\,a_j}\right)e^{(I\,(2\,L_i-L_j))}$$

$$DVTri[2]=,\left(\frac{1}{4096}K\,m_j\,\alpha^2(12250\,a_i\,\alpha^4+12250\,a_i\,\alpha^4\,\phi_i+2304\,a_i+2304\,a_i\,\phi_i+7200\,a_i\,\alpha^2+7200\,a_i\,\alpha^2\,\phi_i - 875\,\phi_i\,\alpha^5\,a_j-384\,\phi_i\,\alpha\,a_j-720\,\phi_i\,\alpha^3\,a_j)xb_i\Big/(n_i\,{a_i}^2\,{a_j}^2\,(1+\phi_i)) - \frac{1}{512}\,\frac{K\,m_j\,\alpha\,(945\,\alpha^4+320\,\alpha^2-384+1575\,\alpha^6)\,xb_j}{n_i\,a_i\,{a_j}^2}\right)e^{(I\,(2\,L_i-L_j))}$$

$$DVTri[3]=,-\frac{1}{2048}\,\frac{I\,\phi_i\,K\,m_j\,\alpha^3\,(875\,\alpha^4+384+720\,\alpha^2)\,e^{(I\,(2\,L_i-L_j))}}{n_i\,{a_i}^2\,a_j}$$

$$DVTri[4]=,0$$

Equations de Lagrange pour une planète **extérieure** (indice *j*)perturbée par une planète **intérieure** (indice *i*)

```
> Up:=TRONC(expand(UPTr[j]),2,varxyij);nops(%);op(2,Up):
```

$$Up := \left(-\frac{K\,m_i\,{eL_i}^2}{a_j\,\alpha^2\,eL_j} - \frac{45}{256}\,\frac{K\,m_i\,{eL_i}^2\,\alpha^5}{a_j\,eL_j} - \frac{3}{32}\,\frac{K\,m_i\,{eL_i}^2\,\alpha^3}{a_j\,eL_j} + \frac{1}{4}\,\frac{K\,m_i\,{eL_i}^2\,\alpha}{a_j\,eL_j} - \frac{875}{4096}\,\frac{K\,m_i\,{eL_i}^2\,\alpha^7}{a_j\,eL_j}\right) xb_i + \left(\frac{1575}{8192}\,\frac{K\,m_i\,\alpha^8\,{eL_i}^2}{a_j\,eL_j} + \frac{5}{64}\,\frac{K\,m_i\,{eL_i}^2\,\alpha^4}{a_j\,eL_j} - \frac{3}{16}\,\frac{K\,m_i\,{eL_i}^2\,\alpha^2}{a_j\,eL_j} + \frac{315}{2048}\,\frac{K\,m_i\,\alpha^6\,{eL_i}^2}{a_j\,eL_j}\right) xb_j$$

$$2$$

```
> DVTrj:=EqLag(Up,"exterieur",i,j):  seq(nops(DVTrj[k]),k=1..4);
```

$$6,\ 2,\ 6,\ 1$$

Affichage des 4 équations:

```
> for k to 4 do:
> coeff(coeff(DVTrj[k],eL[i],pp),eL[j],qq):
> print(`DVTrj[`.k.`]=`,collect(
> %,varxyij,distributed,factor)*exp(I*(pp*L[i]+qq*L[j])));
> od:
```

$$DVTrj[1] =, \left(\frac{1}{2048}\,\frac{I\,K\,m_i\,(4096 + 720\,\alpha^7 + 384\,\alpha^5 - 1024\,\alpha^3 + 875\,\alpha^9)\,xb_i}{n_j\,{a_j}^2\,\alpha^2} - \frac{1}{4096}\,\frac{I\,K\,m_i\,\alpha^2\,(1575\,\alpha^6 + 640\,\alpha^2 - 1536 + 1260\,\alpha^4)\,xb_j}{n_j\,{a_j}^2}\right) e^{(I\,(2\,L_i - L_j))}$$

$$DVTrj[2] =, \left(-\frac{1}{256}\,\frac{K\,m_i\,(-512+192\,\alpha^5-256\,\alpha^3+875\,\alpha^9+540\,\alpha^7)\,xb_i}{\alpha^2\,n_j\,{a_j}^3}+\frac{1}{8192}\right.$$
$$\left.\frac{K\,m_i\,\alpha^2\,(17640\,\alpha^4+18900\,\alpha^4\,\phi_j+28350\,\alpha^6+29925\,\alpha^6\,\phi_j+6400\,\alpha^2+7040\,\alpha^2\,\phi_j-9216-10752\,\phi_j)\,xb_j}{n_j\,{a_j}^3\,(1+\phi_j)}\right)$$
$$e^{(I\,(2\,L_i-L_j))}$$

$$DVTrj[3] =, \frac{1}{4096}\,\frac{I\,\phi_j\,K\,m_i\,\alpha^2\,(1575\,\alpha^6+640\,\alpha^2-1536+1260\,\alpha^4)\,e^{(I\,(2\,L_i-L_j))}}{n_j\,{a_j}^3}$$

$$DVTrj[4] =, 0$$

calculer une nouvelle inégalité

Fonction perturbatrice planétaire issue d'un calcul numérique en a (signet : EqLag(UPn)):

Equations de Lagrange pour une planète **intérieure** (indice *i*) perturbée par une planète **extérieure** (indice *j*)

```
> Up:=TRONC(expand(UPna[i]),2,varxyij);nops(%);
```

$$Up := (.2235447116\,\frac{K\,m_j\,{eL_i}^2}{a_j\,eL_j}-.443102338225822874\,\frac{K\,m_j\,{eL_i}^2\,\alpha}{a_j\,eL_j})\,xb_i$$
$$+(.0575031048904993 79\,\frac{K\,m_j\,{eL_i}^2\,\alpha}{a_j\,eL_j}-.08027971757\,\frac{K\,m_j\,{eL_i}^2}{a_j\,eL_j})\,xb_j$$

$$2$$

```
> DVni:=EqLag(Up,"interieur",i,j):  seq(nops(DVni[k]),k=1..4);
```

$$6,\ 2,\ 6,\ 1$$

Affichage des 4 équations:

```
> for k to 4 do:
> co:=coeff(coeff(DVni[k],eL[i],pp),eL[j],qq):
> co:=collect(co,varxyij,distributed):
> co:=collect(co,[K,m[i],m[j],a[i],a[j],eL[i],eL[j]],distributed):
> print(`DVni[`.k.`]=`,co*exp(I*(pp*L[i]+qq*L[j])));
> print(`DVni[`.k.`]=`,evalf(subs(alpha=alpha0,co),Precision)*exp(I*(pp*
> L[i]+qq*L[j])));
> od:
```

$$DVni[1] =, \frac{\left(2.\,\frac{I\,(.4470894232 - .8862046764\,\alpha)\,xb_i}{n_i} + 2.\,\frac{I\,(.1150062098\,\alpha - .1605594351)\,xb_j}{n_i}\right) K\,m_j\,e^{(I\,(2\,L_i - L_j))}}{a_i\,a_j}$$

$$DVni[1] =, \frac{\left(-.22018805345063727 4\,\frac{I\,xb_i}{n_i} - .176503165847124641\,\frac{I\,xb_j}{n_i}\right) K\,m_j\,e^{(I\,(2\,L_i - L_j))}}{a_i\,a_j}$$

$$DVni[2] =, \left(\frac{\left(.8862046764\,\frac{xb_i}{n_i} - .1150062098\,\frac{xb_j}{n_i}\right) K\,m_j}{{a_j}^2\,a_i} + \frac{\phi_i\,(.2235447116 - .4431023382\,\alpha)\,xb_i\,K\,m_j}{n_i\,(1. + \phi_i)\,a_j\,{a_i}^2}\right) e^{(I\,(2\,L_i - L_j))}$$

$$DVni[2] =, \left(\frac{\left(.8862046764\,\frac{xb_i}{n_i} - .1150062098\,\frac{xb_j}{n_i}\right) K\,m_j}{{a_j}^2\,a_i} - .055047013362659319\,\frac{\phi_i\,xb_i\,K\,m_j}{n_i\,(1. + \phi_i)\,a_j\,{a_i}^2}\right) e^{(I\,(2\,L_i - L_j))}$$

$$DVni[3] =, 2.\,\frac{I\,\phi_i\,(.2235447116 - .4431023382\,\alpha)\,K\,m_j\,e^{(I\,(2\,L_i - L_j))}}{n_i\,{a_i}^2\,a_j}$$

$$DVni[3] =, -.110094026725318638\,\frac{I\,\phi_i\,K\,m_j\,e^{(I\,(2\,L_i-L_j))}}{n_i\,{a_i}^2\,a_j}$$

$$DVni[4] =, 0$$

$$DVni[4] =, 0$$

Equations de Lagrange pour une planète **extérieure** (indice *j*)perturbée par une planète **intérieure** (indice *i*)

```
> Up:=TRONC(expand(UPna[j]),2,varxyij);nops(%);
```

$$Up := (-7.365601386\,\frac{K\,m_i\,{eL_i}^2}{a_j\,eL_j} + 7.85396218804779255\,\frac{K\,m_i\,{eL_i}^2\,\alpha}{a_j\,eL_j})\,xb_i + (-.08027971757\,\frac{K\,m_i\,{eL_i}^2}{a_j\,eL_j} + .057503104890499379\,\frac{K\,m_i\,{eL_i}^2\,\alpha}{a_j\,eL_j})\,xb_j$$

$$2$$

```
> DVnj:=EqLag(Up,"exterieur",i,j):  seq(nops(DVnj[k]),k=1..4);
```

$$6,\ 2,\ 6,\ 1$$

Affichage des 4 équations:

```
> for k to 4 do:
> co:=coeff(coeff(DVnj[k],eL[i],pp),eL[j],qq):
> co:=collect(co,varxyij,distributed):
> co:=collect(co,[K,m[i],m[j],a[i],a[j],eL[i],eL[j]],distributed):
> print(‘DVnj[‘.k.‘]=‘,co*exp(I*(pp*L[i]+qq*L[j])));
> print(‘DVnj[‘.k.‘]=‘,evalf(subs(alpha=alpha0,co),Precision)*exp(I*(pp*
> L[i]+qq*L[j])));
> od:
```

$$DVnj[1]=,\ \frac{\left(2.\ \frac{I\,(7.365601386-7.853962188\,\alpha)\,xb_i}{n_j}+2.\ \frac{I\,(.08027971757-.05750310489\,\alpha)\,xb_j}{n_j}\right)K\,m_i\,e^{(I\,(2\,L_i-L_j))}}{{a_j}^2}$$

$$DVnj[1]=,\ \frac{\left(4.85515976742069354\ \frac{I\,xb_i}{n_j}+.0882515829761369200\ \frac{I\,xb_j}{n_j}\right)K\,m_i\,e^{(I\,(2\,L_i-L_j))}}{{a_j}^2}$$

$$DVnj[2]=,((-2.\ \frac{(-15.70792438\,\alpha+7.365601386)\,xb_i}{n_j}$$
$$+\left(-2.\ \frac{-.1150062098\,\alpha+.08027971757}{n_j}+\frac{\phi_j\,(-.08027971757+.05750310489\,\alpha)}{n_j\,(1.+\phi_j)}\right)xb_j)K\,m_i\,e^{(I\,(2\,L_i-L_j))})\Big/$$
$${a_j}^3$$

$$DVnj[2]=,((5.02088324218845278\ \frac{xb_i}{n_j}+\left(-.0159437307871246410\ \frac{1}{n_j}-.0441257914880684600\ \frac{\phi_j}{n_j\,(1.+\phi_j)}\right)xb_j)\,K$$
$$m_i\,e^{(I\,(2\,L_i-L_j))})\Big/{a_j}^3$$

$$DVnj[3]=,\ 2.\ \frac{I\,\phi_j\,(-.08027971757+.05750310489\,\alpha)\,K\,m_i\,e^{(I\,(2\,L_i-L_j))}}{n_j\,{a_j}^3}$$

$$DVnj[3]=,\ -.0882515829761369200\ \frac{I\,\phi_j\,K\,m_i\,e^{(I\,(2\,L_i-L_j))}}{n_j\,{a_j}^3}$$

$$DVnj[4]=,\ 0$$

$$DVnj[4]=,\ 0$$

calculer une nouvelle inégalité

5.1.4. Equations de Lagrange pour un satellite perturbé par l'aplatissement de sa planète

```
> # read cat(chemin,‘USiJcs_6.m‘):
```

On utilise la fonction perturbatrice de satellite par l'aplatissement, construite en 4.4.1 sous le nom *USiJcs* , et dont on peut éventuellement extraire une partie (ici par exemple les termes en J_2, $c_{2,\,p}$ et $s_{2,\,p}$, en annulant tous les autres coefficients d'aplatissement)

```
> Up:=subs(seq(J[k]=0,k=3..6),seq(seq(c[k,p]=0,p=1..k),k=3..6),
> seq(seq(s[k,p]=0,p=1..k),k=3..6),USiJcs):  nops(Up);
```

$$2788$$

On peut aussi sélectionner par exemple les seuls termes en J_2 ou en $a_{2,\,2}$ ou en $b_{4,\,2}$:

```
> Up:=expand(coeff(USiJcs,J[2])*J[2]):  nops(%);
```

$$292$$

```
> Up:=expand(coeff(USiab,a[2,2])*a[2,2]):nops(%);
```

$$241$$

```
> Up:=expand(coeff(USiab,b[4,2])*b[4,2]):nops(%);
```

$$771$$

Calcul des équations de Lagrange avec cette sélection de fonction perturbatrice :

```
> DVSi:=EqLag(Up,"interieur",i,e):
> print(seq(nops(expand(DVSi[k])),k=1..4));
```

$$254,\ 809,\ 301,\ 209$$

Pour limiter l'affichage du résultat, voici comment restreindre le résultat à une inégalité (*ppi, ppj*) donnée . On y fait aussi la substitution des fonctions des excentricités et inclinaisons. Le degré 6 qu'on pourrait adopter dans la fonction mtaylor permettrait d'obtenir les développements au degré 5, c'est-à-dire le degré convenable pour une fonction perturbatrice développée au degré 6. Ici, on se contente, pour limiter l'affichage, du degré 4 :

```
> ppi:=1:  ppj:=0:
> for k to 4 do:
> coeff(coeff(DVSi[k],eL[i],ppi),eL[e],ppj):
> subs(phi[i]=sqrt(1-x[i]*xb[i]),C[i]=sqrt(1-y[i]*yb[i]),
> C[e]=sqrt(1-y[e]*yb[e]),%):
> mtaylor(%,varxyie,4):
> print('DVSi['.k.']=',
> sort(collect(
> %,varxyie,distributed,factor),varxyie)*exp(I*(ppi*L[i]+ppj*L[e])));
> od:
```

$$DVSi[1] =, \left(\frac{27}{16}\,\frac{I\,K*(M+mu[i])\,{a_e}^2\,J_2\,x_i\,{xb_i}^2}{n_i\,{a_i}^4} - \frac{3}{2}\,\frac{I\,K*(M+mu[i])\,{a_e}^2\,J_2\,x_i\,{yb_i}^2}{n_i\,{a_i}^4} + 3\,\frac{I\,K*(M+mu[i])\,{a_e}^2\,J_2\,x_i\,yb_i\,yb_e}{n_i\,{a_i}^4} - \frac{3}{2}\,\frac{I\,K*(M+mu[i])\,{a_e}^2\,J_2\,x_i\,{yb_e}^2}{n_i\,{a_i}^4} - 9\,\frac{I\,K*(M+mu[i])\,{a_e}^2\,J_2\,xb_i\,y_i\,yb_i}{n_i\,{a_i}^4} + 9\,\frac{I\,K*(M+mu[i])\,{a_e}^2\,J_2\,xb_i\,y_i\,yb_e}{n_i\,{a_i}^4} + 9\,\frac{I\,K*(M+mu[i])\,{a_e}^2\,J_2\,xb_i\,yb_i\,y_e}{n_i\,{a_i}^4} - 9\,\frac{I\,K*(M+mu[i])\,{a_e}^2\,J_2\,xb_i\,y_e\,yb_e}{n_i\,{a_i}^4} + \frac{3}{2}\,\frac{I\,K*(M+mu[i])\,{a_e}^2\,J_2\,xb_i}{n_i\,{a_i}^4}\right) e^{(I\,L_i)}$$

$$DVSi[2] =, \left(\frac{399}{64}\,\frac{K*(M+mu[i])\,a_e{}^2\,J_2\,x_i\,xb_i{}^2}{n_i\,a_i{}^5} - \frac{45}{8}\,\frac{K*(M+mu[i])\,a_e{}^2\,J_2\,x_i\,yb_i{}^2}{n_i\,a_i{}^5} + \frac{21}{2}\,\frac{K*(M+mu[i])\,a_e{}^2\,J_2\,x_i\,yb_i\,yb_e}{n_i\,a_i{}^5}\right.$$
$$- \frac{39}{8}\,\frac{K*(M+mu[i])\,a_e{}^2\,J_2\,x_i\,yb_e{}^2}{n_i\,a_i{}^5} - \frac{135}{4}\,\frac{K*(M+mu[i])\,a_e{}^2\,J_2\,xb_i\,y_i\,yb_i}{n_i\,a_i{}^5} + \frac{63}{2}\,\frac{K*(M+mu[i])\,a_e{}^2\,J_2\,xb_i\,y_i\,yb_e}{n_i\,a_i{}^5}$$
$$\left. + \frac{63}{2}\,\frac{K*(M+mu[i])\,a_e{}^2\,J_2\,xb_i\,yb_i\,y_e}{n_i\,a_i{}^5} - \frac{117}{4}\,\frac{K*(M+mu[i])\,a_e{}^2\,J_2\,xb_i\,y_e\,yb_e}{n_i\,a_i{}^5} + \frac{39}{8}\,\frac{K*(M+mu[i])\,a_e{}^2\,J_2\,xb_i}{n_i\,a_i{}^5}\right) e^{(I\,L_i)}$$

$$DVSi[3] =, \left(\frac{9}{4}\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2\,x_i\,xb_i}{n_i\,a_i{}^5} - 9\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2\,y_i\,yb_i}{n_i\,a_i{}^5} + 9\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2\,y_i\,yb_e}{n_i\,a_i{}^5}\right.$$
$$\left. + 9\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2\,yb_i\,y_e}{n_i\,a_i{}^5} - 9\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2\,y_e\,yb_e}{n_i\,a_i{}^5} + \frac{3}{2}\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2}{n_i\,a_i{}^5}\right) e^{(I\,L_i)}$$

$$DVSi[4] =, \left(-\frac{3}{4}\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2\,x_i\,yb_i}{n_i\,a_i{}^5} + \frac{3}{4}\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2\,x_i\,yb_e}{n_i\,a_i{}^5} - \frac{9}{4}\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2\,xb_i\,y_i}{n_i\,a_i{}^5}\right.$$
$$\left. + \frac{9}{4}\,\frac{I\,K*(M+mu[i])\,a_e{}^2\,J_2\,xb_i\,y_e}{n_i\,a_i{}^5}\right) e^{(I\,L_i)}$$

5.1.5. Equations de Lagrange pour deux satellites massifs

Aux équations relatives aux perturbations mutuelles, identiques à celles qu'on sait obtenir en 5.1.2 ou 5.1.3 pour les planètes, et à celles liées à l'aplatissement qu'on vient de décrire en 5.1.4, il faut maintenant ajouter les perturbations indirectes dues à l'aplatissement.

On utilise la fonction perturbatrice de satellite par l'aplatissement, construite en 3.6 sous le nom *USiIndab* ou *USiIndJcs* , et dont on peut éventuellement extraire une partie (ici par exemple les termes en J_2, $c_{2,\,p}$ et $s_{2,\,p}$, en annulant tous les autres coefficients d'aplatissement)

```
> Up:=subs(seq(J[k]=0,k=3..6),seq(seq(c[k,p]=0,p=1..k),k=3..6),
> seq(seq(s[k,p]=0,p=1..k),k=3..6),USiIndJcs): nops(Up);
```

$$1928$$

On peut aussi sélectionner par exemple les seuls termes en J_2 ou en $a_{2,\,2}$ ou en $b_{4,\,2}$:

```
> Up:=expand(coeff(USiIndJcs,J[2])*J[2]): nops(%);
```

$$248$$

```
> Up:=expand(coeff(USiIndab,a[2,2])*a[2,2]):nops(%);
```

$$182$$

```
> Up:=expand(coeff(USiIndab,b[4,2])*b[4,2]):nops(%);
```

$$251$$

Calcul des équations de Lagrange du satellite S_i avec cette sélection de fonction perturbatrice :

```
> DVSi:=EqLag(Up,"interieur",i,e):
> print(seq(nops(expand(DVSi[k])),k=1..4));
```

$$200,\ 440,\ 184,\ 162$$

Pour limiter l'affichage du résultat, voici comment restreindre le résultat à une inégalité (*ppi*, *ppj*, *ppe*) donnée . On y fait aussi la substitution des fonctions des excentricités et inclinaisons. Le degré 3 adopté dans la fonction mtaylor permet d'obtenir les développements au degré 2, c'est-à-dire le degré convenable pour une fonction perturbatrice développée au degré 3. Notons que cette perturbation étant d'ordre 2, il peut y apparaitre des inégalités dépendant de 3 longitudes moyennes.

Pour *ppi*, *ppj* et *ppe* nuls, on obtient ici un **terme séculaire non nul** dans l'équation en longitude moyenne et dans celle en excentricité (contrairement à la partie indirecte des perturbations héliocentriques de planètes qui ne produit pas de termes séculaires):

```
> ppi:=0:  ppj:=0:  ppe:=0:
> for k to 4 do:
> coeff(coeff(coeff(DVSi[k],eL[i],ppi),eL[j],ppj),eL[e],ppe):
> subs(phi[i]=sqrt(1-x[i]*xb[i]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]), C[e]=sqrt(1-y[e]*yb[e]),%):
> mtaylor(%,varxyije,3):
> print('DVSi['.k.']=',
> sort(collect(
> %,varxyije,distributed,factor),varxyije)*exp(I*(ppi*L[i]+ppj*L[j]+ppe
> *L[e])));
> od:
```

$$DVSi[1] =, 0$$

$$DVSi[2] =, -\frac{27}{16}\,\frac{K\,\mu_j\,{a_e}^2\,J_2\,x_i\,xb_j}{n_i\,a_i\,{a_j}^4} - \frac{27}{16}\,\frac{K\,\mu_j\,{a_e}^2\,J_2\,xb_i\,x_j}{n_i\,a_i\,{a_j}^4}$$

$$DVSi[3] =, \frac{9}{4}\,\frac{I\,K\,\mu_j\,{a_e}^2\,J_2\,x_j}{n_i\,a_i\,{a_j}^4}$$

$$DVSi[4] =, 0$$

Pour le satellite S_j on calcule d'abord sa fonction perturbatrice indirecte due au satellite S_i en permutant les indices i et j , par exemple :

```
> Up:=subs(i=ii,j=i,ii=j,Up):  op(1,Up);
```

$$-\frac{15}{4}\,\frac{K\,\mu_i\,a_j\,{a_e}^2\,J_2\,x_j\,y_i\,yb_e\,C_e\,C_i\,eL_i}{{a_i}^4\,{eL_j}^2}$$

```
> DVSj:=EqLag(Up,"interieur",j,e):
> print(seq(nops(expand(DVSj[k])),k=1..4));
```

$$200,\ 440,\ 184,\ 162$$

```
> ppi:=0:  ppj:=0:  ppe:=0:
> for k to 4 do:
> coeff(coeff(coeff(DVSj[k],eL[i],ppi),eL[j],ppj),eL[e],ppe):
> subs(phi[j]=sqrt(1-x[j]*xb[j]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]), C[e]=sqrt(1-y[e]*yb[e]),%):
> mtaylor(%,varxyije,3):
> print(‘DVSj[‘.k.‘]=‘,
> sort(collect(
> %,varxyije,distributed,factor),varxyije)*exp(I*(ppi*L[i]+ppj*L[j]+ppe
> *L[e])));
> od:
```

$$DVSj[1] =,\ 0$$

$$DVSj[2] =,\ -\frac{27}{16}\,\frac{K\,\mu_i\,{a_e}^2\,J_2\,x_i\,xb_j}{n_j\,a_j\,{a_i}^4} - \frac{27}{16}\,\frac{K\,\mu_i\,{a_e}^2\,J_2\,xb_i\,x_j}{n_j\,a_j\,{a_i}^4}$$

$$DVSj[3] =,\ \frac{9}{4}\,\frac{I\,K\,\mu_i\,{a_e}^2\,J_2\,x_i}{n_j\,a_j\,{a_i}^4}$$

$$DVSj[4] =,\ 0$$

5.2. Equations canoniques

Les développements de fonctions perturbatrices obtenus en variables régulières x, xb, y, yb et en fonction du demi-grand axe et de la longitude moyenne de chaque corps peuvent être convertis en variables canoniques, par exemple celles de Poincaré (voir [1] et [4]) qui s'en rapprochent le plus et sont définies par :

$$\Lambda = \sqrt{\mu\, a} \qquad \text{conjuguée à la longitude moyenne} \qquad L$$

et les variables complexes régulières (où ϖ et Ω sont les longitudes du péricentre et de nœud):

$$\xi = \sqrt{\Lambda\,(1 - \sqrt{1 - e^2})}\, \exp\,(I\, \varpi) \qquad \text{et} \qquad \zeta = \sqrt{2\,\Lambda\, \sqrt{1 - e^2}\, \sin(\frac{i}{2})}\, \exp\,(I\, \Omega)$$

On a alors les relations entre x et ξ et entre y et ζ et leurs conjugués $\overline{\xi}$ et $\overline{\zeta}$:

$$x = \xi \sqrt{\frac{2}{\Lambda}} \sqrt{1 - \frac{\xi\, \overline{\xi}}{2\,\Lambda}} \qquad \text{et} \qquad y = \frac{\zeta}{\sqrt{2\,\Lambda\,(1 - \frac{\xi\,\overline{\xi}}{\Lambda})}}$$

En posant $u = \xi \sqrt{\frac{2}{\Lambda}}$ et $U = \overline{u}\, \exp\,(I\, L)$ ainsi que $v = \frac{\zeta}{\sqrt{2\,\Lambda}}$ et $V = \overline{v}\, \exp\,(I\, L)$ on peut écrire plus simplement:

$$x = u \sqrt{1 - \frac{u\, \overline{u}}{4}} \qquad \text{et} \qquad X = U \sqrt{1 - \frac{U\, \overline{U}}{4}}$$

ainsi que

$$y = \frac{v}{\sqrt{1 - \frac{u\,\overline{u}}{2}}} \qquad \text{et} \qquad Y = \frac{V}{\sqrt{1 - \frac{U\,\overline{U}}{2}}}$$

On pourra donc passer par simple substitution des variables initiales aux variables de Poincaré, par exemple:

```
> R:=TRONC(RSA,5,[X,Xb]);
```

$$R := -\frac{125}{768}X^5 + \frac{45}{256}X^4\,Xb - \frac{5}{384}X^3\,Xb^2 - \frac{5}{384}X^2\,Xb^3 + \frac{45}{256}X\,Xb^4 - \frac{125}{768}Xb^5 - \frac{1}{6}X^4 + \frac{1}{6}X^3\,Xb + \frac{1}{6}X\,Xb^3 - \frac{1}{6}Xb^4 - \frac{3}{16}X^3 + \frac{3}{16}X^2\,Xb + \frac{3}{16}X\,Xb^2 - \frac{3}{16}Xb^3 - \frac{1}{4}X^2 + \frac{1}{2}X\,Xb - \frac{1}{4}Xb^2 - \frac{1}{2}X - \frac{1}{2}Xb + 1$$

```
> S:=mtaylor(sqrt(1-U*Ub/4),[U,Ub],7);
```

$$S := 1 - \frac{1}{8}U\,Ub - \frac{1}{128}U^2\,Ub^2 - \frac{1}{1024}U^3\,Ub^3$$

```
> SUBS([X=expand(U*S),Xb=expand(Ub*S)],R,5,[U,Ub]);
```

$$-\frac{125}{768}U^5 + \frac{63}{256}U^4\,Ub - \frac{61}{768}U^3\,Ub^2 - \frac{61}{768}U^2\,Ub^3 + \frac{63}{256}U\,Ub^4 - \frac{125}{768}Ub^5 - \frac{1}{6}U^4 + \frac{11}{48}U^3\,Ub - \frac{1}{8}U^2\,Ub^2 + \frac{11}{48}U\,Ub^3 - \frac{1}{6}Ub^4 - \frac{3}{16}U^3 + \frac{1}{4}U^2\,Ub + \frac{1}{4}U\,Ub^2 - \frac{3}{16}Ub^3 - \frac{1}{4}U^2 + \frac{1}{2}U\,Ub - \frac{1}{4}Ub^2 - \frac{1}{2}U - \frac{1}{2}Ub + 1$$

Les variables canoniques conjuguées sont alors: $(\xi, -I\,\overline{\xi})$ et $(\zeta, -I\,\overline{\zeta})$, et les équations canoniques sont de la forme:

$$\frac{d\xi}{dt} = I\,\frac{\partial F}{\partial \overline{\xi}} \qquad \text{et} \qquad \frac{d\overline{\xi}}{dt} = -I\,\frac{\partial F}{\partial \xi}$$

donc aussi plus simples à calculer. D'ailleurs, l'une de ces deux équations suffit *a priori* puisqu'elles sont conjuguées aussi au sens des nombres complexes.

Dans cette version, on laisse au lecteur le loisir de développer cette formulation des équations.

6. Bibliographie

[1] L. Duriez, 1989, "Les méthodes modernes de la Mécanique Céleste", Goutelas 1989, Editions Frontières, D. Benest et C. Froeschlé eds. pp 9-62

[2] L. Duriez, 2007, "Cours de Mécanique Céleste", Licence de Mathématiques, Université de Lille 1.

[3] J. Laskar, 1985, "Accurate methods in general planetary theory", Astron. Astrophys. 144,133-146

[4] J. Laskar, 1989, "Les méthodes modernes de la Mécanique Céleste", Goutelas 1989, Editions Frontières, D. Benest et C. Froeschlé eds. pp 63-107

C

Contents